
Change 1 – Common Cell Configuration
Change 1.1 – New global variable DiscardULDataUM

	Type/Function name
	EUTRA_Ctrl_Type
fl_EUTRA_Common_Init

f_EUTRA_DiscardUlDataUM_Set (new)

f_EUTRA_DiscardUlDataUM_Get (new)

	Reason for change
	DRBs are not preconfigured anymore but configured dynamically at the SS. Therefore DRB configuration is done within common functions which don't know about whether or not UL UM data needs to be discarded

	Summary of changes
	New global variable to store whether or not SS shall discard UM data in UL

	TTCN module
	EUTRA_Component


Change 2 – Generic procedures for RB establishment
Change 2.1 – New local type definition

	Type name
	DRB_ConfigurationAtSS_Type

	Reason for change
	To support dynamic DRB configuration in several procedures for RB establishment we need to know whether DRBs are pre-configured already, default DRB configuration can be applied or specific DRB configuration needs to be applied.

	Summary of changes
	New type definition for parameter of procedures for RB establishment

	TTCN module
	EUTRA_NASSteps


	  type union DRB_ConfigurationAtSS_Type {

    RadioBearerList_Type  Specific,

    Null_Type             Preconfigured,

    Null_Type             Default

  };

	  template (value) DRB_ConfigurationAtSS_Type cs_DRB_ConfigurationAtSS_Specific(template (value) RadioBearerList_Type p_RadioBearerList) := {Specific := p_RadioBearerList};

	  const DRB_ConfigurationAtSS_Type tsc_DRB_ConfigurationAtSS_Preconfigured := {Preconfigured := true};

	  const DRB_ConfigurationAtSS_Type tsc_DRB_ConfigurationAtSS_Default := {Default := true};


Change 2.2 – f_EUTRA_SS_DRB_AddConfig replaced by f_EUTRA_SS_GetDrbConfigList
	Function name
	f_EUTRA_SS_DRB_AddConfig

	Reason for change
	f_EUTRA_Activate_SRB2_DRB_SendRrcMsg needs to get DRB configuration before DRBs gets configured at the SS (see next change)

	Summary of changes
	New function f_EUTRA_SS_GetDrbConfigList;
f_EUTRA_SS_DRB_AddConfig removed as not being used any longer

	TTCN module
	EUTRA_ConfigurationSteps


	  function f_EUTRA_SS_GetDrbConfigUM(DRB_Identity p_Drb,

                                     template (value) PDCP_Config p_PdcpConfig := cs_508_PDCP_Config_DRB_UM,

                                     template (value) RLC_Configuration_Type p_RlcConfig := cs_RLC_Configuration_UM) runs on EUTRA_PTC return template (value) RadioBearer_Type

  {

    var boolean v_DiscardULDataUM := ispresent(f_EUTRA_DiscardUlDataUM_Get());

    return cs_OneDRB_ConfigUM(p_Drb, p_PdcpConfig, p_RlcConfig, v_DiscardULDataUM);

  }

	  function f_EUTRA_SS_GetDrbConfigList(integer p_NoOfAddDRBs_AM,

                                       integer p_NoOfAddDRBs_UM,

                                       DRB_Identity p_FirstAddDRB) runs on EUTRA_PTC return template (omit) RadioBearerList_Type

  { /* @sic R5s130681: optional parameter p_DiscardULDataUM sic@ */

    /* @sic R5s141286 change 8: additional parameter p_FirstAddDRB sic@ */

    var template (omit) RadioBearerList_Type v_DRB_ToAddModList := omit;

    var integer v_DRB_Id;

    var integer n, m;

    if (p_NoOfAddDRBs_AM + p_NoOfAddDRBs_UM > 0) {          // @sic R5s100100 sic@

      v_DRB_Id := p_FirstAddDRB;

      for (n := 0; n < p_NoOfAddDRBs_AM; n := n + 1) {

        v_DRB_ToAddModList[n] := cs_OneDRB_ConfigAM(v_DRB_Id);

        v_DRB_Id := v_DRB_Id + 1;

      }

      for (m := 0; m < p_NoOfAddDRBs_UM; m := m + 1) { 
        v_DRB_ToAddModList[n+m] := f_EUTRA_SS_GetDrbConfigUM(v_DRB_Id);
        v_DRB_Id := v_DRB_Id + 1;

      }

    }

    return v_DRB_ToAddModList;

  }


Change 2.3 – New type used as parameter for f_EUTRA_Activate_SRB2_DRB_SendRrcMsg

	Function name
	f_EUTRA_Activate_SRB2_DRB_SendRrcMsg

	Reason for change
	DRBs shall be configured dynamically at the SS whenever an RRC connection gets reconfigured with changing the DRB configuration (RadioResourceConfigDedicated.drb_ToAddModList != omit);

This is the case in the common function f_EUTRA_RRCConnectionReconfiguration_Common;

Nevertheless DRBs shall be added before AS security and therefore before calling f_EUTRA_RRCConnectionReconfiguration_Common

	Summary of changes
	DRB configuration at SS just before AS security; three cases are considered:

a) DRBs are configured already (DRB_ConfigurationAtSS_Type.Preconfigured)

b) DRBs need to be configured as per default in 36.508 (DRB_ConfigurationAtSS_Type.Default)
NOTE: new function f_EUTRA_SS_GetDrbConfig introduced to allow generic implementation

c) Specific DRB configuration shall be applied (DRB_ConfigurationAtSS_Type.Specific)

	TTCN module
	EUTRA_NASSteps


	  function f_EUTRA_Activate_SRB2_DRB_SendRrcMsg(EUTRA_CellId_Type p_CellId,

                                                RRC_TransactionIdentifier p_RRC_TI,

                                                NasCount_Type p_NasCountUL,

                                                UInt_Type p_NoOfAddDRBs_AM := 0,

                                                UInt_Type p_NoOfAddDRBs_UM := 0, 
                                                template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default,
                                                template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                                template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,

                                                template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,

                                                template (omit) EPS_QoSList_Type p_QoS := omit,

                                                template (omit) Tft_List_Type p_Tft := omit,

                                                boolean p_PDCP_StatusReportRequired := true,

                                                template (omit) RLF_TimersAndConstants_r9 p_RLF_TimersAndConstants_r9 := omit,

                                                template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit,

                                                template (omit) DRB_Identity p_DrbFor2ndPDN := omit) runs on EUTRA_PTC

  { /* @sic R5-102180: Establishment of SRB2 and DRBs is split into 2 procedures for sending/receiving sic@ */

    /* @sic R5-144797: new parameter p_DrbFor2ndPDN sic@ */
    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();
    var template (omit) RadioBearerList_Type v_DrbConfigList; 
    var integer n;
    // =====================================================

    //

    if (ischosen(p_DrbConfigAtSS.Specific)) {                   /* @sic R5s150338: new parameter p_DrbConfigAtSS sic@ */

      v_DrbConfigList := p_DrbConfigAtSS.Specific;

    }

    else if (ischosen(p_DrbConfigAtSS.Default)) {               /* @sic R5s150338: new parameter p_DrbConfigAtSS sic@ */

      v_DrbConfigList := f_EUTRA_SS_GetDrbConfigList(p_NoOfAddDRBs_AM + 1, p_NoOfAddDRBs_UM, tsc_DRB1);     /* default DRB + additional AM DRBs */

      if (ispresent(p_DrbFor2ndPDN)) {

        n := lengthof(v_DrbConfigList);

        v_DrbConfigList[n] := cs_OneDRB_ConfigAM(valueof(p_DrbFor2ndPDN));

      }

    }
    /* step 6-8 of procedure in 36.508 cl. 4.5.3.3 */

    v_AuthParams := f_EUTRA_Authentication_InitAS(v_AuthParams, p_NasCountUL);

    v_AuthParams := f_EUTRA_RRC_ActivateSecurity(p_CellId, v_AuthParams, p_NasCountUL, v_DrbConfigList);

    f_EUTRA_Security_Set(v_AuthParams);

    /* Step 8 of procedure in 36.508 cl. 4.5.3.3 */

    f_EUTRA_RRCConnectionReconfiguration_Common(p_CellId,

                                                p_RRC_TI,

                                                p_NoOfAddDRBs_AM,

                                                p_NoOfAddDRBs_UM,

                                                {cs_508_SRB2_ToAddMod_DEFAULT},

                                                p_DRB_ToAddModList,

                                                p_MAC_MainConfig,

                                                p_PhysicalConfigDedicated,

                                                p_QoS,

                                                p_Tft,

                                                p_PDCP_StatusReportRequired,

                                                p_RLF_TimersAndConstants_r9,// @sic R5s120808 sic@

                                                p_DownlinkAntennaGroupConfig,

                                                p_DrbFor2ndPDN);

  }


Change 2.4 – Parameter p_DrbConfigAtSS for functions using f_EUTRA_Activate_SRB2_DRB_SendRrcMsg
	Function name
	(several functions)

	Reason for change
	Due to change above parameter p_DrbConfigAtSS (DRB_ConfigurationAtSS_Type) needs to be provided in other function calls

	Summary of changes
	parameter p_DrbConfigAtSS added to several functions; appropriate information handed over

	TTCN module
	(several modules)


	  function f_EUTRA_ServiceRequestAndActivate_SRB2_DRB(EUTRA_CellId_Type p_CellId,

                                                      UInt_Type p_NoOfAddDRBs_AM := 0,

                                                      UInt_Type p_NoOfAddDRBs_UM := 0, 
                                                      template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default,
                                                      template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                                      template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,

                                                      template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,

                                                      template (omit) EPS_QoSList_Type p_QoS := omit,

                                                      template (omit) Tft_List_Type p_Tft := omit,

                                                      boolean p_PDCP_StatusReportRequired := true,

                                                      template (present) EstablishmentCause p_EstablishmentCause := ?,

                                                      template (omit) RLF_TimersAndConstants_r9 p_RLF_TimersAndConstants_r9 := omit,

                                                      template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit,

                                                      template (omit) DRB_Identity p_DrbFor2ndPDN := omit) runs on EUTRA_PTC

  { /* @sic R5-144797: new parameter p_DrbFor2ndPDN sic@ */

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_KsiValue v_KsiValue;

    // =====================================================

    // step 3-10 of procedure in 36.508 cl. 4.5.3.3

    // Step 3 - 5 of Generic RB establishment procedure

    //Check the Service Request message

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                               v_RRC_TI,

                                               p_EstablishmentCause, //@sic R5s100539 sic@

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(v_KsiValue)));

    /* Step 6 - 8 */

    f_EUTRA_Activate_SRB2_DRB_SendRrcMsg(p_CellId,

                                         v_RRC_TI,

                                         v_NasInd.SecurityProtection.NasCount,

                                         p_NoOfAddDRBs_AM,

                                         p_NoOfAddDRBs_UM, 
                                         p_DrbConfigAtSS,
                                         p_DRB_ToAddModList,

                                         p_MAC_MainConfig,

                                         p_PhysicalConfigDedicated,

                                         p_QoS,

                                         p_Tft,

                                         p_PDCP_StatusReportRequired,

                                         p_RLF_TimersAndConstants_r9, // @sic R5s120808 sic@

                                         p_DownlinkAntennaGroupConfig,

                                         p_DrbFor2ndPDN);

    /* Step 9 - 10 */

    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(p_CellId,

                                                  v_RRC_TI,

                                                  p_NoOfAddDRBs_AM,

                                                  p_NoOfAddDRBs_UM);          /* @sic R5s100315 sic@ */

  }

	  function f_EUTRA_GenericRbEst_Common(EUTRA_CellId_Type p_CellId,

                                       integer p_NoOfAddDRBs_AM,

                                       integer p_NoOfAddDRBs_UM, 
                                       template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default,
                                       template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                       template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,

                                       template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,

                                       template (omit) EPS_QoSList_Type p_QoS := omit,

                                       template (omit) Tft_List_Type p_Tft := omit,

                                       boolean p_PDCP_StatusReportRequired := true,

                                       template (omit) RLF_TimersAndConstants_r9 p_RLF_TimersAndConstants_r9 := omit,

                                       template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit,

                                       template (omit) DRB_Identity p_DrbFor2ndPDN := omit)

    runs on EUTRA_PTC

  { /* @sic R5s130703 additional change: parameters p_DRB_ToAddModList, p_MAC_MainConfig, p_PhysicalConfigDedicated, p_QoS, p_Tft made optional sic@ */

    /* @sic R5-144797: new parameter p_DrbFor2ndPDN sic@ */

    var GutiParameters_Type v_GutiParameters := f_EUTRA_CellInfo_GetGuti(p_CellId);

    /* Step 2 */

    f_EUTRA_UE_Page(p_CellId,

                    cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParameters.MME_Code, v_GutiParameters.M_TMSI))); //@sic R5s100398 sic@

    // Steps 3 - 10

    f_EUTRA_ServiceRequestAndActivate_SRB2_DRB(p_CellId,

                                               p_NoOfAddDRBs_AM,

                                               p_NoOfAddDRBs_UM, 
                                               p_DrbConfigAtSS,
                                               p_DRB_ToAddModList,

                                               p_MAC_MainConfig,

                                               p_PhysicalConfigDedicated,

                                               p_QoS,

                                               p_Tft,

                                               p_PDCP_StatusReportRequired,

                                               -,

                                               p_RLF_TimersAndConstants_r9, // @sic R5s120808 sic@

                                               p_DownlinkAntennaGroupConfig,

                                               p_DrbFor2ndPDN);

  }

	  function fl_EUTRA_GenericRbEst_DisableBSR_PHR(EUTRA_CellId_Type p_CellId,

                                                integer p_NoOfAddDRBs_AM,

                                                integer p_NoOfAddDRBs_UM, 
                                                template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default,
                                                template (omit) EPS_QoSList_Type p_QoS := omit,

                                                template (omit) Tft_List_Type p_Tft := omit,

                                                boolean p_PDCP_StatusReportRequired := true) runs on EUTRA_PTC

  {

    var template (omit) DRB_ToAddModList v_DRB_ToAddModList := omit;                // => apply default setiing acc. 36.508 cl. 4.8.2.1.7 */

    var template (value) MAC_MainConfig_Type v_MAC_MainConfig ;

    var template (omit) PhysicalConfigDedicated v_PhysicalConfigDedicated := omit;  // => apply default setting acc. 36.508 cl. 4.8.2.1.6 */

    if (p_NoOfAddDRBs_UM > 0) {

      v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS_DisableBSR_PHR; // short DRX cycle acc. to note in 36.508 table 4.6.3-16

    } else {

      v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL_DisableBSR_PHR; // long DRX cycle acc. to note in 36.508 table 4.6.3-16

    }

    f_EUTRA_GenericRbEst_Common(p_CellId,

                                p_NoOfAddDRBs_AM,

                                p_NoOfAddDRBs_UM, 
                                p_DrbConfigAtSS,
                                v_DRB_ToAddModList,

                                v_MAC_MainConfig,

                                v_PhysicalConfigDedicated,

                                p_QoS,

                                p_Tft,

                                p_PDCP_StatusReportRequired);

	  function f_EUTRA_RbEst_508_UM_DisableBSR_PHR(EUTRA_CellId_Type p_CellId,

                                               template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default) runs on EUTRA_PTC

  { /* additional UM bearer will be established */

    var template (value) EPS_QoSList_Type v_Qos_List := {cs_508_EPS_QoS_Dedicated_1};

    var template (value) Tft_List_Type v_Tft_List := {f_EUTRA_Get_508_TrafficFlowTemplate(1)};

    fl_EUTRA_GenericRbEst_DisableBSR_PHR(p_CellId, 0, 1, p_DrbConfigAtSS, v_Qos_List, v_Tft_List);

  }

	  function f_EUTRA_LoopBackActivation_State4_12_2(EUTRA_CellId_Type p_CellId,

                                                  template (value) UE_TestLoopModeA_LB_Setup_Type p_UE_TestLoopModeA_LB_Setup,

                                                  integer p_NoOfAddDRBs_AM,

                                                  integer p_NoOfAddDRBs_UM,

                                                  template (omit) EPS_QoSList_Type p_QoS,

                                                  template (omit) Tft_List_Type p_Tft,

                                                  template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit) runs on EUTRA_PTC

  {


    var DRB_ConfigurationAtSS v_DrbConfigAtSS_Type := tsc_DRB_ConfigurationAtSS_Preconfigured;     // DRBs are configured at the SS already
    var template (omit) MAC_MainConfig_Type v_MAC_MainConfig := omit;               // => apply default setting acc. 36.508 cl. 4.8.2.1.5 */

    var template (omit) PhysicalConfigDedicated v_PhysicalConfigDedicated := omit;  // => apply default setting acc. 36.508 cl. 4.8.2.1.6 */

    f_EUTRA_GenericRbEst_Common(p_CellId, /* @sic R5s100293 sic@

                                             @sic R5-104796 DRX support mandated at SS sic@ */

                                p_NoOfAddDRBs_AM,

                                p_NoOfAddDRBs_UM, 
                                v_DrbConfigAtSS,
                                p_DRB_ToAddModList,

                                v_MAC_MainConfig,

                                v_PhysicalConfigDedicated,

                                p_QoS,

                                p_Tft);   // bring UE to State 3

    f_EUTRA_CloseUE_TestLoopModeA(p_CellId, p_UE_TestLoopModeA_LB_Setup);             // Step 1-2

  }

	  function f_EUTRA_GenericRbEst_DisableBSR_PHR(EUTRA_CellId_Type p_CellId,

                                               integer p_NoOfAddDRBs_AM := 0,

                                               integer p_NoOfAddDRBs_UM := 0, 
                                               template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default,
                                               template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                               template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,

                                               template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,

                                               template (omit) EPS_QoSList_Type p_QoS := omit,

                                               template (omit) Tft_List_Type p_Tft := omit,

                                               boolean p_PDCP_StatusReportRequired := true, //@sic R5s100852 p_PDCP_StatusReportRequired added sic@

                                               template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit)

    runs on EUTRA_PTC

  {

    var template (value) MAC_MainConfig_Type v_MAC_MainConfig;

    if (isvalue(p_MAC_MainConfig)) {

      v_MAC_MainConfig := valueof(p_MAC_MainConfig);

    } else {

      if (p_NoOfAddDRBs_UM > 0) {

        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS_DisableBSR_PHR; // short DRX cycle acc. to note in 36.508 table 4.6.3-16

      } else {

        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL_DisableBSR_PHR; // long DRX cycle acc. to note in 36.508 table 4.6.3-16

      }

    }

    f_EUTRA_GenericRbEst_Common(p_CellId,

                                p_NoOfAddDRBs_AM,

                                p_NoOfAddDRBs_UM,

                                p_DRB_ToAddModList,

                                v_MAC_MainConfig,

                                p_PhysicalConfigDedicated,

                                p_QoS,

                                p_Tft,

                                p_PDCP_StatusReportRequired,

                                -,

                                p_DownlinkAntennaGroupConfig);

  }

	  function f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling_DSR(

                                                                          EUTRA_CellId_Type p_CellId,

                                                                          Dsr_TransMax_Type p_Dsr_TransMax,

                                                                          boolean p_PDCP_StatusReportRequired := true) //@sic CA test Development)

    runs on EUTRA_PTC

  { //@sic R5-101183 R5s110576 sic@

    var template (omit) MAC_MainConfig_Type v_MAC_MainConfig := cs_MAC_MainConfig_Explicit_DisableBSR_TA_Infinity;   // time alignment timer is infinity

    var template (omit) PhysicalConfigDedicated v_PhysicalConfigDedicated := f_PhysicalConfigDedicated_Dsr_TransMax(p_CellId, p_Dsr_TransMax); 
    var template (omit) DRB_ToAddModList v_DRB_ToAddModList:= omit;

    var DRB_ConfigurationAtSS_Type v_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default;

    f_EUTRA_GenericRbEst_DisableBSR_PHR(p_CellId,

                                        0,

                                        0, 
                                        v_DrbConfigAtSS,
                                        v_DRB_ToAddModList,

                                        v_MAC_MainConfig,

                                        v_PhysicalConfigDedicated,

                                        omit,

                                        omit,

                                        p_PDCP_StatusReportRequired);     // bring UE to State 3

    f_EUTRA_CloseUE_TestLoopModeA(p_CellId, cs_UE_TestLoopModeA_LB_Setup_NoScaling);             // Step 1-2

  }


Further related changes: additional parameter tsc_DRB_ConfigurationAtSS_Default when calling

· f_EUTRA_GenericRbEst_Common in 

· f_EUTRA_GenericRbEst

· f_EUTRA_GenericRbEst_DisablePHR

· f_EUTRA_GenericRbEst_TAInfinity

· f_EUTRA_IdleUpdated_ReleaseSecondPDN

· f_EUTRA_LoopBackActivation_State4_71315

· f_EUTRA_LoopBackActivation_State4_7147a

· f_EUTRA_LoopBackActivation_State4_LongOnTimer_DrxS

· f_EUTRA_LoopBackActivation_State4_NoDrx

· f_TC_10_8_7_EUTRA

· f_TC_7_1_4_11_EUTRA

· f_TC_8_1_1_6_EUTRA

· f_TC_8_5_1_3_EUTRA

· f_TC_8_6_4_8_EUTRA

· fl_EUTRA_GenericRbEst_DSR_TransMax

· f_EUTRA_GenericRbEst_DisableBSR_PHR in

· f_EUTRA_LoopBackActivation_State4_7123

· f_EUTRA_LoopBackActivation_State4_7131

· f_EUTRA_LoopBackActivation_State4_71314

· f_EUTRA_LoopBackActivation_State4_71412

· f_EUTRA_LoopBackActivation_State4_7143

· f_EUTRA_LoopBackActivation_State4_7146

· f_EUTRA_LoopBackActivation_State4_7147_71413

· f_EUTRA_LoopBackActivation_State4_7148

· f_EUTRA_LoopBackActivation_State4_717

· f_EUTRA_LoopBackActivation_State4_717_MIMO

· f_EUTRA_LoopBackActivation_State4_DSR_TransMax

· f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling_AntennaInfo_ExplicitValue

· f_TC_7_2_3_9_EUTRA

· f_TC_7_2_3_9_EUTRA

· f_TC_7_3_5_2_EUTRA

· f_TC_7_3_5_4_EUTRA

· fl_EUTRA_GenericRbEst_WithRLC_AM_Params

· f_EUTRA_ServiceRequestAndActivate_SRB2_DRB in 

· f_EUTRA_IPCAN_XCAP_Signalling

· f_EUTRA_IdleUpdated_ReleaseSecondPDN

· f_EUTRA_RbEst_NoPaging

· f_TC_19_3_2a_EUTRA

· f_TC_8_1_2_7_EUTRA

· fl_EUTRA_GenericIMS_MO_SpeechCall_11_2_6

· fl_EUTRA_IPCAN_ServiceRequestAndActivate_SRB2_DRB

· f_EUTRA_Activate_SRB2_DRB_SendRrcMsg in

· f_TC_17_4_7_EUTRA
( For all these functions DRBs are configured acc. to 36.508 clause 4.5.3.3
Change 2.5 – Merge of fl_EUTRA_GenericRbEst_DisableBSR_PHR and f_EUTRA_GenericRbEst_DisableBSR_PHR

	Function name
	fl_EUTRA_GenericRbEst_DisableBSR_PHR

f_EUTRA_GenericRbEst_DisableBSR_PHR

	Reason for change
	fl_EUTRA_GenericRbEst_DisableBSR_PHR can easily be replaced by f_EUTRA_GenericRbEst_DisableBSR_PHR

	Summary of changes
	· fl_EUTRA_GenericRbEst_DisableBSR_PHR replaced by f_EUTRA_GenericRbEst_DisableBSR_PHR

· parameter p_DrbConfigAtSS as used for f_EUTRA_Activate_SRB2_DRB_SendRrcMsg
· default values for parameters

Changes of f_EUTRA_GenericRbEst_DisableBSR_PHR require further changes for f_EUTRA_RbEst_Def_DisableBSR_PHR and f_EUTRA_RbEst_508_UM_DisableBSR_PHR (see below)

	TTCN module
	L2_CommonFunctions


	

























	  function f_EUTRA_GenericRbEst_DisableBSR_PHR(EUTRA_CellId_Type p_CellId,

                                               integer p_NoOfAddDRBs_AM := 0,

                                               integer p_NoOfAddDRBs_UM := 0, 
                                               template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default,
                                               template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                               template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,

                                               template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,

                                               template (omit) EPS_QoSList_Type p_QoS := omit,

                                               template (omit) Tft_List_Type p_Tft := omit,

                                               boolean p_PDCP_StatusReportRequired := true, //@sic R5s100852 p_PDCP_StatusReportRequired added sic@

                                               template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit)

    runs on EUTRA_PTC

  {

    var template (value) MAC_MainConfig_Type v_MAC_MainConfig ;

    if (isvalue(p_MAC_MainConfig)) {

      v_MAC_MainConfig := valueof(p_MAC_MainConfig);

    } else {

      if (p_NoOfAddDRBs_UM > 0) {

         v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS_DisableBSR_PHR; // short DRX cycle acc. to note in 36.508 table 4.6.3-16

      } else {

        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL_DisableBSR_PHR; // long DRX cycle acc. to note in 36.508 table 4.6.3-16

      }

    }

    f_EUTRA_GenericRbEst_Common(p_CellId,

                                p_NoOfAddDRBs_AM,

                                p_NoOfAddDRBs_UM, 
                                p_DrbConfigAtSS,
                                p_DRB_ToAddModList,

                                v_MAC_MainConfig,

                                p_PhysicalConfigDedicated,

                                p_QoS,

                                p_Tft,

                                p_PDCP_StatusReportRequired,

                                -,

                                p_DownlinkAntennaGroupConfig);

  }


Changes applied to 

	  function f_EUTRA_RbEst_Def_DisableBSR_PHR(EUTRA_CellId_Type p_CellId,

                                            boolean p_PDCP_StatusReportRequired := true) runs on EUTRA_PTC

  {


    f_EUTRA_GenericRbEst_DisableBSR_PHR(p_CellId, 0, 0, -, -, -, -, -, -, p_PDCP_StatusReportRequired);

  }

	  function f_EUTRA_RbEst_508_UM_DisableBSR_PHR(EUTRA_CellId_Type p_CellId,

                                               template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default) runs on EUTRA_PTC

  { /* additional UM bearer will be established */

    var template (value) EPS_QoSList_Type v_Qos_List := {cs_508_EPS_QoS_Dedicated_1};

    var template (value) Tft_List_Type v_Tft_List := {f_EUTRA_Get_508_TrafficFlowTemplate(1)};


    f_EUTRA_GenericRbEst_DisableBSR_PHR(p_CellId, 0, 1, p_DrbConfigAtSS, -, -, -, v_Qos_List, v_Tft_List);
  }


Change 2.6 – Removal of f_EUTRA_GenericRbEst_WithRLC_UM_Params

	Function name
	f_EUTRA_GenericRbEst_WithRLC_UM_Params

f_EUTRA_GenericRbEst_UM_Params

	Reason for change
	f_EUTRA_GenericRbEst_WithRLC_UM_Params is used by f_EUTRA_GenericRbEst_UM_Params only

	Summary of changes
	f_EUTRA_GenericRbEst_WithRLC_UM_Params integrated into f_EUTRA_GenericRbEst_UM_Params only

	TTCN module
	L2_CommonFunctions


	
















	  function f_EUTRA_GenericRbEst_UM_Params(EUTRA_CellId_Type p_CellId, 
                                          template (value) DRB_ConfigurationAtSS_Type p_DrbConfigAtSS := tsc_DRB_ConfigurationAtSS_Default,
                                          template (value) PDCP_Config  p_PDCP_Config,

                                          template (value) UL_UM_RLC p_UL_UM_RLC,

                                          template (value) DL_UM_RLC p_DL_UM_RLC,

                                          template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,

                                          template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,

                                          boolean p_PDCP_StatusReportRequired := true) runs on EUTRA_PTC

  {


    var template (value) EPS_QoSList_Type v_Qos_List := {cs_508_EPS_QoS_Dedicated_1};

    var template (value) Tft_List_Type v_Tft_List := {f_EUTRA_Get_508_TrafficFlowTemplate(1)};

    var template (omit) DRB_ToAddModList v_DRB_ToAddModList;

    if(p_PDCP_StatusReportRequired)  // @sic R5s110207 sic@

      {

        v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1);

      }

    else

      {

        v_DRB_ToAddModList[0] := cs_DRB_ToAddMod_NoPDCP_StatusReport (tsc_DRB1);

      }

    v_DRB_ToAddModList[1] := cs_DRB_ToAddMod_RLC_UM (tsc_DRB2,

                                                     p_PDCP_Config,

                                                     p_UL_UM_RLC,

                                                     p_DL_UM_RLC);





    f_EUTRA_GenericRbEst_DisableBSR_PHR(p_CellId,

                                        0,

                                        1,

                                        ,

                                        v_DRB_ToAddModList,

                                        p_MAC_MainConfig,

                                        p_PhysicalConfigDedicated,

                                        v_Qos_List,

                                        v_Tft_List);

  }


Change 3 – DRB configuration before activation of AS security

Change 3.1 – DRB configuration in f_EUTRA_RRC_ActivateSecurity
	Function name
	f_EUTRA_RRC_ActivateSecurity

	Reason for change
	For dynamic DRB configuration DRBs need to be configured at the SS whenever configured at the UE 

	Summary of changes
	During RRC connection establishment DRBs can be configured just before RRC security is established: In general all DRBs shall be configured which are configured at the UE by the RRCConnectionReconfiguration.

	TTCN module
	EUTRA_SecuritySteps


	  function f_EUTRA_RRC_ActivateSecurity(EUTRA_CellId_Type p_CellId,

                                        EUTRA_SecurityParams_Type p_SecurityParams,

                                        NasCount_Type p_NasCountUL ,

                                        template (omit) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_DRB1) runs on EUTRA_PTC

    return EUTRA_SecurityParams_Type

  {

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Authentication_InitAS(p_SecurityParams,

                                                                                    p_NasCountUL); /* Calculate Kenb and KRRCenc, KRRCint, KUPenc */

    if (ispresent(p_DrbConfigList)) {    

      f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, valueof(p_DrbConfigList));

    }
    // Read PDCP SQN for SRB1 and calcuate Ciphering Activation times

    v_SecurityParams.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetInitial(p_CellId);         /* note: this works only when timing info is now */

    // Note by default DRB is also pre-configured on SS side

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL (p_CellId,

                                              v_SecurityParams.AS_Integrity ,

                                              v_SecurityParams.AS_Ciphering); //@sic R5s110149 sic@

    f_EUTRA_RRC_InitialSecurityActivation(p_CellId,

                                          v_SecurityParams.AS_Integrity.Algorithm,

                                          v_SecurityParams.AS_Ciphering.Algorithm);    /* SecurityModeComplete is not ciphered

                                                                                          acc. to 36.331 cl. 5.3.4.3 and 5.3.1.1 */

    return v_SecurityParams;

  }


( Due to the above change the default DRB (DRB1) is configured by function implicitly by the following functions too:

	Function
	Comment

	f_EUTRA_ActivateAdditionalPDNOnDef
	further DRB is configured by calling f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig

	f_EUTRA_508_EstablishIMSEmergencyCall_Steps5_8
	

	f_EUTRA_508_IMSEmergencyCallLimitedService_Steps7_15
	

	f_EUTRA_IdleUpdated_Step5_13
	further AM DRB is configured in case of multiple PDN

	f_EUTRA_Authentication_NAS_AS_Security
	

	f_EUTRA_Activate_SRB2_DRB_SendRrcMsg
	per default function configures default DRB + n AM DRBs + m UM DRBs acc. to 36.508 cl. 4.5.3.2 and in addition a further AM DRB for 2nd PDN if needed (see change 2.3)

	f_EUTRA_SendUE_LocationInformation
	f_EUTRA_RRC_ActivateSecurity is called only when the UE is in IDLE mode, in which case the RRC connection gets release again at the end of the function

	f_EUTRA_PdcpScurityProcedure_Common_9_4,
f_EUTRA_PdcpScurityProcedure_Integrity_9_4
	


Change 3.2 – DRB configuration during inital registration
	Function name
	f_EUTRA_IdleUpdated_Step5_13, f_EUTRA_IdleUpdated_Step14_15

	Reason for change
	Due to dynamic DRB configuration DRB1 needs to be configured just before activation of RRC security; this is done per default in f_EUTRA_RRC_ActivateSecurity.

Nevertheless to reduce complexity the optional DRB of the 2nd PDN shall be configured in one go at the same time

	Summary of changes
	configuration of 2nd DRB moved from f_EUTRA_IdleUpdated_Step14_15 to f_EUTRA_IdleUpdated_Step5_13

	TTCN module
	EUTRA_InitialRegistration


	  function f_EUTRA_IdleUpdated_Step5_13(EUTRA_CellId_Type p_CellId,

                                        IDLEUPDATED_STATE_Type p_State,

                                        NAS_MSG_Indication_Type p_NAS_Ind) runs on EUTRA_PTC return APNandPCOs_Type

  { /* @sic R5-110796: split into f_EUTRA_IdleUpdated_Step5_13 and f_EUTRA_IdleUpdated_Step14_15 sic@

       the returned APNandPCOs_Type contains the valid APN (i.e. accessPointName is present) */

    var ProcedureTransactionIdentifier v_EPS_TI;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var template (present) SecurityHeaderType v_ExpectedSecurityHeader; // @sic R5s090373 sic@

    var NasCount_Type v_NasCountUL;

    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();

    var APNandPCOs_Type v_APNandPCOs;

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(p_CellId);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity); 
    var DRB_Identity v_DrbForInternet; 
    var template (value) RadioBearerList_Type v_DrbConfigList := cs_DrbConfigList_DRB1;
    // Set UE network capabilities as sent by the UE to be used later (see TS 24.301 clause 5.4.3.3)

    v_AuthParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability(p_NAS_Ind.Pdu.Msg.aTTACH_REQUEST);  // @sic R5s100135 sic@

    f_EUTRA_Security_Set(v_AuthParams);

    if (p_NAS_Ind.SecurityProtection.Status == tsc_SHT_NoSecurityProtection) { // @sic R5s090210, R5s090373 sic@

      v_ExpectedSecurityHeader := tsc_SHT_NoSecurityProtection;

    } else {

      v_ExpectedSecurityHeader := (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected);

    }

    v_PdnConnectivityRequest := p_NAS_Ind.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;

    // Step 5-6: Authentication

    v_AuthParams := f_EUTRA_NAS_Authentication(p_CellId,

                                               tsc_SRB1,

                                               v_AuthParams,

                                               tsc_SHT_NoSecurityProtection,

                                               v_ExpectedSecurityHeader,       // @sic R5s090210, R5s090373 sic@

                                               v_PLMN);                        // AUTHENTICATION REQUEST & RESPONSE

    // Step 7-8: NAS Security

    v_NasCountUL := f_EUTRA_NAS_ActivateSecurity(p_CellId, v_AuthParams);   // NAS SECURITY COMMAND & COMPLETE

    // Step 9a1 & 2 - ESM information optionally transferred;  @sic R5-110708 sic@

    v_APNandPCOs := f_ESM_InitAPNandPCOs(p_CellId, f_CheckEsmInfoTransferFlag(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST), v_EPS_TI);  // @sic R5s130382 sic@

    // optional

    if (p_State == STATE2A_TESTLOOP_ModeA) {

      f_EUTRA_Activate_TestMode (p_CellId, tsc_UE_TestLoopMode_TypeA);

    }

    else if (p_State == STATE2A_TESTLOOP_ModeB) {

      f_EUTRA_Activate_TestMode (p_CellId, tsc_UE_TestLoopMode_TypeB);

    }

    else if (p_State == STATE2A_TESTLOOP_ModeC) {

      f_EUTRA_Activate_TestMode (p_CellId, tsc_UE_TestLoopMode_TypeC); //@sic eMBMS sic@

    }

    if (tsc_EUTRA_Registration_MultiplePDN) {   /* @sic R5-144797, R5s150338 sic@ */

      v_DrbForInternet := f_EUTRA_EpsBearerAssociatedDRB(tsc_EpsDefaultBearerId2ndPDN);

      f_IP_Handling_Start(IP, PDN_Internet, cs_DrbInfo_EUTRA(p_CellId, v_DrbForInternet));

      v_DrbConfigList[1] := cs_OneDRB_ConfigAM(v_DrbForInternet);

    }

    // Step 10-11: RRC Security Mode procedures

    v_AuthParams := f_EUTRA_RRC_ActivateSecurity(p_CellId, v_AuthParams, v_NasCountUL, v_DrbConfigList); // RRC SECURITY COMMAND & COMPLETE

    f_EUTRA_Security_Set(v_AuthParams); //Saving security parameters

    // Step 12-13 UE Capability check

    f_EUTRA_Capability(p_CellId, tsc_RRC_TI_Def); // This enquires for the eutra capability but does not check the contents of the message received

    return v_APNandPCOs;

  }

	  function f_EUTRA_IdleUpdated_Step14_15(EUTRA_CellId_Type p_CellId,

                                         EUTRA_ATTACH_Type p_ForcedAttach,

                                         template (omit) AdditionalUpdateType p_AdditionalUpdateType,

                                         ProcedureTransactionIdentifier p_EPS_TI,

                                         boolean p_IPv4AllocationViaNasFlag,

                                         template (omit) ProtocolConfigOptions p_PcoToUE,

                                         AccessPointName p_APN,

                                         PDN_TypeToBeUsed_Type p_PDN_TypeToBeUsed := pdnTypeAsSupportedByUE,

                                         template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact,

                                         template (omit) GPRS_Timer p_T3402 := omit,

                                         template (omit) GPRS_Timer p_T3423 := omit,

                                         template (omit) PLMN_List p_EquivalentPlmnList := omit,

                                         integer p_NoOfEmergencyNumbers := 0, // @sic R5-115673 sic@

                                         template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice,

                                         template (omit) EMM_Cause p_EMM_Cause := omit) runs on EUTRA_PTC

  { /* @sic R5-112600: IP address allocation sic@ */

    /* @sic R5-110796: split into f_EUTRA_IdleUpdated_Step5_13 and f_EUTRA_IdleUpdated_Step14_15 sic@ */

    /* @sic R5-120234, R5-120541, R5-120542 sic@ */

    /* @sic R5-144370, R5-144797: p_APN shall not be omit sic@ */

    /* @sic R5-145971: new parameter p_EMM_Cause sic@ */

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5-103673, R5s100795 sic@

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(p_CellId);

    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@

    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(p_CellId);

    var template (omit) EPS_NetworkFeatureSupport v_NtwkFeatSupport := p_NtwkFeatSupport;

    var template (omit) EmergNumList v_EmergNumList := omit;

    var template (omit) MobileIdentity v_MSId := f_GetMSId(p_ForcedAttach); // @sic R5-103673 sic@

    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(firstPDN);               // @sic R5-144797 sic@

    var EPS_BearerIdentity v_EpsDefaultBearerId := f_EUTRA_InitialRegistrationEpsBearerId(firstPDN);  // @sic R5-144797 sic@


    var template (omit) ESM_Cause v_ESM_Cause := omit;        // @sic R5s120136 sic@

    var template (value) PDN_Address v_PDN_Address;

    var EmergencyNumList v_EmergencyNumList;

    var BEARER_CONTEXT_TYPE v_BearerContextNumber := DEF_1;

    if (pc_IPv6 and pc_IPv4) {              /* @sic R5-120234, R5-120541, R5-120542 sic@ */

      select (p_PDN_TypeToBeUsed) {

        case (forceIPv4only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110010'B); }     // PDN type IPv4 only allowed

        case (forceIPv6only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110011'B); }     // PDN type IPv6 only allowed

        case (pdnTypeAsSupportedByUE) { v_ESM_Cause := omit; }

      }

    }

    // @sic R5s130681 sic@

    select (f_GetTestcaseAttrib_Eutra_Release(testcasename())) { // @sic R5s130508 sic@

      case ("","rel_10") { // rel8 or rel-10 - 36.508 just says rel-9

      }

      case ("rel_9") {

        v_NtwkFeatSupport.imsVoPS := '1'B; // @sic R5s130902 sic@

        v_NtwkFeatSupport.emcBS := '1'B; // @sic R5s130977 sic@

      }

      case else {}

    }

    if (p_NoOfEmergencyNumbers > 0) {

      v_EmergencyNumList := f_Get_EmergencyNumList (p_NoOfEmergencyNumbers, f_EUTRA_MobileInfo_GetLocalEmergencyNumberList()); /* @sic R5s141315 sic@ */

      f_EUTRA_MobileInfo_SetLocalEmergencyNumberList (v_EmergencyNumList); // save the list of locally usable emergency numbers

      v_EmergNumList := f_Build_EmergNumList(p_NoOfEmergencyNumbers, v_EmergencyNumList);

    }

    if (tsc_EUTRA_Registration_PDN1isIMS) {     // @sic R5-144797, R5s141367 sic@

      v_BearerContextNumber := DEF_2;

    }






    v_PDN_Address := f_EUTRA_GetPDNAddress(p_IPv4AllocationViaNasFlag, v_PdnIndex, p_PDN_TypeToBeUsed);

    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(p_CellId,

                                                                  v_EpsDefaultBearerId,    // @sic R5-144797 sic@

                                                                  tsc_RRC_TI_Def,

                                                                  p_EPS_TI,

                                                                  v_AttachType,

                                                                  v_PDN_Address,

                                                                  v_Guti,

                                                                  cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  v_ESM_Cause, // @sic R5-104703 sic@

                                                                  p_APN,

                                                                  p_PcoToUE,

                                                                  f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@

                                                                  p_T3412,

                                                                  p_T3402,

                                                                  p_T3423,

                                                                  p_EquivalentPlmnList,

                                                                  v_EmergNumList,         // @sic R5s130681 sic@

                                                                  v_NtwkFeatSupport,      // @sic R5s130681 sic@

                                                                  -,

                                                                  -,

                                                                  v_BearerContextNumber, // @sic R5-144797 sic@

                                                                  p_EMM_Cause); // @sic R5-145971 sic@

  }


Change 3.3 – DRB configuration during RRC connection establishment
	Function name
	fl_TC_10_2_1_Body, fl_TC_10_4_1_Body, fl_TC_10_6_1_Body, fl_TC_10_7_2_Body, fl_TC_11_2_1_Body, fl_ReactivateEPS_BearerContext

	Reason for change
	In several test cases not only DRB1 is established during RRC connection establishment

	Summary of changes
	List of DRB configurations is handed over to f_EUTRA_RRC_ActivateSecurity

	TTCN module
	ESM_Testcases, IMS_EmergencyCall, MBMS_MCCH_TestCases


Change 3.3.1 - fl_TC_10_2_1_Body
	  function fl_TC_10_2_1_Body() runs on EUTRA_PTC

  {

    var ProcedureTransactionIdentifier v_EpsProcedureTransactionId := '00'O;

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NasCount_Type v_NasCountUL;

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_CellA); //Get FDD or TDD mode from Cell-configuration

    var NAS_MSG_Indication_Type v_NasInd;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA); //@sic R5-100786 sic@ 
    var template (value) RadioBearerList_Type v_DrbConfigList := { cs_OneDRB_ConfigAM(tsc_DRB1), cs_OneDRB_ConfigAM(tsc_DRB2) };
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1);

    v_DRB_ToAddModList[1] := cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB2); // @sic R5s110365 sic@

    //The SS pages the UE in order to establish a dedicated EPS bearer context.

    //@siclog "Step 1" siclog@

    f_EUTRA_UE_Page(eutra_CellA,

                    cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParams.MME_Code,

                                                                       v_GutiParams.M_TMSI)));

    // The UE transmits a SERVICE REQUEST for downlink signalling.

    //@siclog "Step 2" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_mt_Access,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));

    //  The SS establishes SRB2 and DRB associated with default EPS bearer context

    // (a first PDN obtained during the attach procedure).

    //@siclog "Step 2A" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL); 
    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL, v_DrbConfigList);

    f_EUTRA_Security_Set(v_SecurityParams);

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST with

    //        particular settings (use Test Reference dedicated EPS bearer context2 context #1 - see TS 36.508)

    /* Note: This message is included in a RRCConnectionReconfiguration message to

     * setup the new radio bearer associated with the dedicated EPS bearer context

     * activated by the NAS message

     */

    //@siclog "Step 3" siclog@

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                           omit,

                                                                           omit,

                                                                           cs_508_RadioResourceConfigDedicated_SRB2( v_DRB_ToAddModList,

                                                                                                                     cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                     cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                                v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                                v_AntennaInfo)),

                                                                           omit),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_ActDedicatedEPSBearerCxtReq(tsc_EpsDedicatedBearerId,

                                                                                      v_EpsProcedureTransactionId,

                                                                                      cs_LinkedId(hex2bit( tsc_EpsDefaultBearerId)),

                                                                                      cs_508_EPS_QoS_Dedicated_1, // According to Context#1 in TS 36.508

                                                                                      f_EUTRA_Get_508_TrafficFlowTemplate(1, tsc_EpsDedicatedBearerId)))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Step 4. Check: Does the UE transmit an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified?

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    //  Check: Does the test result of CALL generic procedure indicate

    //     that the UE can transmit data on the newly activated EPS bearer context ?

    //     TS 36.508 clause 6.4.2.6

    //@siclog "Step 5" siclog@

    f_EUTRA_508Check_DedicatedEPSIsActive(eutra_CellA,

                                          tsc_EpsDedicatedBearerId,

                                          cds_508_EPS_QoS_Dedicated_1_tlv,

                                          f_EUTRA_Get_508_TrafficFlowTemplate_2_Modify(tsc_EpsDedicatedBearerId));

  }


Change 3.3.2 - fl_TC_10_4_1_Body
	   function fl_TC_10_4_1_Body() runs on EUTRA_PTC

  {

    var EPS_BearerIdentity v_EpsBearerId2 := '6'H; // default bearer on additional PDN

    var EPS_BearerIdentity v_EpsDedicatedBearerId2 := '7'H; // dedicated bearer id on additional PDN

    var ProcedureTransactionIdentifier v_EpsTi2 := tsc_PTI_Unassigned;

    var ProcedureTransactionIdentifier v_PTId_UE;

    var template (omit) ProtocolConfigOptions v_Pco;  // @sic R5s100761 sic@

    var template (value) AccessPointName v_AccessPointName;

    var PDN_Address v_PDN_Address2;

    var NasCount_Type v_NasCountUL;

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var NAS_MSG_Indication_Type v_NasInd;

    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;

    var template (value) ESM_Cause v_EsmCause43 := cs_ESM_Cause_v ('00101011'B); // Invalid EPS bearer identity

    var template (value) ESM_Cause v_EsmCause36 := cs_ESM_Cause_v ('00100100'B); // Regular deactivation

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var GutiParameters_Type v_GutiParams;

    var NAS_PlmnId v_PLMN;

    var TrackingAreaCode v_Tac;

    var SRB_COMMON_IND v_ReceivedAsp;

    var template (value) TrackingAreaIdList v_TaiList;

    var template (value) CellPowerList_Type v_CellPowerList;

    var M_TMSI_Type v_M_Tmsi_Def := tsc_M_TMSI1;

    var GutiParameters_Type v_GutiParameters := f_EUTRA_CellInfo_GetGuti ( eutra_CellA );

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info ( eutra_CellA);

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA);

    var octetstring v_APN_Value := tsc_APN1; // APN-1(New PDN name), different from default

    var template (omit) LocAreaId v_LAI := omit;

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(NORMAL); // @sic R5s110176 sic@

    var template (value) RadioBearerList_Type v_DrbConfigList := {cs_OneDRB_ConfigAM(tsc_DRB1), cs_OneDRB_ConfigAM(tsc_DRB2), cs_OneDRB_ConfigAM(tsc_DRB3)};  // @sic R5s130077 sic@

    timer t_Wait := 1.0;

    // Set the cell type of cell A to the ''Serving cell''.

    // Set the cell type of cell B to the ''Suitable cell''.

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_ServingCellRS_EPRE  ),

      cs_CellPower ( eutra_CellB, tsc_SuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList ); // @sic R5s110498 sic@

    // Cause the UE to request connectivity to an additional PDN.

    //@siclog "Step 1" siclog@

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_string); // @sic R5s100466 sic@

    // The UE transmits a SERVICE REQUEST

    //@siclog "Step 1A" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    // The SS establishes SRB2 and DRB associated with default EPS bearer context obtained during the attach procedure.

    //@siclog "Step 1B" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA,

                                                     v_SecurityParams,

                                                     v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The UE transmits a PDN CONNECTIVITY REQUEST message as specified to request an additional PDN.

    //@siclog "Step 2" siclog@

    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_CellA,

                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

    v_AccessPointName := cs_AccessPointName(v_APN_Value);

    v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    // Change from DRB1 to DRB2 to handle U Plane data

    f_EUTRA_StartIPfor2ndPDN(eutra_CellA);  // @sic R5s110449, R5-113734 change 8 sic@

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message with IE EPS Bearer Identity set to new EPS bearer context.

    // Note: the SS allocates a PDN address of a PDN type which is compliant with the PDN type requested by the UE.

    //@siclog "Step 3" siclog@

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), PDN_2); // @sic R5s110011 R5-120234 sic@

    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefaultProtocolConfigOptions(v_Pco));

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 4" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling

    f_EUTRA_DelayForUserPlaneSignalling();

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST relative to the additional PDN.

    // Note: the same PDN address is applicable because the linked EPS bearer ID refers to the default EBC allocated in step 3.

    //@siclog "Step 5" siclog@

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,

                                        v_EpsDedicatedBearerId2,

                                        v_EpsTi2,

                                        cs_LinkedId(hex2bit( v_EpsBearerId2)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2));

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 6" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId2))));

    //  The SS releases the RRC connection.

    //@siclog "Step 7" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA, -, -, -, v_DrbConfigList);  // @sic R5s130077 sic@

    //  The SS transmits a Paging message to the UE using S-TMSI with CN domain indicator set to "PS".

    //@siclog "Step 8" siclog@

    f_EUTRA_UE_Page ( eutra_CellA,

                      cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParameters.MME_Code, v_M_Tmsi_Def )) );

    //  The UE transmits the SERVICE REQUEST message.

    //@siclog "Step 9" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));

    // The SS establishes SRB2 and DRBs associated with two default EPS bearer context (a first PDN obtained during the attach procedure and an additional PDN).

    //@siclog "Step 9A" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL, v_DrbConfigList);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2({cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1),

                                                                                                                  cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB2), // @sic R5s110365 sic@

                                                                                                                  cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB3)},

                                                                                                                 cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                 cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                            v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                            v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                  cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the dedicated EPS bearer to the additional PDN.

    //@siclog "Step 10" siclog@

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release (tsc_RRC_TI_Def, tsc_DRB3), // @sic R5s110275 sic@

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(v_EpsDedicatedBearerId2,

                                                                                    v_EpsTi2,

                                                                                    v_EsmCause36))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    //@siclog "Step 11" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_DeactivateEPSBearerCxtAccept(v_EpsDedicatedBearerId2,

                                                                                      v_EpsTi2))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST relative to additional PDN.

    // Note: the same PDN address is applicable because the linked EPS bearer ID refers to the default EBC allocated in step 3.

    //@siclog "Step 12" siclog@

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,

                                        v_EpsDedicatedBearerId2,

                                        v_EpsTi2,

                                        cs_LinkedId(hex2bit( v_EpsBearerId2)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2));

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 13" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId2))));

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the dedicated EPS bearer to the additional PDN.

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    //@siclog "Step 14-15" siclog@

    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,

                                         v_EpsBearerId2);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 15");

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the dedicated EPS bearer.

    // This message is included in a DLInformationTransfer message.

    //@siclog "Step 16" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsDedicatedBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             f_EUTRA_Get_508_TrafficFlowTemplate_2_Modify(v_EpsDedicatedBearerId2))))); // @sic R5-103625 sic@

    // Check: Does UE transmits a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43?

    //@siclog "Step 17" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsDedicatedBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST that does not point an existing EPS bearer context.

    //@siclog "Step 18" siclog@

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release (tsc_RRC_TI_Def, tsc_DRB2),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(v_EpsBearerId2,

                                                                                    v_EpsTi2,

                                                                                    v_EsmCause36))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    //@siclog "Step 19" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_DeactivateEPSBearerCxtAccept(v_EpsBearerId2,

                                                                                      v_EpsTi2))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 19");

    //  The SS releases the RRC connection.

    //@siclog "Step 20" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA, -, -, -, v_DrbConfigList);  // @sic R5s130077 sic@

    //  Cause the UE to request connectivity to an additional PDN.

    //@siclog "Step 21" siclog@

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_string); // APN-1(New PDN name), different from default @sic R5s100466 sic@

    // The UE transmits a SERVICE REQUEST.

    //@siclog "Step 21A" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    // The SS establishes SRB2 and DRBs associated with two default EPS bearer context (a first PDN obtained during the attach procedure and an additional PDN).

    //@siclog "Step 21B" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The UE transmits a PDN CONNECTIVITY REQUEST message as specified to request an additional PDN.

    //@siclog "Step 22" siclog@

    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_CellA,

                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

    v_AccessPointName := cs_AccessPointName(v_APN_Value);

    v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message with IE EPS Bearer Identity set to new EPS bearer context.

    // Note: the SS allocates a PDN address of a PDN type which is compliant with the PDN type requested by the UE.

    //@siclog "Step 23" siclog@

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), PDN_2); // @sic R5s110011 R5-120234 sic@

    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefaultProtocolConfigOptions(v_Pco));

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 24" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling

    f_EUTRA_DelayForUserPlaneSignalling();

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST relative to additional PDN.

    // Note: the same PDN address is applicable because the linked EPS bearer ID refers to the default EBC allocated in step 3.

    //@siclog "Step 25" siclog@

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,

                                        v_EpsDedicatedBearerId2,

                                        v_EpsTi2,

                                        cs_LinkedId(hex2bit( v_EpsBearerId2)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2));

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 26" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId2))));

    // The SS releases the RRC connection.

    //@siclog "Step 27" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA, -, -, -, v_DrbConfigList);  // @sic R5s130077 sic@

    // The SS transmits a Paging message to the UE using S-TMSI with CN domain indicator set to "PS".

    //@siclog "Step 28" siclog@

    f_EUTRA_UE_Page ( eutra_CellA,

                      cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParameters.MME_Code, v_M_Tmsi_Def )) );

    // The UE transmits SERVICE REQUEST message.

    //@siclog "Step 29" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));

    // activate security in AS

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    // The SS performs a radio bearer establishment procedure.

    // The RRCConnectionReconfiguration message doesn't include the EPS bearer ID of the additional PDN.

    //@siclog "Step 30" siclog@

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    // Check: Does UE transmit a RRCConnectionReconfigurationComplete message?

    //@siclog "Step 31" siclog@

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_CellA, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 31");

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the default EPS bearer to the additional PDN.

    // This message is included in a DLInformationTransfer message.

    //@siclog "Step 32" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             omit))));

    // Check: Does UE transmit a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43?

    //@siclog "Step 32A" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 32A");

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the dedicated EPS bearer.

    //@siclog "Step 32B" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsDedicatedBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             f_EUTRA_Get_508_TrafficFlowTemplate_2_Modify(v_EpsDedicatedBearerId2))))); // @sic R5-103625 sic@

    // Check: Does UE transmit a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43? (see Note 4)

    //@siclog "Step 32C" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsDedicatedBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 32C");

    // The SS releases the RRC connection.

    //@siclog "Step 33" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA, -, -, -, v_DrbConfigList);  // @sic R5s130077 sic@

    //  Cause the UE to request connectivity to an additional PDN.

    //@siclog "Step 34" siclog@

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_string); // APN-1(New PDN name), different from default @sic R5s100466 sic@

    // The UE transmits a SERVICE REQUEST

    //@siclog "Step 34A" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    // The SS establishes SRB2 and DRB associated with default EPS bearer context obtained during the attach procedure

    //@siclog "Step 34B" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The UE transmits a PDN CONNECTIVITY REQUEST message as specified to request an additional PDN.

    //@siclog "Step 35" siclog@

    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_CellA,

                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

    v_AccessPointName := cs_AccessPointName(v_APN_Value);

    v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message

    // with IE EPS Bearer Identity set to new EPS bearer context.

    //@siclog "Step 36" siclog@

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), PDN_2); // @sic R5s110011 R5-120234 sic@

    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefaultProtocolConfigOptions(v_Pco));

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 37" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling

    f_EUTRA_DelayForUserPlaneSignalling();

    //  The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST relative to additional PDN.

    //@siclog "Step 38" siclog@

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,

                                        v_EpsDedicatedBearerId2,

                                        v_EpsTi2,

                                        cs_LinkedId(hex2bit( v_EpsBearerId2)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2));

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 39" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId2))));

    // The SS releases the RRC connection.

    //@siclog "Step 40" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA, -, -, -, v_DrbConfigList);  // @sic R5s130077 sic@

    //  Cell A is a suitable cell, Cell B is the serving cell

    //@siclog "Step 41" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_SuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE  )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    // The UE transmit a TRACKING AREA UPDATE REQUEST message as specified on Cell B.

    //@siclog "Step 42" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cdr_TAU_Request_EPSBearerCtxtStatus( v_UpdateType,

                                                                                                       f_EUTRA_SecurityKSIasme_Get (),

                                                                                                       cr_DRXparameter_Any ('5C'O) ifpresent, // @sic R5s120210, R5-123117 sic@

                                                                                                       v_AdditionalUpdateType, // @sic R5-103681 sic@

                                                                                                       cs_EPSBearerCtxtStatus_3Ctxts)));

    // The SS establishes SRB2 and DRB associated with default EPS bearer context obtained during the attach procedure.

    //@siclog "Step 42A" siclog@

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;  // @sic R5s141327 sic@

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellB,

                                                     v_SecurityParams,

                                                     v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellB,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_CellB, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );

    // The SS transmits a TRACKING AREA UPDATE ACCEPT indicating only one EPS bearer (default EBId-1) active in the EPS bearer context status IE.

    // The EPS bearer ID linked to the additional PDN is deactivated by SS.

    //@siclog "Step 43" siclog@

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellB);

    v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellB );

    v_TaiList := cds_TAIListNonConsecutive_tlv( v_PLMN, { bit2oct(v_Tac) } );

    if (v_UpdateType == tsc_EpsUpdate_Combined_TaLaUpdate) // @sic R5s110086 sic@

    {

      v_LAI := f_EUTRA_GetCellLAI(eutra_CellB, NORMAL); // @sic R5s110176 sic@

    }

    SRB.send ( cas_SRB2_NasPdu_REQ ( eutra_CellB, // @sic R5s141327 sic@

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_TAU_Accept ( v_UpdateType,

                                                                          f_GutiParameters2MobileIdentity (tsc_IEI_Guti, v_GutiParams ),

                                                                          v_TaiList,

                                                                          cs_EPSBearerCtxtStatus_DefaultBearerOnly,

                                                                          v_LAI,

                                                                          omit, // v_MSId

                                                                          f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    // The UE transmits TRACKING AREA UPDATE COMPLETE message.

    //@siclog "Step 43AA" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB, // @sic R5s141327 sic@

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete))) -> value v_ReceivedAsp;

    // VOID   @sic R5-113188 sic@

    //@siclog "Step 43AB" siclog@

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the default EPS bearer to the additional PDN.

    // This message is included in a DLInformationTransfer message.

    //@siclog "Step 43A" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellB, // @sic R5s110593 sic@ @sic R5s141327 sic@

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             omit))));

    // Check: Does UE transmits a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43?

    //@siclog "Step 43B" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,  // @sic R5s110593 sic@ @sic R5s141327 sic@

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 43B");

    // Step 43C. The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the dedicated EPS bearer.

    //@siclog "Step 43C" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellB,  // @sic R5s110593 sic@ @sic R5s141327 sic@

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsDedicatedBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             f_EUTRA_Get_508_TrafficFlowTemplate_2_Modify(v_EpsDedicatedBearerId2))))); // @sic R5-103625 sic@

    // Check: Does UE transmits a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43?

    //@siclog "Step 43D" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,  // @sic R5s110593 sic@ @sic R5s141327 sic@

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsDedicatedBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 43D");

    //  The SS releases the RRC connection.

    //@siclog "Step 44" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);

    // EXCEPTION: Step 45 & 46 describe behaviour that depends on the UE capability.

    if ( pc_ESM_MO_Bearer_Allocation == true ) {  // @sic R5-121749 sic@

      //  Cause the UE to request bearer resource allocation of dedicated EPS bearer associated with additional PDN connectivity.

      //@siclog "Step 45" siclog@

      f_UT_RequestActivateDedicatedBearer (UT,

                                           hex2int(v_EpsDedicatedBearerId2),

                                           hex2int(v_EpsBearerId2),

                                           cs_508_EPS_QoS_Dedicated_1,

                                           f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2),

                                           LOCAL_CNF_REQUIRED); // @sic R5s110055 sic@

      // Check: Does the UE transmit SERVICE REQUEST?

      //@siclog "Step 46" siclog@

      t_Wait.start(3.0);

      alt {

        [] SRB.receive ( car_SRB1_NasPdu_IND ( eutra_CellB,

                                               cr_NAS_Indication ( ?,

                                                                   ?)))

          {

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 46");

            t_Wait.stop;

          }

        [] t_Wait.timeout

          {

          }

      }

    }

  }


Change 3.3.3 - fl_TC_10_6_1_Body
	  function fl_TC_10_6_1_Body() runs on EUTRA_PTC

  { // @sic R5s100461 sic@ @sic R5-104733 sic@

    var EPS_BearerIdentity v_EpsBearerId2 := '6'H; // default bearer on additional PDN

    var template (value) ESM_Cause v_EsmCause;

    var NAS_MSG_Indication_Type v_NasInd;

    var template (value) LinkedEpsBearerIdentity v_LinkedEpsBearerId := {idValue := hex2bit(v_EpsBearerId2)};

    var PDN_DISCONNECT_REQUEST v_PDNDiscReq;

    var ProcedureTransactionIdentifier v_EpsTi := tsc_PTI_Unassigned;

    var NasCount_Type v_NasCountUL;

    var SRB_COMMON_IND v_ReceivedAsp;

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_CellA); //Get FDD or TDD mode from Cell-configuration
    var template (value) RadioBearerList_Type v_DrbConfigList := { cs_OneDRB_ConfigAM(tsc_DRB1), cs_OneDRB_ConfigAM(tsc_DRB2) };
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    // Cause the UE to request disconnection from the additional PDN

    //@siclog "Step 1" siclog@

    f_UT_DeactivateBearer(UT, hex2int(v_EpsBearerId2));

    // The UE transmits a SERVICE REQUEST message.

    //@siclog "Step 1A" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    //@siclog "Step 1B" siclog@

    // The SS establishes SRB2 and DRBs associated with two default EPS bearer contexts

    // (a first PDN obtained during the attach procedure and an additional PDN).

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL, v_DrbConfigList);

    f_EUTRA_Security_Set(v_SecurityParams);

    v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1);

    v_DRB_ToAddModList[1] := cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB2);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( v_DRB_ToAddModList,

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                             f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA))),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Step 2. The UE transmits PDN DISCONNECT REQUEST

    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_CellA,

                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_PDNDisconnectReq(v_LinkedEpsBearerId)))) -> value v_ReceivedAsp;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNDiscReq  := v_NasInd.Pdu.Msg.pDN_DISCONNECT_REQUEST;

    v_EpsTi := v_PDNDiscReq.procedureTransactionIdentifier;

    //  The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST

    //@siclog "Step 3" siclog@

    v_EsmCause := cs_ESM_Cause_v ('00100100'B); // #36 regular deactivation

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release (tsc_RRC_TI_Def, tsc_DRB2),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(v_EpsBearerId2,

                                                                                    v_EpsTi,

                                                                                    v_EsmCause))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Step 4. Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_DeactivateEPSBearerCxtAccept(v_EpsBearerId2,

                                                                                      tsc_PTI_Unassigned))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    //@siclog "Step 5 Void" siclog@

    // The SS releases the RRC connection.

    //@siclog "Step 5A" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

  }


Change 3.3.4 - fl_TC_10_7_2_Body
	  function fl_TC_10_7_2_Body() runs on EUTRA_PTC

  {  // @sic R5-105012 sic@ @sic R5-106111 sic@

    var ProcedureTransactionIdentifier v_EpsTi := tsc_PTI_Unassigned;

    var SRB_COMMON_IND v_ReceivedAsp;

    var BEARER_RESOURCE_ALLOCATION_REQUEST v_BEARER_RESOURCE_ALLOCATION_Req;

    var template (value) NAS_DL_Message_Type v_ModifyEPSBearerCxtReq;

    var NAS_MSG_Indication_Type v_NasInd;

    var NasCount_Type v_NasCountUL;

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_CellA); //Get FDD or TDD mode from Cell-configuration 
    var template (value) RadioBearerList_Type v_DrbConfigList := { cs_OneDRB_ConfigAM(tsc_DRB1), cs_OneDRB_ConfigAM(tsc_DRB2) };
    //  The SS releases the RRC connection.

    //@siclog "Step 0" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);  // @sic R5-106111 sic@

    //  Cause the UE to request bearer resource allocation of dedicated EPS bearer associated with first PDN connectivity

    //@siclog "Step 1" siclog@

    f_UT_RequestActivateDedicatedBearer (UT,

                                         hex2int(tsc_EpsDedicatedBearerId2),

                                         hex2int(tsc_EpsDefaultBearerId),

                                         cs_508_EPS_QoS_Dedicated_1,

                                         f_EUTRA_Get_508_TrafficFlowTemplate(2, tsc_EpsDedicatedBearerId2));

    //  The UE transmits a SERVICE REQUEST message.

    //@siclog "Step 2" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_MoData,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    //  The SS establishes SRB2 and DRB associated with default EPS bearer context

    // (a first PDN obtained during the attach procedure).

    //@siclog "Step 2A" siclog@

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL, v_DrbConfigList);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1), cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB2)},   // @sic R5-106111 sic@

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                             f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA))),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Check: Does the UE transmit a BEARER RESOURCE ALLOCATION REQUEST message?

    //@siclog "Step 3" siclog@

    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_CellA,

                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_508_BearerResourceAllocReq(?,

                                                                                          tsc_EpsDefaultBearerId)))) -> value v_ReceivedAsp;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    // store PTI received

    v_BEARER_RESOURCE_ALLOCATION_Req  := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.bEARER_RESOURCE_ALLOCATION_REQUEST; // @sic R5s100538 sic@

    v_EpsTi := v_BEARER_RESOURCE_ALLOCATION_Req.procedureTransactionIdentifier;

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message.

    // This message is included in a DLInformationTransfer message.

    //@siclog "Step 4" siclog@

    v_ModifyEPSBearerCxtReq := cs_508_ModifyEPSBearerCxtReq(tsc_EpsDedicatedBearerId,

                                                            v_EpsTi,

                                                            cds_508_EPS_QoS_Dedicated_1_tlv,

                                                            f_EUTRA_Get_508_TrafficFlowTemplate_2_Add(tsc_EpsDedicatedBearerId));  // @sic R5s100549 sic@ @sic R5-106111 sic@

    //Send DLInformationTransfer as this only Modify EPS Bearer Request

    SRB.send(cas_SRB_NasPdu_REQ(eutra_CellA,

                                tsc_SRB2,

                                cs_TimingInfo_Now,

                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                               v_ModifyEPSBearerCxtReq)));

    //  Check:  Does the UE transmit a MODIFY EPS BEARER CONTEXT ACCEPT message as specified?

    //@siclog "Step 5" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtAccept(tsc_EpsDedicatedBearerId)))); // @sic R5s110008 sic@

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

  }


Change 3.3.5 - fl_TC_11_2_1_Body
	  function fl_TC_11_2_1_Body() runs on EUTRA_PTC

  {

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NasCount_Type v_NasCountUL;

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var NAS_MSG_Indication_Type v_NasInd;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_CellA);

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA);

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var EmergencyNumList v_EmergencyNumList := f_EUTRA_MobileInfo_GetLocalEmergencyNumberList (); // list of locally usable emergency numbers @sic R5-125276 sic@

    var charstring v_EmergencyNumber := hex2str(v_EmergencyNumList[0]); // 1st number in the list of locally usable emergency numbers

    var template (value) RadioBearerList_Type v_DrbConfigList := {cs_OneDRB_ConfigAM(tsc_DRB1), cs_OneDRB_ConfigAM(tsc_DRB2), cs_OneDRB_ConfigAM(tsc_DRB3)};  // @sic R5s140873 sic@

    timer t_Wait;

    v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1);

    v_DRB_ToAddModList[1] := cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB2);

    v_DRB_ToAddModList[2] := cs_508_DRB_ToAddMod_DEFAULT_UM(tsc_DRB3);

    //The user a PDN context for emergency bearer services by MMI or by AT command.

    //@siclog "Step 1 - 13" siclog@

    f_EUTRA_508_EstablishIMSEmergencyCall(eutra_CellA, v_EmergencyNumber, "Step 2");

    //IMS call has been released.

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN2]); // @sic R5s13???? sic@

    // The SS releases the RRC connection.

    //@siclog "Step 14" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA,-,-,-, v_DrbConfigList ); // @sic R5s140873 sic@

    // The user intiates an emergency call by MMI or by AT command.

    // @siclog "Step 15" siclog@

    f_UT_RequestAdditionalPDNConnectivity (UT, 7, -, true, CNF_REQUIRED); // @sic R5s141345 sic@

    // Check: Does the UE transmit SERVICE REQUEST?

    //@siclog "Step 16" siclog@

    t_Wait.start(3.0);

    alt {

      [] SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellA, cr_RRCConnectionRequest(?))) // @sic R5s130748 sic@

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 16");

          t_Wait.stop;

        }

      [] t_Wait.timeout

        {

        }

    }

    //The SS pages the UE in order to establish a dedicated EPS bearer context.

    //@siclog "Step 17" siclog@

    f_EUTRA_UE_Page(eutra_CellA,

                    cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParams.MME_Code,

                                                                       v_GutiParams.M_TMSI)));

    // The UE transmits a SERVICE REQUEST for downlink signalling.

    //@siclog "Step 18" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_mt_Access,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));

    //  The SS establishes SRB2 and DRB associated with default EPS bearer context

    // (a first PDN obtained during the attach procedure).

    //@siclog "Step 19" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL , v_DrbConfigList);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2(v_DRB_ToAddModList,

                                                                                                                 cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                 cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                            v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                            v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the dedicated EPS bearer to the additional PDN.

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    //@siclog "Step 20-21" siclog@

    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA, '6'H);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 21");
  }


Change 3.3.6 - fl_ReactivateEPS_BearerContext
	  function fl_ReactivateEPS_BearerContext(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  {

    var GutiParameters_Type v_GutiParameters;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_KsiValue v_KsiValue;

    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_CellId); //Get FDD or TDD mode from Cell-configuration 
    var template (value) RadioBearerList_Type v_DrbConfigList := { cs_OneDRB_ConfigAM(tsc_DRB1), cs_OneDRB_ConfigAM(tsc_DRB2) };
    /* Step 2 */

    v_GutiParameters := f_EUTRA_CellInfo_GetGuti(p_CellId);

    f_EUTRA_UE_Page(p_CellId,

                    cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParameters.MME_Code, v_GutiParameters.M_TMSI)));

    // Step 3 - 5 of Generic RB establishment procedure

    //Check the Service Request message

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                               v_RRC_TI,

                                               ?,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(v_KsiValue)));

    /* step 6-8 of procedure in 36.508 cl. 4.5.3.3 */

    v_AuthParams := f_EUTRA_Authentication_InitAS(v_AuthParams, v_NasInd.SecurityProtection.NasCount);

    v_AuthParams := f_EUTRA_RRC_ActivateSecurity(p_CellId, v_AuthParams, v_NasInd.SecurityProtection.NasCount , v_DrbConfigList);

    f_EUTRA_Security_Set(v_AuthParams);

    //The SS transmits an RRCConnectionReconfiguration message to configure data radio bearer(s) associated with the existing EPS bearer context    <-- RRC: RRCConnectionReconfiguration

    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2({cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1), cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB2)},

                                                                                                                 cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                 cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                            v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                            f_EUTRA_CellInfo_GetAntennaInfoDedicated(p_CellId))),

                                                                        omit)));

    //The UE transmits an RRCConnectionReconfigurationComplete message  --> RRC: RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(p_CellId,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

  }


Change 4 - DRB configuration in RRC CONNECTED state
Change 4.1 – New functions: DRB configuration and restart of security
	Function name
	f_EUTRA_SS_ConnectedMode_AddDrbs, f_EUTRA_SS_ConnectedMode_AddUmDrb, f_EUTRA_SS_ConnectedMode_AddAmDrb

	Reason for change
	Adding DRBs for already established RRC connection requires restart of RRC security

	Summary of changes
	New functions to do DRB configuration and restart of RRC security

	TTCN module
	EUTRA_NASSteps


	  function f_EUTRA_SS_ConnectedMode_AddDrbs(EUTRA_CellId_Type p_CellId ,

                                            template (value) RadioBearerList_Type p_DrbConfigList) runs on EUTRA_PTC

  {

    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, p_DrbConfigList);
    f_EUTRA_SS_RRC_RestartSecurity(p_CellId);

  }

	  function f_EUTRA_SS_ConnectedMode_AddUmDrb(EUTRA_CellId_Type p_CellId,

                                             DRB_Identity p_DrbId) runs on EUTRA_PTC

  { 
    var template (value) RadioBearerList_Type v_DrbConfigList := f_EUTRA_SS_GetDrbConfigList(0, 1, p_DrbId);

    f_EUTRA_SS_ConnectedMode_AddDrbs(p_CellId, v_DrbConfigList);
  }

	  function f_EUTRA_SS_ConnectedMode_AddAmDrb(EUTRA_CellId_Type p_CellId,

                                             DRB_Identity p_DrbId) runs on EUTRA_PTC

  { 
    var template (value) RadioBearerList_Type v_DrbConfigList := f_EUTRA_SS_GetDrbConfigList(1, 0, p_DrbId);

    f_EUTRA_SS_ConnectedMode_AddDrbs(p_CellId, v_DrbConfigList);
  }


Change 4.2 – Common functions adding DRBs
	Function name
	f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS_DrbReconfig, f_EUTRA_ActivateDedicatedEpsBearer, f_EUTRA_TxDedicatedEpsBearerCtxtReq

	Reason for change
	When DRBs are added at the UE dynamic DRB configuration requires configuring the DRBs at the SS side too.

	Summary of changes
	DRB configuration added to common functions

	TTCN module
	EUTRA_NASSteps


Change 4.2.1 - f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS_DrbReconfig
	  function f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS_DrbReconfig(EUTRA_CellId_Type p_CellId,

                                                                        RRC_TransactionIdentifier p_RRC_TI,

                                                                        DRB_IdentityList_Type p_DrbList,

                                                                        template (value) NAS_MSG_RequestList_Type p_NAS_MsgList,

                                                                        RLC_Mode_Type p_RLC_Mode_Type := AM_Mode)

    runs on EUTRA_PTC

  {

    var template (omit) DRX_Config v_DRX_Config;

    var template (omit) MAC_MainConfig_Type v_MAC_MainConfig := omit;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var integer i;

    // EPS bearer contexts are to be activated with default parameters.

    if (p_RLC_Mode_Type == AM_Mode) {  // @sic R5s110340 ch. 8 sic@

      for (i:=0; i<lengthof(p_DrbList); i:=i+1) {

        v_DRB_ToAddModList[i] := cs_508_DRB_ToAddMod_DEFAULT_AM(p_DrbList[i]); 
        f_EUTRA_SS_ConnectedMode_AddAmDrb(p_CellId, p_DrbList[i]);
      }

    }

    else {

      for (i:=0; i<lengthof(p_DrbList); i:=i+1) {

        v_DRB_ToAddModList[i] := cs_508_DRB_ToAddMod_DEFAULT_UM(p_DrbList[i]); 
        f_EUTRA_SS_ConnectedMode_AddUmDrb(p_CellId, p_DrbList[i]);
      }

      v_DRX_Config := f_GetUeDrxConfig(DRX_S);

      v_MAC_MainConfig := cs_MAC_MainConfig_Explicit(cs_MAC_MainConfig_Common(-, v_DRX_Config));     /* @sic R5s140618 change 2 sic@ */

      if (isvalue(v_DRX_Config)) {

        f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_Config(valueof(v_DRX_Config)));     /* configure DRX at the SS; @sic R5s140618 change 2 sic@ */

      }

    }

    SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,

                                        cs_TimingInfo_Now,

                                        cs_RRCConnectionReconfiguration_DRB_Reconfig(p_RRC_TI, v_DRB_ToAddModList, v_MAC_MainConfig),

                                        p_NAS_MsgList));

    //Receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND (p_CellId,

                                     cr_508_RRCConnectionReconfigurationComplete(p_RRC_TI)));

  }


( Aditional DRBs are configured by all functions using f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS_DrbReconfig:
	Function
	DRBs added

	f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14
	1 UM DRB

	
f_EUTRA_508_EstablishIMSEmergencyCall_Steps10_14
	

	

f_EUTRA_508_EstablishIMSEmergencyCall_Steps9_15
	

	


f_EUTRA_508_EstablishIMSEmergencyCall_Steps2_15
	

	



f_EUTRA_508_EstablishIMSEmergencyCall
	

	f_EUTRA_508_IMSEmergencyCallLimitedService_Steps17_19
	DRB2 UM

	
f_EUTRA_508_IMSEmergencyCallLimitedService5a1_19
	

	

f_EUTRA_508_IMSEmergencyCallLimitedService_Steps2_19
	

	


f_EUTRA_508_IMSEmergencyCallLimitedService
	

	f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig
	1 DRB (AM or UM)

	
f_EUTRA_ActivateAdditionalPDNOnDef
	1 AM DRB

	

f_EUTRA_IPCAN_XCAP_Signalling
	

	

f_EUTRA_508_ActivateAdditionalPDN
	

	
f_EUTRA_508_EstablishIMSEmergencyCall_Steps10_12
	1 AM DRB

	

f_EUTRA_508_EstablishIMSEmergencyCall_Steps10_14
	

	


f_EUTRA_508_EstablishIMSEmergencyCall_Steps9_15
	

	



f_EUTRA_508_EstablishIMSEmergencyCall_Steps2_15
	

	




f_EUTRA_508_EstablishIMSEmergencyCall
	

	




f_EUTRA_IPCAN_EmergencyCall
	

	



f_EUTRA_IPCAN_EmergencyCall
	

	f_EUTRA_TxActivateDedicatedBearerCtxtOnAdditionalPDN
	DRB2 AM, DRB3 AM

	fl_EUTRA_GenericIMS_MO_SpeechCall_11_2_6
	None: RRC connection is release at the end of the function


Change 4.2.2 - f_EUTRA_ActivateDedicatedEpsBearer
	  function f_EUTRA_ActivateDedicatedEpsBearer(EUTRA_CellId_Type p_CellId,


                                              template (value) DRB_ToAddModList p_DRB_ToAddModList,

                                              template (omit)  MAC_MainConfig_Type p_MAC_MainConfig,

                                              template (value) NAS_MSG_RequestList_Type p_NAS_MsgList)

    runs on EUTRA_PTC

  {

    var SRB_COMMON_IND v_ReceivedMsg;

    var IntegerList_Type v_EpsBearerListTX := {};

    var IntegerList_Type v_EpsBearerListRX := {};

    var template (present) IntegerList_Type v_EpsBearerListExpected; 
    var template (value) RadioBearerList_Type v_RadioBearerList;
    var integer n := lengthof(p_NAS_MsgList);

    var integer i;

    for (i := 0; i < n; i := i + 1) {

      v_RadioBearerList[i] := f_EUTRA_SS_GetDrbConfigUM(p_DRB_ToAddModList[i].drb_Identity);
    }

    f_EUTRA_SS_ConnectedMode_AddDrbs(p_CellId, v_RadioBearerList);
    f_EUTRA_TxDedicatedEpsBearerCtxtReq_Common(p_CellId, -, p_DRB_ToAddModList, p_MAC_MainConfig, p_NAS_MsgList);

    for (i:=0; i<n; i:=i+1) {

      v_EpsBearerListTX[i] := hex2int(valueof(p_NAS_MsgList[i].Pdu.Msg.aCTIVATE_DEDICATED_EPS_BEARER_CONTEXT_REQUEST.epsBearerId));

    }

    for (i:=0; i<n; i:=i+1) {

      SRB.receive(car_SRB2_NasPdu_IND(p_CellId, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_508_ActDedicatedEPSBearerCxtAccept(?)))) -> value v_ReceivedMsg;

      v_EpsBearerListRX[i] := hex2int(v_ReceivedMsg.Signalling.Nas[0].Pdu.Msg.aCTIVATE_DEDICATED_EPS_BEARER_CONTEXT_ACCEPT.epsBearerId);

    }

    v_EpsBearerListExpected := {permutation(all from v_EpsBearerListTX)};

    if (not match(v_EpsBearerListRX, v_EpsBearerListExpected)) {

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "EPS bearer ids are different in ACTIVATE_DEDICATED_EPS_BEARER_CONTEXT_REQUEST and ACTIVATE_DEDICATED_EPS_BEARER_CONTEXT_ACCEPT messages");

    }

  }


( Aditional DRBs are configured by all functions using f_EUTRA_ActivateDedicatedEpsBearer:

	Function
	DRBs added

	f_EUTRA_ActivateDedicatedEpsBearer_Speech
	1 UM DRB

	
f_EUTRA_GenericIMS_MT_SpeechCall
	DRB2 UM

	
f_EUTRA_IPCAN_MO_SpeechCall
	DRB2 UM

	
f_EUTRA_IPCAN_MT_SpeechCall
	1 UM DRB

	f_EUTRA_ActivateDedicatedEpsBearer_AddVideo
	1 UM DRB

	
f_EUTRA_IPCAN_AddVideo
	1 UM DRB

	f_EUTRA_ActivateDedicatedEpsBearer_SpeechVideo
	DRB2 UM, DRB3 UM

	
f_EUTRA_IPCAN_MO_VideoCall
	DRB2 UM, DRB3 UM

	
f_EUTRA_IPCAN_MT_VideoCall
	DRB2 UM, DRB3 UM


Change 4.2.3 - f_EUTRA_TxDedicatedEpsBearerCtxtReq
	  function f_EUTRA_TxDedicatedEpsBearerCtxtReq(EUTRA_CellId_Type p_CellId,

                                               DRB_Identity p_DRB_Id,

                                               EPS_BearerIdentity p_Id,

                                               template (value) ProcedureTransactionIdentifier p_ProcTi, // octet

                                               template (value) LinkedEpsBearerIdentity p_LinkedId, //  1/2 octet

                                               template (value) EPS_QualityOfService p_QoS,

                                               template (value) TrafficFlowTemplate p_Tft,

                                               RLC_Mode_Type p_RLC_Mode_Type := AM_Mode)

    runs on EUTRA_PTC

  { /* RRC connection reconfiguration acc. to 36.331 cl. 5.3.5 */

    /* @sic R5s141221 - Further changes by MCC10: use of f_EUTRA_TxDedicatedEpsBearerCtxtReq_Common sic@ */

    var template (value) NAS_DL_Message_Type v_ActDedicatedEPSBearerCxtReq := cs_508_ActDedicatedEPSBearerCxtReq(p_Id, p_ProcTi, p_LinkedId, p_QoS, p_Tft);

    var template (value) NAS_MSG_RequestList_Type v_NAS_MsgList := {cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_ActDedicatedEPSBearerCxtReq)};

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var template (omit) MAC_MainConfig_Type v_MAC_MainConfig := omit;

    if (p_RLC_Mode_Type == AM_Mode) {

      v_DRB_ToAddModList := {cs_508_DRB_ToAddMod_DEFAULT_AM(p_DRB_Id)}; 
      f_EUTRA_SS_ConnectedMode_AddAmDrb(p_CellId, p_DRB_Id);
    }

    else {

      v_DRB_ToAddModList := {cs_508_DRB_ToAddMod_DEFAULT_UM(p_DRB_Id)};

      v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS; 
      f_EUTRA_SS_ConnectedMode_AddUmDrb(p_CellId, p_DRB_Id);
    }

    f_EUTRA_TxDedicatedEpsBearerCtxtReq_Common(p_CellId, -, v_DRB_ToAddModList, v_MAC_MainConfig, v_NAS_MsgList);

  }


Change 5 - DRB configuration during handover
Change 5.1 – Intercell handover: Common cases
	Function name
	f_EUTRA_SS_AS_ActivateSecurity_HO_Reest, f_EUTRA_SS_AS_ActivateSecurity_HO_Common

	Reason for change
	In general during an intercell (or IRAT) handover the RRC security needs to be updated (f_EUTRA_SS_AS_ActivateSecurity_HO_Reest) what requires the DRBs to be configured at the target cell.

On the other hand f_EUTRA_SS_AS_ActivateSecurity_HO_Reest is used not only for handover to a new cell but also used e.g. for intracell handover or DRBs at the target cell may be configured already.

	Summary of changes
	· f_EUTRA_SS_AS_ActivateSecurity_HO_Reest: Automatic configuration of DRBs which are not configured at the SS yet 

· Split of f_EUTRA_SS_AS_ActivateSecurity_HO_Reest: new function f_EUTRA_SS_AS_ActivateSecurity_HO_Common

	TTCN module
	EUTRA_SecuritySteps


Change 5.1.1 - f_EUTRA_SS_AS_ActivateSecurity_HO_Common
	  function f_EUTRA_SS_AS_ActivateSecurity_HO_Common(EUTRA_CellId_Type p_TargetCellId,

                                                   EUTRA_SecurityParams_Type p_Auth_Params,

                                                   NextHopChainingCount p_NCC,

                                                   template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                                   DRB_IdentityList_Type p_AM_DrbList := tsc_DrbList_Def,

                                                   template (omit) SecurityActTimeList_Type p_SecurityActTimeList := omit) runs on EUTRA_PTC

    return EUTRA_SecurityParams_Type

  { /* @sic R5s110019/R5s110030: add. parameter p_SecurityActTimeList; this may be used to avoid async. query of PDCP count when security activation is scheduled sic@ */

    var PhysCellId v_TargetPhysicalCellIdentity := f_EUTRA_CellInfo_GetPhyCellId(p_TargetCellId);

    var ARFCN_ValueEUTRA v_TargetCellEARFCN_DL := f_EUTRA_CellInfo_GetEARFCN_DL(p_TargetCellId); // @sic R5s100138 sic@

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_InitAS_KeyChaining_KeyRefresh(v_TargetPhysicalCellIdentity,

                                                                                            v_TargetCellEARFCN_DL,

                                                                                            p_Auth_Params,

                                                                                            p_NCC); /* Calculate Kenb and KRRCenc, KRRCint, KUPenc */

    // @sic R5s100138 sic@

    // Read PDCP SQN for SRB1 and calcuate Ciphering Activation times

    if (isvalue(p_SecurityActTimeList)) {

      v_SecurityParams.AS_Ciphering.ActTimeList := valueof(p_SecurityActTimeList);

    } else {

      v_SecurityParams.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetForHO (p_TargetCellId, p_AM_DrbList);

    }

    // Note by default DRB is also pre-configured on SS side

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL (p_TargetCellId,

                                              v_SecurityParams.AS_Integrity, //@sic R5s110149 sic@

                                              v_SecurityParams.AS_Ciphering,

                                              p_TimingInfo);

    return v_SecurityParams;

  }


Further changes: f_EUTRA_SS_AS_ActivateSecurity_HO_Common replaces f_EUTRA_SS_AS_ActivateSecurity_HO_Reest in 
· f_EUTRA_508RRC_IntraLTE_HO_IntraCell_Step1_5
· f_EUTRA_508RRC_IntraLTE_CA_HO_InterCell
Change 5.1.2 - f_EUTRA_SS_AS_ActivateSecurity_HO_Reest

	  function fl_DrbInPdcpCountInfoList(DRB_Identity p_DrbId,

                                         PdcpCountInfoList_Type p_PdcpCountInfoList) return boolean

  {

    var RadioBearerId_Type v_RadioBearerId;

    var integer i;

    for (i := 0; i < lengthof(p_PdcpCountInfoList); i := i + 1) {

      v_RadioBearerId := p_PdcpCountInfoList[i].RadioBearerId;

      if (ischosen(v_RadioBearerId.Drb) and (v_RadioBearerId.Drb == p_DrbId)) {

        return true;

      }

    }

    return false;

  }

	  function fl_EUTRA_SS_AS_ConfigureMissingDRBsForHandover_AMorUM(EUTRA_CellId_Type p_TargetCellId,

                                                                 PdcpCountInfoList_Type p_PdcpCountInfoList,

                                                                 DRB_IdentityList_Type p_DrbList,

                                                                 RLC_Mode_Type p_RLC_Mode_Type) runs on EUTRA_PTC

  { /* local function to configure DRBs at the SS which are not included in p_PdcpCountInfoList */

    var DRB_Identity v_DrbId;

    var integer i;

    for (i := 0; i < lengthof(p_DrbList); i := i + 1) {

      v_DrbId := p_DrbList[i];

      if (not fl_DrbInPdcpCountInfoList(v_DrbId, p_PdcpCountInfoList)) {

        select (p_RLC_Mode_Type) {

          case (AM_Mode) {

            f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(p_TargetCellId, cs_OneDRB_ConfigAM(v_DrbId));

          }

          case (UM_Mode) {

            f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(p_TargetCellId, cs_OneDRB_ConfigUM(v_DrbId));

          }

        }

      }

    }

  }

	  function fl_EUTRA_SS_AS_ConfigureMissingDRBsForHandover(EUTRA_CellId_Type p_TargetCellId,

                                                          DRB_IdentityList_Type p_AM_DrbList,

                                                          DRB_IdentityList_Type p_UM_DrbList) runs on EUTRA_PTC

  {

    var PdcpCountInfoList_Type v_PdcpCountInfoList := f_EUTRA_SS_PdcpCount_GetAll(p_TargetCellId);

    fl_EUTRA_SS_AS_ConfigureMissingDRBsForHandover_AMorUM(p_TargetCellId, v_PdcpCountInfoList, p_AM_DrbList, AM_Mode);

    fl_EUTRA_SS_AS_ConfigureMissingDRBsForHandover_AMorUM(p_TargetCellId, v_PdcpCountInfoList, p_UM_DrbList, UM_Mode);

  }

	  function f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(EUTRA_CellId_Type p_TargetCellId,

                                                   EUTRA_SecurityParams_Type p_Auth_Params,

                                                   NextHopChainingCount p_NCC,

                                                   DRB_IdentityList_Type p_AM_DrbList := tsc_DrbList_Def,

                                                   DRB_IdentityList_Type p_UM_DrbList := {}) 

    runs on EUTRA_PTC return EUTRA_SecurityParams_Type

  {
    fl_EUTRA_SS_AS_ConfigureMissingDRBsForHandover(p_TargetCellId, p_AM_DrbList, p_UM_DrbList);
    return f_EUTRA_SS_AS_ActivateSecurity_HO_Common(p_TargetCellId, p_Auth_Params, p_NCC, -, p_AM_DrbList);

  }


Change 5.2 – Intercell handover: generic procedure acc. 36.523-3 clause 7.16.1
	Function name
	f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common

	Reason for change
	· For dynamic configuration of DRBs at the SS DRBs at the target cell need to configured just before the handover of the PDCP count

· In the source cell the DRBs shall not be reconfigured

	Summary of changes
	· DRBs are configured at the beginning of the procedure for the target cell

· only SRBs are re-configured at the end of the procedure in the source cell

NOTE: 36.523-3 cl. 7.16.1 need to be changed acc.

	TTCN module
	EUTRA_RRCSteps


	  function f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common(EUTRA_CellId_Type                         p_SourceCellId,

                                                       EUTRA_CellId_Type                         p_TargetCellId,

                                                       template (value) DL_DCCH_Message          p_RRC_ConnRecongMsg,

                                                       template (value) RachProcedureConfig_Type p_RachProcedureConfig,

                                                       template (value) RadioBearerList_Type     p_DrbConfigList := cs_DrbConfigList_Def,

                                                       C_RNTI                                    p_C_RNTI        := tsc_C_RNTI_Def2,

                                                       template (omit)  DciUlInfo_Type           p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                       template (value) UL_GrantConfig_Type      p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                       template (value) UL_GrantConfig_Type      p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                       boolean                                   p_CnfFlag := tsc_NoCnfReq,

                                                       template (present) UL_DCCH_Message        p_RRCConnReconfigComplete := cr_508_RRCConnectionReconfigurationComplete)

    runs on EUTRA_PTC

  { /* @sic R5s120709 function introduced to cope with "Full Handover" as needed for TC_13_4_1_5 sic@ */

    /* @sic R5s140147: additional parameter p_RRCConnReconfigComplete sic@ */

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var template (omit) MeasConfig v_MeasConfig := p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.measConfig;

    var boolean v_MeasGapConfigIsPresent := isvalue(v_MeasConfig) and ispresent(v_MeasConfig.measGapConfig);

    var DrxCtrl_Type v_DrxCtrl;

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    // Step 0: Configuration of DRBs at the target cell

    f_EUTRA_SS_CommonRadioBearerConfig(p_TargetCellId, p_DrbConfigList);
    // Step 1: Transfer of the PDCP Count for AM DRBs from source to target cell   @sic R5-120721 change 7 sic@

    f_EUTRA_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_DRB_IdentityList.AMonly);

    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 3: Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetCellId, p_C_RNTI, p_RachProcedureConfig);

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetCellId, p_C_RNTI);

    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(p_TargetCellId,

                                                             v_Auth_Params,

                                                             v_NextHopChainingCount,

                                                             -,

                                                             v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set(v_Auth_Params);

    // Step 5: configure DRX and MeasGap at the target cell (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present

    }

    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@

    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_REQ(p_SourceCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    f_IP_ChangeEutraCell(IP, p_TargetCellId);      /* @sic R5s150031 Change 1 sic@ */

    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE

    SRB.send(cas_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, p_RRC_ConnRecongMsg));

    // Step 9:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, p_RRCConnReconfigComplete));       // @sic R5s100430  Change 2, R5s140147 sic@

    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 11: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetCellId, -, -, -, -, -, p_CnfFlag);     /* @sic acc. to R5s100029 change 3 sic@ */

    // Step 13: Re-establish SRBs and DRB(s) in source cell

    f_EUTRA_SS_SRBs_DRBs_Release(p_SourceCellId, -, v_DRB_IdentityList.All, p_CnfFlag); // @sic R5s100801 sic@

    f_EUTRA_SS_SRBs_DRBs_Config(p_SourceCellId, -, omit, p_CnfFlag); // @sic R5s100801 sic@

    // Step 14: Release DRX and MeasGap configuration

    f_EUTRA_SS_DrxCtrlConfig(p_SourceCellId, cs_DrxCtrl_None, -, p_CnfFlag);  // @sic R5s100801 sic@

    f_EUTRA_SS_MeasGapCtrlConfig(p_SourceCellId, cs_MeasGapCtrl_None, -, p_CnfFlag);  // @sic R5s110743 sic@

  }


Change 5.3 – CA intercell handover: generic procedure acc. 36.523-3 clause 7.16.1a
	Function name
	f_EUTRA_508RRC_IntraLTE_CA_HO_InterCell

	Reason for change
	· For dynamic configuration of DRBs at the SS DRBs at the target cell need to configured just before the handover of the PDCP count

· In the source cell the DRBs shall not be reconfigured

	Summary of changes
	· DRBs are configured at the beginning of the procedure for the target cell

· only SRBs are re-configured at the end of the procedure in the source cell

NOTE: 36.523-3 cl. 7.16.1a need to be changed acc.

	TTCN module
	EUTRA_CommonFunctions_CA


	  function f_EUTRA_508RRC_IntraLTE_CA_HO_InterCell(EUTRA_CellId_Type                     p_SourcePCellId, //Source Pcell

                                                   EUTRA_CellId_Type                     p_SourceSCellId, //Source Scell

                                                   EUTRA_CellId_Type                     p_TargetPCellId, //Target Pcell

                                                   EUTRA_CellId_Type                     p_TargetSCellId, // Target Scell

                                                   template (omit) SCellToAddModList_r10  p_SCellToAddModList := omit,

                                                   template (omit) SCellToReleaseList_r10 p_SCellToReleaseList := omit,

                                                   boolean p_UL_CA := false, // UL CA desired

                                                   template (value) Scell_Capability_Type    p_Scell_Capability := UL_DL,

                                                   template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_DRB1,

                                                   C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,

                                                   template (omit)  MeasConfig           p_MeasConfig    := omit,

                                                   template (omit)  DciUlInfo_Type       p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                   template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                   template (value) UL_GrantConfig_Type  p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                   Dsr_TransMax_Type                     p_Dsr_TransMax := n4,

                                                   template (omit) CarrierFreqEUTRA      p_CarrierFreqEutra := omit,

                                                   boolean                               p_CnfFlag := tsc_NoCnfReq,

                                                   boolean                               p_AddAdditionalSpectrumEmission := false)  runs on EUTRA_PTC

  {

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;

    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var template (omit) MeasConfig v_MeasConfig := p_MeasConfig;

    var boolean v_MeasGapConfigIsPresent := isvalue(v_MeasConfig) and ispresent(v_MeasConfig.measGapConfig);

    var DrxCtrl_Type v_DrxCtrl;

    var SubFrameTiming_Type v_Timing, v_Timing1;

    var SCellIndex_r10  v_SCellIndex;

    var MAC_MainConfig_ScellDeactivationTimer_Type v_DeactTimer := rf32;

    var template (value) Scell_Capability_Type v_Scell_Capability := DlOnly;

    var boolean v_UL_CA := p_UL_CA;

    var boolean v_NoUL_SCell := f_EUTRA_CellInfo_GetNoUL_Frequency(p_TargetSCellId) ; //@sic R5s140083 sic@

    if (v_UL_CA) {

      v_Scell_Capability := UL_DL;

    }

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    v_RachProcedureConfig := cs_RachProcedureConfig_CRNTI_HO(f_EUTRA_CellInfo_GetRAR_TA(p_TargetPCellId),

                                                             f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetPCellId));

    v_RRCConnectionReconfiguration := f_EUTRA_RRCConnectionReconfiguration_IntraLTE_HO_InterCell(p_SourcePCellId,

                                                                                                 p_TargetPCellId,

                                                                                                 p_C_RNTI,

                                                                                                 p_MeasConfig,

                                                                                                 p_Dsr_TransMax,

                                                                                                 p_CarrierFreqEutra,

                                                                                                 p_AddAdditionalSpectrumEmission,

                                                                                                 cs_RRCConnectionReconfiguration_SCell_AddMod_IEs(p_SCellToAddModList, p_SCellToReleaseList));

    v_Timing := f_EUTRA_GetNextSendOccasion(p_SourcePCellId, 700); // 500 ms // Time for schedule of DL RRC message

    //@sic R5s130964 sic@

    v_Timing1 := f_EUTRA_TimingInfoAdd (v_Timing, 40); // time for Target Pcell SS configuration

    // The 40 ms is sufficient for 4 retransmissions and less than 15 [RRC processing delay] + 100 [5 retransmissions of PRACH preamble on trarget pCell]

    // Prepare target Pcell

    // Step 1: stop periodic TA on source Pcell

    f_EUTRA_SS_CommonCellConfig(p_SourcePCellId, cas_ULGrantAllocation_REQ(p_SourcePCellId,cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 2: Configure no RACH response in target Pcell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetPCellId, p_C_RNTI, cs_RachProcedureConfig_NoResponse(f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetPCellId)), cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    // Step 3: Transmit RRCConnectionReconfiguration/HO command to UE

    SRB.send(cas_SRB1_RrcPdu_REQ(p_SourcePCellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),v_RRCConnectionReconfiguration));

    // Step 4: optional if target Pcell is Source Scell

    if (p_SourceSCellId == p_TargetPCellId) {  //Else, the target Pcell is expected to be configured with SRB1-3 and DRB already

      f_EUTRA_CellConfig_ScellToPcell_Def(p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    }

    // Step 4a: Configuration of DRBs at the target cell

    f_EUTRA_SS_CommonRadioBearerConfig(p_TargetPCellId, p_DrbConfigList);       /* @sic R5s150338 - dynamic DRB configuration sic@ */
    // Step 5: Transfer of the PDCP Count for AM DRBs from source to target cell

    f_EUTRA_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_DRB_IdentityList.AMonly, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    // Step 6: Tell the SS about the HO

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_PdcpHandoverInit_REQ(p_SourcePCellId, p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number)));

    // Step 7: Configure C-RNTI based contention resolution in target Pcell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetPCellId, p_C_RNTI, v_RachProcedureConfig, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetPCellId, p_C_RNTI);

    // Step 8: activate RRC security at SS target Pcell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(p_TargetPCellId,

                                                             v_Auth_Params,

                                                             v_NextHopChainingCount,

                                                             cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number),

                                                             v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set(v_Auth_Params);

    // Step 9: configure DRX and MeasGap at the target P and S cells (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourcePCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourcePCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetPCellId, v_DrxCtrl, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

      f_EUTRA_SS_DrxCtrlConfig(p_TargetSCellId, v_DrxCtrl, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetPCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)),cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetSCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)),cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    }

    // Step 10: configure UL grant configuration on target Pcell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_ULGrantAllocation_REQ(p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    // Step 11: Configure target PCell as PCell

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_PCellConfig_REQ(p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), {p_TargetSCellId}, p_CnfFlag));

    // Step 12: Configure target Scell

    if(isvalue (p_SCellToAddModList)) {

      v_SCellIndex := valueof(p_SCellToAddModList[0].sCellIndex_r10);

    } else {

      v_SCellIndex := 1;

    }

    //if Scell change does not happen, steps 1-2 are still needed due to possible change in C-RNTI

    // Save new C-RNTI in cell configuration for p_SCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetSCellId, p_C_RNTI);

    if ( not v_NoUL_SCell) //@sic R5s140083 sic@

    {

    // Step 12.1: Configure C-RNTI based contention resolution in S-cell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetSCellId, p_C_RNTI, v_RachProcedureConfig, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), p_CnfFlag);

    // Step 12.2: configure def UL grant configuration and no auto TA transmision on target Scell

    f_EUTRA_SS_CommonCellConfig(p_TargetSCellId, cas_ULGrantAllocation_REQ(p_TargetSCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), cs_DciInfo_CcchDcchDtchUL_Def, cs_PUCCH_Synch_None, cs_UL_GrantConfig_Def, p_CnfFlag));

    };

    // Step 12.3: Configure SCell as Scell with new Pcell association

    f_EUTRA_SS_CommonCellConfig(p_TargetSCellId, cas_SCellConfig_REQ(p_TargetSCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), cs_SCellConfig(p_TargetPCellId, v_SCellIndex, p_Scell_Capability, v_DeactTimer), p_CnfFlag));

    // Step 13:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetPCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Step 14: configure periodic TA on target cell

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_PUCCH_Synch_Config_REQ(p_TargetPCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 15: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_PdcpHandoverComplete_REQ(p_TargetPCellId));

    // Step 16: configure RACH procedure for target cell back to default

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetPCellId, -, -, -, -, -, p_CnfFlag);

    if (p_SourcePCellId != p_TargetSCellId) {  //Source cell is no more a Carrier component

      // Step 17: Re-establish SRBs and DRB(s) in source cell

      f_EUTRA_SS_SRBs_DRBs_Release(p_SourcePCellId, -, v_DRB_IdentityList.All, p_CnfFlag);

      f_EUTRA_SS_SRBs_DRBs_Config(p_SourcePCellId, -, omit, p_CnfFlag);

      // Step 18: Release DRX and MeasGap configuration

      f_EUTRA_SS_DrxCtrlConfig(p_SourcePCellId, cs_DrxCtrl_None, -, p_CnfFlag);

      f_EUTRA_SS_MeasGapCtrlConfig(p_SourcePCellId, cs_MeasGapCtrl_None, -, p_CnfFlag);

      //Step 19: Configure source Pcell as normal cell

      f_EUTRA_SS_CommonCellConfig(p_SourcePCellId, cas_ServingCellRelease_REQ(p_SourcePCellId, cs_TimingInfo_Now, p_CnfFlag));

    }

    if (p_SourceSCellId != p_TargetSCellId and p_SourceSCellId != p_TargetPCellId) {

      // No RACH configuration as Already configured for C-RNTI based contention along with Pcell

      //No update of C-RNTI

      //Step 20

      //  Release DRX and MeasGap at the S cell (if necessary)

      if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceSCellId)) {

        f_EUTRA_SS_DrxCtrlConfig(p_SourceSCellId, cs_DrxCtrl_None, -, p_CnfFlag);

      }

      f_EUTRA_SS_MeasGapCtrlConfig(p_SourceSCellId, cs_MeasGapCtrl_None, -, p_CnfFlag);

      //Step 21

      //  Configure SCell as normal cell

      f_EUTRA_SS_CommonCellConfig(p_SourceSCellId, cas_ServingCellRelease_REQ(p_SourceSCellId, cs_TimingInfo_Now, p_CnfFlag));

      //  configure Default grant configuration on S cell

      f_EUTRA_SS_CommonCellConfig(p_SourceSCellId, cas_ULGrantAllocation_REQ(p_SourceSCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, -, p_CnfFlag));

    }

  }


Change 5.4 – Intercell handover: DRB configuration at target cell before handover of PDCP count
	Function name
	f_TC_13_3_1_2_EUTRA, f_TC_13_3_1_3_EUTRA

	Reason for change
	As for f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common DRBs at target cell need to be configured before calling f_EUTRA_SS_PdcpCount_Handover

	Summary of changes
	DRBs at target cell need to be configured before calling f_EUTRA_SS_PdcpCount_Handover

	TTCN module
	MultiLayer_Procedures


Change 5.4.1 - f_TC_13_3_1_2_EUTRA
	  function f_TC_13_3_1_2_EUTRA() runs on EUTRA_PTC

  { /* Intra-system connection re-establishment / Re-establishment of a new connection when further data is to be transferred */

    var template (value) CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var ShortMAC_I v_ShortMAC_I;

    var PhysCellId v_PhysicalCellIdentity_Cell1;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var NextHopChainingCount v_NextHopChainingCount := 0; // Default configuration value according to TS 36.508 Table 4.6.1-10

    var SR_PUCCH_ResourceIndex_Type v_Sr_PUCCH_ResourceIndex; //@sic R5100744 sic@

    var SR_ConfigIndex_Type v_SR_ConfigIndex; //@sic R5100744 sic@

    var SubFrameTiming_Type v_TimingInfo;

    var PDCP_SDUList_Type v_PDCP_SDUList_DRB1 := { f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE()) };      /* @sic R5s120470 sic@ */

    //Init cell array and security

    f_EUTRA_Init(c1);

    //Get cell specific parameters for Cell 1

    v_PhysicalCellIdentity_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);

    //Get FDD or TDD mode from Cell-configuration

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell2);

    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated(eutra_Cell2);

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    //The UE is in state Loopback Activated (state 4) on Cell 1 according to [18] using the UE TEST LOOP MODE B.

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeB);

    fl_EUTRA_GenericRbEst_DSR_TransMax(eutra_Cell1);     //@sic R5s100744 sic@

    f_EUTRA_CloseUE_TestLoopModeB(eutra_Cell1, '05'O);   //@sic R5s110430 sic@

    //Set initial cell power levels

    v_CellPowerList_AtT0 := {

      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_NonSuitableCellRS_EPRE)

    };

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);

    v_ShortMAC_I := f_Calculate_ShortMAC(eutra_Cell1, eutra_Cell2);

    f_EUTRA_TestBody_Set(true, layer2);

    //@siclog "Step 0" siclog@

    // Void //@sic R5-113190 sic@

    //@siclog "Step 1" siclog@

    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context on Cell 1. */

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1, tsc_RbId_DRB1, cs_TimingInfo_Now, v_PDCP_SDUList_DRB1));     /* @sic R5s120470 sic@ */

    //PDCP count shall be kept in PDCP re-establishment for AM DRB 
    f_EUTRA_SS_CommonRadioBearerConfig(eutra_Cell2, cs_DrbConfigList_DRB1);
    f_EUTRA_SS_PdcpCount_Handover ( eutra_Cell1, eutra_Cell2, { tsc_DRB1 } ); //@sic R5100744, R5-120721 change 7: f_EUTRA_SS_PdcpCount_Handover is done by f_EUTRA_508RRC_IntraLTE_HO_InterCell sic@

    //@siclog "Step 2" siclog@

    /* 100 ms after step 1 the SS changes the power level of cell 1 to non-suitable "Off" and changes the power level of cell 2 to suitable

       according to TS 36.508 subclause 6.2.2.1 in order that the radio link quality of Cell 1 is degraded. Note: 100 ms delay after step 1

       is to ensure that UE has time to transmit STATUS PDU for data received in step 1 */

    v_TimingInfo := f_EUTRA_GetNextSendOccasion(eutra_Cell2, 100); //@sic R5-112703, R5s110499 sic@

    v_CellPowerList_AtT1 := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE) //@sic R5s110499 sic@

    };

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), eutra_Cell2); //@sic R5-112703, R5s110318 Change 3.1 sic@

    //@siclog "Step 3" siclog@

    /* Check: Does the UE initiate an RRC connection re-establisment procedure and transmit an RRCConnectionReestablishmentRequest message on Cell 2?  */

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell2,

                                    cr_508_RRCConnectionReestablishmentRequest(tsc_C_RNTI_Def,

                                                                               v_PhysicalCellIdentity_Cell1,

                                                                               cr_ReestablishmentCause_OtherFailure,

                                                                               v_ShortMAC_I)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.3.1.2 Step 3");

    // Activate RRC security at SS target cell (Cell2) for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(eutra_Cell2, f_EUTRA_Security_Get(), v_NextHopChainingCount); //@sic R5100744 sic@

    f_EUTRA_Security_Set(v_Auth_Params); //@sic R5100744 sic@

    //@siclog "Step 4" siclog@

    /* The SS transmits RRCConnectionReestablishment message. */

    f_EUTRA_RRC_RRCConnectionReestablishment_Def(eutra_Cell2, -, v_NextHopChainingCount);

    //@siclog "Step 5" siclog@

    /* UE transmits an RRCConnectionReestablishmentComplete message */

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2,  cr_508_RRCConnectionReestablishmentComplete(tsc_RRC_TI_Def)));

    v_SR_ConfigIndex := f_EUTRA_CellInfo_GetSR_ConfigIndex (eutra_Cell2);

    v_Sr_PUCCH_ResourceIndex := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex(eutra_Cell2);

    //@siclog "Step 6" siclog@

    /* The SS transmits RRCConnectionReconfiguration message */

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell2, //@sic R5100744 sic@

                                 cs_TimingInfo_Now,

                                 cs_RRCConnReconfig_Resume_SRB2_DRBs(tsc_RRC_TI_Def,

                                                                     v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                     v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                     v_AntennaInfo,

                                                                     v_Sr_PUCCH_ResourceIndex,

                                                                     v_SR_ConfigIndex)));

    //EXCEPTION : the steps 7 and 8 can happen in any order

    //@siclog "Step 7 - 8" siclog@

    /* UE transmits an RRCConnectionConfigurationComplete message */

    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context on Cell 2? */

    interleave { //@sic R5100744 sic@

      [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def))) {}

      [] DRB.receive(car_DRB_COMMON_IND_PDCP_SDUList(eutra_Cell2, tsc_RbId_DRB1, ?, v_PDCP_SDUList_DRB1))  {} /* @sic R5s120470 sic@ */

    }

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.3.1.2 Step 8");

    f_EUTRA_TestBody_Set(false);

    //Postamble

    f_EUTRA_Postamble(eutra_Cell2, E3_TESTMODE); //@sic R5100744 sic@

  }


Change 5.4.2 - f_TC_13_3_1_3_EUTRA
	  function f_TC_13_3_1_3_EUTRA() runs on EUTRA_PTC

  { /* RRC connection reconfiguration / Full configuration / DRB establishment */

    var template (value) CellPowerList_Type v_CellPowerList_AtT0 := {

      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_NonSuitableCellRS_EPRE)

    };

    var template (value) CellPowerList_Type v_CellPowerList_AtT1 := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE) //@sic R5s110499 sic@

    };

    var ShortMAC_I v_ShortMAC_I;

    var PhysCellId v_PhysicalCellIdentity_Cell1;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var NextHopChainingCount v_NextHopChainingCount := 0; // Default configuration value according to TS 36.508 Table 4.6.1-10

    var SubFrameTiming_Type v_TimingInfo;

    var PDCP_SDUList_Type v_PDCP_SDUList_DRB1 := { f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE()) };      /* @sic R5s120470 sic@ */

    //var template (value) DL_DCCH_Message v_RRCConnReconfig;

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated;

    //Init cell array and security

    f_EUTRA_Init(c1);

    //Get cell specific parameters for Cell 1

    v_PhysicalCellIdentity_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);

    //Get FDD or TDD mode from Cell-configuration

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell2);

    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated(eutra_Cell2);

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    //The UE is in state Loopback Activated (state 4) on Cell 1 according to [18] using the UE TEST LOOP MODE B.

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeB);

    f_EUTRA_RbEst_Def(eutra_Cell1);

    f_EUTRA_CloseUE_TestLoopModeB(eutra_Cell1, '05'O);

    //Set initial cell power levels

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);

    v_ShortMAC_I := f_Calculate_ShortMAC(eutra_Cell1, eutra_Cell2);

    f_EUTRA_TestBody_Set(true, layer2);

    //@siclog "Step 1" siclog@

    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context on Cell 1. */

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1, tsc_RbId_DRB1, cs_TimingInfo_Now, v_PDCP_SDUList_DRB1));     /* @sic R5s120470 sic@ */

    //PDCP count shall be kept in PDCP re-establishment for AM DRB 
    f_EUTRA_SS_CommonRadioBearerConfig(eutra_Cell2, cs_DrbConfigList_DRB1);
    f_EUTRA_SS_PdcpCount_Handover(eutra_Cell1, eutra_Cell2, { tsc_DRB1 });

    //@siclog "Step 2" siclog@

    /* 100 ms after step 1 the SS changes the power level of cell 1 to non-suitable "Off" and changes the power level of cell 2 to suitable

       according to TS 36.508 subclause 6.2.2.1 in order that the radio link quality of Cell 1 is degraded. Note: 100 ms delay after step 1

       is to ensure that UE has time to transmit STATUS PDU for data received in step 1 */

    v_TimingInfo := f_EUTRA_GetNextSendOccasion(eutra_Cell2, 100); //

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), eutra_Cell2);

    //@siclog "Step 3" siclog@

    /* Check: Does the UE initiate an RRC connection re-establisment procedure and transmit an RRCConnectionReestablishmentRequest message on Cell 2?  */

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell2,

                                    cr_508_RRCConnectionReestablishmentRequest(tsc_C_RNTI_Def,

                                                                               v_PhysicalCellIdentity_Cell1,

                                                                               cr_ReestablishmentCause_OtherFailure,

                                                                               v_ShortMAC_I)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.3.1.3 Step 3");

    // Activate RRC security at SS target cell (Cell2) for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(eutra_Cell2, f_EUTRA_Security_Get(), v_NextHopChainingCount); //@sic R5100744 sic@

    f_EUTRA_Security_Set(v_Auth_Params); //@sic R5100744 sic@

    //@siclog "Step 4" siclog@

    /* The SS transmits RRCConnectionReestablishment message. */

    f_EUTRA_RRC_RRCConnectionReestablishment_Def(eutra_Cell2, -, v_NextHopChainingCount);

    //@siclog "Step 5" siclog@

    /* UE transmits an RRCConnectionReestablishmentComplete message */

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_508_RRCConnectionReestablishmentComplete(tsc_RRC_TI_Def)));

    // Configure SS for dedicated RACH

    //f_EUTRA_SS_ConfigRachProcedure(eutra_Cell2, tsc_C_RNTI_Def2, cs_RachProcedureConfig_CRNTI_Dedicated(f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell2),f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell2)));

    v_RadioResourceConfigDedicated :=  cs_RadioResourceConfigDedicated_SrbDrbConfig( {cs_508_SRB2_ToAddMod_DEFAULT},

                                                                                      {cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1)},

                                                                                      cs_508_MAC_MainConfig_Explicit_RBC_FullConfig, //@sic R5-131840 sic@

                                                                                      cs_508_PhysicalConfigDedicated_Default_RBC_HO(-, //@sic R5s130426 sic@

                                                                                                                                  v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                 v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                 v_AntennaInfo,

                                                                                                                                 cs_508_SchedulingRequest_Config_Default (f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex(eutra_Cell2), f_EUTRA_CellInfo_GetSR_ConfigIndex (eutra_Cell2), n4)));

    //@siclog "Step 6" siclog@

    /* The SS transmits RRCConnectionReconfiguration message */

    SRB.send ( cas_SRB1_RrcPdu_REQ ( eutra_Cell2, //@sic R5120826 sic@

                                     cs_TimingInfo_Now,

                                     cs_RRCConnectionReconfiguration_Common ( tsc_RRC_TI_Def,

                                                                           omit,

                                                                           omit,

                                                                           v_RadioResourceConfigDedicated,

                                                                           omit,

                                                                           cs_RRCConnectionReconfiguration_v890_IEs_FullConfig

                                                                     ) ) );

    //EXCEPTION : the steps 7 and 8 can happen in any order

    //@siclog "Step 7 - 8" siclog@

    /* UE transmits an RRCConnectionConfigurationComplete message */

    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context on Cell 2? */

    interleave {

      [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def))) {}

      [] DRB.receive(car_DRB_COMMON_IND_PDCP_SDUList(eutra_Cell2, tsc_RbId_DRB1, ?, v_PDCP_SDUList_DRB1)) {}  /* @sic R5s120470 sic@ */

    }

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.3.1.3 Step 8");

    f_EUTRA_TestBody_Set(false);

    //Postamble

    f_EUTRA_Postamble(eutra_Cell2, E3_TESTMODE);

  }


Change 5.5 – IRAT handover: new common function
	Function name
	f_EUTRA_SS_AS_ActivateSecurity_HO_IRAT

	Reason for change
	When the test case comes back to EUTRA at several places in TTCN the same code sequence is used; furthermore in these cases DRBs need to be established in the target 

	Summary of changes
	New common function

	TTCN module
	EUTRA_SecuritySteps


	  function f_EUTRA_SS_AS_ActivateSecurity_HO_IRAT(EUTRA_CellId_Type p_TargetCellId,

                                                  DRB_IdentityList_Type p_AM_DrbList := tsc_DrbList_Def,

                                                  DRB_IdentityList_Type p_UM_DrbList := {}) runs on EUTRA_PTC

  { // activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    fl_EUTRA_SS_AS_ConfigureMissingDRBsForHandover(p_TargetCellId, p_AM_DrbList, p_UM_DrbList);
    v_Auth_Params := f_EUTRA_Security_Get();

    v_Auth_Params := f_EUTRA_MappedContext_HO_Init(v_Auth_Params, v_Auth_Params.AuthParams);  //@sic R5s130010 sic@

    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetForHO(p_TargetCellId, tsc_DrbList_Def);  //@sic R5s130010, R5s130142 sic@

    f_EUTRA_SS_NAS_ConfigureSecurity(v_Auth_Params.NAS_Integrity, v_Auth_Params.NAS_Ciphering); // @sic R5s120686 sic@

    f_EUTRA_SS_NAS_CountSet('00000000'O, omit); // @sic R5s120686 sic@

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL(p_TargetCellId,

                                             v_Auth_Params.AS_Integrity,

                                             v_Auth_Params.AS_Ciphering,

                                             cs_TimingInfo_Now);

    f_EUTRA_Security_Set(v_Auth_Params);

  }


Further changes: function applied to 

· f_TC_8_4_2_7_Common
· f_TC_8_6_5_1_EUTRA
· f_TC_8_6_3_1_EUTRA
· f_TC_8_4_2_2_EUTRA
· f_TC_8_4_2_4_EUTRA
Change 6 – RRC Connection release
	Function name
	f_EUTRA_RRC_ConnectionRelease_Local

	Reason for change
	For dynamic configuration of DRBs at the SS no DRBs shall be re-configured when the RRC Connection gets released

	Summary of changes
	· Only SRBs are re-configured at the end of the procedure; 
NOTE 1: to achieve this a further change is required for f_EUTRA_SS_SRBs_DRBs_Config
NOTE: 36.523-3 cl. 7.18 needs to be changed acc.
· All DRBs being configured SS shall be released; there is no need for any parameter

	TTCN module
	EUTRA_RRCSteps


	  function f_EUTRA_RRC_ConnectionRelease_Local(EUTRA_CellId_Type p_CellId,

                                               SubFrameTiming_Type p_TimingAtT
) runs on EUTRA_PTC

  { /* NOTE: the function schedules all steps to release and reconfigure the SRBs and (default) DRB

     *       => this will in fact happen in the future */

    /* @sic R5-110803 change 13 sic@ */

    /* @sic R5s110634 MCC160 implementation of change 1.1: new parameter p_DrbConfigList sic@ */

    /* @sic R5s110743: release MeasGapConfig sic@ */

    var SubFrameTiming_Type v_Timing := p_TimingAtT;
 
    var DRB_IdentityList_Type v_DrbIdList := {};

    var PdcpCountInfoList_Type v_PdcpCountInfoList := f_EUTRA_SS_PdcpCount_GetAll(p_CellId);

    var integer k := 0;

    var integer i;


    for (i := 0; i < lengthof(v_PdcpCountInfoList); i := i + 1) {

      if (ischosen(v_PdcpCountInfoList[i].RadioBearerId.Drb)) {

        v_DrbIdList[k] := v_PdcpCountInfoList[i].RadioBearerId.Drb;     /* @sic R5s150338: release all DRBs being configured at the SS sic@ */

      }

    }
    /* Step 1 at T: disable the UL grant

     *               => it is not possible for the UE to send any ACK for the RRC Connecion Release)

     *               => UE has to wait until 60ms timer has timed out (-> 36.331 cl. 5.3.8.3) */

    f_EUTRA_StopULGrantTransmission(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 2 at T+5ms: release security */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 5);

    f_EUTRA_SS_RRC_ReleaseSecurity(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 3 at T+10ms: release DRX */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 10);

    f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_None, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 3A at T+15ms: release MeasGapConfig */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 15);

    f_EUTRA_SS_MeasGapCtrlConfig(p_CellId, cs_MeasGapCtrl_None, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 4 (obsolete) */

    /* Step 5 at T+55ms: Release SRBs and DRBs */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 55);

    f_EUTRA_SS_SRBs_DRBs_Release(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), v_DrbIdList);

    /* Step 6 at T+60ms: Configure SRBs and DRBs */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 60);

    f_EUTRA_SS_SRBs_DRBs_Config(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), omit);

  }


Related Changes:
1. p_DrbConfigList removed from f_EUTRA_RRC_ConnectionRelease_Common, f_EUTRA_RRC_ConnectionRelease and all functions calling f_EUTRA_RRC_ConnectionRelease
2. f_EUTRA_SS_SRBs_DRBs_Config adapted to allow p_DrbConfigList to be omitted:
	  function f_EUTRA_SS_SRBs_DRBs_Config(EUTRA_CellId_Type p_CellId,

                                       template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                       template (omit) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def,

                                       template (omit) boolean p_CnfFlag := omit) runs on EUTRA_PTC

  { /* @sic R5s100801, R5s110634 additional comments: optional parameter p_CnfFlag, R5s110602 sic@ */

    var template (value) RadioBearerList_Type v_DrbConfigList;
    var template (value) RadioBearerList_Type v_SrbDrbList := {

      cs_SRB1_ConfigDef,

      cs_SRB2_ConfigDef

    };

    var integer i := lengthof(v_SrbDrbList);

    var integer k;




    if (ispresent(p_DrbConfigList)) {

      v_DrbConfigList := valueof(p_DrbConfigList);

      for (k:=0; k < lengthof(v_DrbConfigList); k:= k+1) {

        v_SrbDrbList[i+k] := v_DrbConfigList[k];          /* @sic R5s100279 change 2 sic@ */

      }

    }
    f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, v_SrbDrbList, p_TimingInfo, p_CnfFlag);

  }


Change 7 – Cell configuration of test cases

Change 7.1 – Cell Configuration – default DRB only
	Function name
	f_EUTRA_CellConfig_Def, f_EUTRA_CellConfig_SRBs_Only

	Reason for change
	In contrast to static DRB configuration for support of dynamic DRB configuration in general DRBs shall not be pre-configured at the SS; for that reason DRB1 gets configured e.g. in f_EUTRA_RRC_ActivateSecurity now

	Summary of changes
	· DRB configuration removed from default cell configuration function f_EUTRA_CellConfig_Def;

· f_EUTRA_CellConfig_Def enhanced with p_DiscardULDataUM and p_UseBigGrants
· Due to changes of f_EUTRA_CellConfig_Def, f_EUTRA_CellConfig_Def and f_EUTRA_CellConfig_SRBs_Only in fact are the same but f_EUTRA_CellConfig_SRBs_Only is kept as wrapper function

	TTCN module
	EUTRA_ConfigurationSteps


	  function f_EUTRA_CellConfig_Def(EUTRA_CellId_Type p_CellId,

                                  template (omit) EUTRA_DiscardUlDataUM_Type p_DiscardULDataUM := omit,

                                  template (omit) EUTRA_UseBigGrants_Type p_UseBigGrants := omit,
                                  template (omit) CA_CellInitialConfig_Type p_CA_CellInitialConfig := omit,//@sic R5s150252 sic@

                                  EUTRA_CyclicPrefix_Type p_CyclicPrefixLength := normal) runs on EUTRA_PTC

  {

    f_EUTRA_SS_ConfigureActiveCell(p_CellId, p_CA_CellInitialConfig, p_CyclicPrefixLength); //@sic R5s150252 sic@

    f_EUTRA_SS_ConfigureSRBs(p_CellId);


    if (ispresent(p_DiscardULDataUM)) {      /* @sic R5s150338 sic@ */

      f_EUTRA_DiscardUlDataUM_Set(p_DiscardULDataUM);

    }

    if (ispresent(p_UseBigGrants)) {         /* @sic R5s150338 sic@ */

      f_EUTRA_ConfigureBigULGrant(p_CellId);

    }
  }

	  function f_EUTRA_CellConfig_SRBs_Only (EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  {




    f_EUTRA_CellConfig_Def(p_CellId);
  }


Further changes: Enhancement of formal parameter list when calling f_EUTRA_CellConfig_Def in
· f_EUTRA_CellConfig_Pcell_Def
· f_TC_7_1_3_12_Common

· f_TC_7_1_3_13_Common
Change 7.2 – Cell Configuration – default DRB + 1 UM DRB
	Function name
	f_EUTRA_CellConfig_DefPlusUM

	Reason for change
	DRBs shall not be pre-configured at the SS 

	Summary of changes
	1. f_EUTRA_CellConfig_DefPlusUM is removed from TTCN and replaced by f_EUTRA_CellConfig_Def wherever being used
2. f_EUTRA_CellConfig_DefPlusUM_SN5 is removed from TTCN and replaced by f_EUTRA_CellConfig_Def wherever being used
3. f_EUTRA_CellConfig_DefPlusUM_PDCP_SN7 is removed from TTCN and replaced by f_EUTRA_CellConfig_Def wherever being used
( Only SRBs are pre-configured

	TTCN module
	EUTRA_ConfigurationSteps


Change 7.2.1 - Further changes due to removal of f_EUTRA_CellConfig_DefPlusUM_SN5
	function f_TC_7_2_2_1_EUTRA() runs on EUTRA_PTC

{ // @sic R5s090314 sic@

  var RLC_SS_State v_RLC_Rec;
  var template (value) RadioBearerList_Type v_DrbList := {   

    cs_OneDRB_ConfigAM,

    cs_OneDRB_ConfigUM(tsc_DRB2, cs_508_PDCP_Config_DRB_UM, cs_RLC_Configuration_UM_SN5)

  };
  // Initialization, call setup, DRB loopback

  f_EUTRA_Init(c1);

  f_InitRLC_Record(v_RLC_Rec, UM_Mode);

  v_RLC_Rec.UM_SN_Size := tsc_UM_SN_Size_5;

  // Cell configuration, with one UM DRB at the SS side

  f_EUTRA_CellConfig_Def(eutra_Cell1);  // @sic R5s100076 sic@

  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  //entering UE test state state 3 with test case specific RLC parameters, with BSR and PHR disabled

  f_EUTRA_GenericRbEst_UM_Params (eutra_Cell1,

                                  cs_DRB_ConfigurationAtSS_Specific(v_DrbList),

                                  cs_508_PDCP_Config_DRB_UM,

                                  cds_TX_UM_RLC_DRB_SN5,

                                  cds_RX_UM_RLC_DRB_SN5);

  // reconfigure DRB locally to set the required test mode UM mapped to TM

  f_EUTRA_SS_DRB2_ConfigUM_TM_Mode (eutra_Cell1, v_RLC_Rec.UM_SN_Size);

  //closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA (eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);

  f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

  fl_TC_7_2_2_1_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

  //postamble

  f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE);

}

	  function f_TC_7_2_2_3_EUTRA() runs on EUTRA_PTC

  { // @sic R5s090318 sic@

    var RLC_SS_State v_RLC_Rec; 
    var template (value) RadioBearerList_Type v_DrbList := {   // @sic R5s150338 change 7.2.1 sic@ 

      cs_OneDRB_ConfigAM,

      cs_OneDRB_ConfigUM(tsc_DRB2, cs_508_PDCP_Config_DRB_UM, cs_RLC_Configuration_UM_SN5)

    };
    //EUTRA initialisation

    f_EUTRA_Init(c1);

    f_InitRLC_Record(v_RLC_Rec, UM_Mode);

    v_RLC_Rec.UM_SN_Size := tsc_UM_SN_Size_5;

    // Cell configuration, with one UM DRB at the SS side
    f_EUTRA_CellConfig_Def(eutra_Cell1);  /* @sic R5s150338: f_EUTRA_CellConfig_DefPlusUM_SN5 replaced by f_EUTRA_CellConfig_Def sic@ */

    // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

    f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    //set null AS ciphering algorithm

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

    //entering UE test state state 3 with test case specific RLC parameters, with BSR and PHR disabled

    f_EUTRA_GenericRbEst_UM_Params (eutra_Cell1, 
                                    cs_DRB_ConfigurationAtSS_Specific(v_DrbList),     // @sic R5s150338 change 7.2.1 sic@
                                    cs_508_PDCP_Config_DRB_UM,

                                    cds_TX_UM_RLC_DRB_SN5,

                                    cds_RX_UM_RLC_DRB_SN5);

    // reconfigure DRB locally to set the required test mode UM mapped to TM

    f_EUTRA_SS_DRB2_ConfigUM_TM_Mode (eutra_Cell1, v_RLC_Rec.UM_SN_Size);

    //closing the test loop (mode A), entering state 4

    f_EUTRA_CloseUE_TestLoopModeA (eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);

    f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

    fl_TC_7_2_2_3_TestBody (v_RLC_Rec);

    f_EUTRA_TestBody_Set(false);

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    //postamble

    f_EUTRA_Postamble (eutra_Cell1, E3_TESTMODE);

  }

	  function f_TC_7_2_2_5_1_EUTRA() runs on EUTRA_PTC

  { // @sic R5s090340 sic@

    var RLC_SS_State v_RLC_Rec; 
    var template (value) RadioBearerList_Type v_DrbList := {   // @sic R5s150338 change 7.2.1 sic@ 

      cs_OneDRB_ConfigAM,

      cs_OneDRB_ConfigUM(tsc_DRB2, cs_508_PDCP_Config_DRB_UM, cs_RLC_Configuration_UM_SN5)

    };
    // Initialization, call setup, DRB loopback

    f_EUTRA_Init(c1);

    f_InitRLC_Record(v_RLC_Rec, UM_Mode);

    v_RLC_Rec.UM_SN_Size := tsc_UM_SN_Size_5;

    // Cell configuration, with one UM DRB at the SS side
    f_EUTRA_CellConfig_Def(eutra_Cell1);  /* @sic R5s150338: f_EUTRA_CellConfig_DefPlusUM_SN5 replaced by f_EUTRA_CellConfig_Def sic@ */

    // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

    f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    //set null AS ciphering algorithm

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

    //entering UE test state state 3 with test case specific RLC parameters, with BSR and PHR disabled

    f_EUTRA_GenericRbEst_UM_Params (eutra_Cell1, 
                                    cs_DRB_ConfigurationAtSS_Specific(v_DrbList),     // @sic R5s150338 change 7.2.1 sic@
                                    cs_508_PDCP_Config_DRB_UM,

                                    cds_TX_UM_RLC_DRB_SN5,

                                    cds_RX_UM_RLC_DRB_SN5);

    // reconfigure DRB locally to set the required test mode UM mapped to TM

    f_EUTRA_SS_DRB2_ConfigUM_TM_Mode (eutra_Cell1, v_RLC_Rec.UM_SN_Size);

    //closing the test loop (mode A), entering state 4

    f_EUTRA_CloseUE_TestLoopModeA (eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);

    f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

    fl_TC_7_2_2_5_1_TestBody (v_RLC_Rec);

    f_EUTRA_TestBody_Set(false);

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    //postamble

    f_EUTRA_Postamble (eutra_Cell1, E3_TESTMODE);

  }


Change 7.2.2 - Further changes due to removal of f_EUTRA_CellConfig_DefPlusUM_PDCP_SN7

	  function f_TC_7_3_1_2_EUTRA() runs on EUTRA_PTC

  { // @sic R5-100082 sic@

    var PDCP_SS_State_Type v_PDCP_Rec;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info; 
    var template (value) RadioBearerList_Type v_DrbList := {   // @sic R5s150338 change 7.2.2 sic@ 

      cs_OneDRB_ConfigAM,

      cs_OneDRB_ConfigUM(tsc_DRB2, cds_PDCP_Config_DRB_UM_SN7)

    };
    // Initialization, generic setup, DRB loopback

    f_EUTRA_Init(c1);

    fl_InitPDCP_Record(v_PDCP_Rec, UM_Mode, PDCP_SNLength7);

    // get FDD_TDD_Info

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);

    // Cell configuration, with one UM DRB at the SS side

    f_EUTRA_CellConfig_Def(eutra_Cell1);  /* @sic R5s150338: f_EUTRA_CellConfig_DefPlusUM_PDCP_SN7 replaced by f_EUTRA_CellConfig_Def sic@ */

    // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    //set null AS ciphering algorithm

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

    //entering UE test state state 3 with test case specific RLC parameters, with BSR and PHR disabled

    f_EUTRA_GenericRbEst_UM_Params(eutra_Cell1, 
                                   cs_DRB_ConfigurationAtSS_Specific(v_DrbList),     // @sic R5s150338 change 7.2.2 sic@
                                   cds_PDCP_Config_DRB_UM_SN7, // @sic R5s100302 sic@

                                   cs_508_DRB_UL_UM_RLC,

                                   cs_508_DRB_DL_UM_RLC);

    // reconfigure DRB locally to set the required PDCP test mode

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_DRB2_ConfigUM_PDCP_Mode(v_PDCP_Rec.PDCP_SN_Size));

    // closing the test loop (mode A), entering state 4

    f_EUTRA_CloseUE_TestLoopModeA(eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);

    f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

    f_EUTRA_7_3_1_2_TestBody(v_PDCP_Rec);

    f_EUTRA_TestBody_Set(false);

    // leaving UE test state 4

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    //postamble

    f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE);

  }


Change 7.3 – Cell Configuration – default DRB + n AM DRBs + m UM DRBs
	Function name
	f_EUTRA_CellConfig_DRB_Common

	Reason for change
	DRBs shall not be pre-configured at the SS 

	Summary of changes
	f_EUTRA_CellConfig_DRB_Common is removed from TTCN and replaced by f_EUTRA_CellConfig_Def wherever being used
( Only SRBs are pre-configured

	TTCN module
	EUTRA_ConfigurationSteps


Change 7.4 – Cell Configuration – SRVCC
	Function name
	f_EUTRA_CellConfig_DefPlusUM_SRVCC

	Reason for change
	DRBs shall not be pre-configured at the SS

	Summary of changes
	f_EUTRA_CellConfig_DefPlusUM_SRVCC renamed to f_EUTRA_CellConfig_SRVCC and calling f_EUTRA_CellConfig_Def instead of f_EUTRA_CellConfig_DefPlusUM
( Only SRBs are pre-configured

	TTCN module
	EUTRA_CommonSRVCC


	  function f_EUTRA_CellConfig_ SRVCC(EUTRA_CellId_Type p_CellId,

                                              template (omit) EUTRA_DiscardUlDataUM_Type p_DiscardULDataUM := DISCARD_UM_DATA,

                                              template (omit) EUTRA_UseBigGrants_Type p_UseBigGrants := USE_BIG_GRANTS) runs on EUTRA_PTC

  { /* @sic R5s130681: optional parameter p_DiscardULDataUM sic@ */

    /* @sic R5s140160, R5w140109: optional parameter p_DiscardULDataUM sic@ */

    /* @sic R5-145773: no ROHC for SRVCC sic@ */

    f_EUTRA_CellConfig (p_CellId, p_DiscardULDataUM, p_UseBigGrants);

  }


Change 7.4 – Cell Configuration – CA
	Function name
	f_EUTRA_CellConfig_ScellToPcell_Def

	Reason for change
	DRBs shall not be pre-configured at the SS; f_EUTRA_CellConfig_DefPlusUM has been removed (see above)

	Summary of changes
	f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB removed from f_EUTRA_CellConfig_ScellToPcell_Def
( Only SRBs are pre-configured

NOTE: 
f_EUTRA_CellConfig_ScellToPcell_Def is called by f_TC_8_2_4_23_Common and f_EUTRA_508RRC_IntraLTE_CA_HO_InterCell:

· in case of f_TC_8_2_4_23_Common DRB1 is configured by f_EUTRA_SS_AS_ActivateSecurity_HO_Reest after f_EUTRA_CellConfig_ScellToPcell_Def has been called

· in case of f_EUTRA_508RRC_IntraLTE_CA_HO_InterCell additional DRBs are configured by f_EUTRA_SS_CommonRadioBearerConfig directly after f_EUTRA_CellConfig_ScellToPcell_Def has been called

	TTCN module
	EUTRA_CommonFunctions_CA


	  function f_EUTRA_CellConfig_ScellToPcell_Def(EUTRA_CellId_Type p_CellId, template (value) TimingInfo_Type p_TimingInfo) runs on EUTRA_PTC

  {

    f_EUTRA_SS_CommonCellConfig(p_CellId, cas_CellConfig_PcellMod_REQ(p_CellId, p_TimingInfo, f_EUTRA_CellInfo_Get(p_CellId), cs_BasicCellConfig_CA_CellType(PCell)));

    //Update Cell Type and UL power control

    f_EUTRA_SS_ConfigureSRBs(p_CellId, p_TimingInfo);


  }


Further related change: f_EUTRA_SS_DRB1_DefConfig is removed as not being used anymore

Change 8 – Further changes

Change 8.1 – fl_TC_10_3_1_and_10_9_1_Prepare2ndRun
	Function name
	fl_TC_10_3_1_and_10_9_1_Prepare2ndRun

	Reason for change
	Due to dynamic DRB configuration the DRBs shall not be reconfigured in fl_TC_10_3_1_and_10_9_1_Prepare2ndRun

	Summary of changes
	Reconfiguration of DRBs removed 

	TTCN module
	ESM_Testcases


	  function fl_TC_10_3_1_and_10_9_1_Prepare2ndRun (EUTRA_CellId_Type p_CellId,

                                                  template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_DRB1) runs on EUTRA_PTC  // @sic R5s130366 sic@

  {

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    // Activate DHCP and ICMPv6 - had been stopped previously when test loop was activated

    f_IP_Handling_Start(IP, PDN_1, cs_DrbInfo_EUTRA(p_CellId, tsc_DRB1)); // @sic R5s110673 sic@ @sic R5s110776 sic@

    // open test loop

    // deactivate test mode

    f_EUTRA_OpenUE_TestLoopMode_Deactivate_TestMode (p_CellId);

    // The UE is brought into state Switched OFF (state 1) according to 36.508.

    f_EUTRA_UE_Detach_SwitchOffUe(eutra_CellA, RRC_CONNECTED);  // @sic R5s130122 sic@

    // reset DRBs explicitly  @sic R5s130366 sic@

    f_EUTRA_SS_SRBs_DRBs_Release(eutra_CellA, -, v_DRB_IdentityList.All);

    f_EUTRA_SS_SRBs_DRBs_Config(eutra_CellA);

    // Switched ON the UE

    f_UT_SwitchOnUE(UT, false);   // @sic R5s130122 sic@

  }


Change 8.2 – Configuration of DRB1 when f_EUTRA_RRC_ActivateSecurity is not called
	Function name
	f_TC_8_2_1_5_EUTRA, f_TC_8_2_1_5_EUTRA, fl_TC_9_1_3_3_Body, f_TC_8_2_1_1_EUTRA, f_TC_8_2_1_8_EUTRA

	Reason for change
	Due to change of f_EUTRA_CellConfig_Def DRB1 is not pre-configured anymore

	Summary of changes
	explicit configuration of DRB1 in cases where no common functions are doing this

	TTCN module
	RRC_ConnReconfig, NAS_SecurityMode


Change 8.2.1 - f_TC_8_2_1_5_EUTRA
	  function f_TC_8_2_1_5_EUTRA() runs on EUTRA_PTC

  { /* RRC connection reconfiguration / Radio bearer establishment for transition from RRC_Idle to RRC CONNECTED / Success / Latency check */

    var EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    var PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var O4_Type v_NasCountUL;

    var NAS_KsiValue v_KsiValue;

    var SRB_COMMON_IND v_ReceivedAsp;

    var SubFrameTiming_Type v_Timing;

    var integer v_N;

    var boolean v_IsFDD;

    f_EUTRA_Init(c1);

    //Get FDD or TDD mode from Cell-configuration

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell1);

    v_IsFDD := (v_EUTRA_FDD_TDD_Mode == FDD);

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    /* Preamble to enter UE in E-UTRA RRC_IDLE (state 2) */

    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);

    //Get security parameters

    v_Auth_Params := f_EUTRA_Security_Get();

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    // enable reporting of HARQ errors

    f_EUTRA_SS_CommonL1MacIndCtrlConfig(eutra_Cell1, cas_HarqErrorIndication_REQ(eutra_Cell1));

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //The SS sends a Paging message to the UE on the appropriate paging block, and including the UE identity in one entry of the IE pagingRecordLists.

    f_EUTRA_UE_Page_Def(eutra_Cell1);

    //@siclog "Step 2" siclog@

    //The UE transmits an RRCConnectionRequest message.

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1, cr_EstablishmentCause_mt_Access);    // @sic R5-120721 change 3 sic@

    //@siclog "Step 3" siclog@

    //The SS transmit an RRCConnectionSetup message. (Note 3)

    //@siclog "Step 4" siclog@

    //Check: Does the UE transmit an RRCConnectionSetupComplete message within 20 (FDD)/23 (TDD) subframes after successful completion of step 3? (Note 2)

    //Note: to avoid unexpected TA commands (resulting in additional HARQ indications) the time alignment timer needs to be set to infinitiv

    v_N := 15;

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    v_ReceivedAsp := fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,

                                                    cas_SRB0_RrcPdu_REQ(eutra_Cell1,

                                                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                                                        f_EUTRA_508_RRCConnectionSetup(eutra_Cell1, -, infinity_)),   /* @sic R5-120721 change 3 sic@ */

                                                    -,

                                                    car_SRB1_RrcNasPdu_IND(eutra_Cell1,

                                                                           cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                                             cr_508_SERVICE_REQUEST(v_KsiValue))),

                                                    -,

                                                    v_IsFDD,

                                                    v_N,

                                                    "Test Case 8.2.1.5 Step 4");

    v_NasCountUL := v_ReceivedAsp.Signalling.Nas[0].SecurityProtection.NasCount;

    fl_EUTRA_RRC_InitialTA(eutra_Cell1);     /* @sic R5s120201 sic@ */

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_OneDRB_ConfigAM);    // @sic R5s150338: DRB1 is not configured in f_EUTRA_CellConfig_Def anymore sic@

    v_Auth_Params := f_EUTRA_Authentication_InitAS(v_Auth_Params, v_NasCountUL);        /* Calculate Kenb and KRRCenc, KRRCint, KUPenc */

    // Read PDCP SQN for all RBs and calcuate Ciphering Activation times

    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetInitial(eutra_Cell1);

    f_EUTRA_SS_RRC_EnableIntegrityProtection(eutra_Cell1, v_Auth_Params.AS_Integrity);  /* SecurityModeCommand is integrity protected with the algorithm

                                                                                           being contained in the message; all sub-sequent RRC messages

                                                                                           are integrity protected in UL and DL (36.331 cl. 5.3.4.3)    */

    f_EUTRA_SS_RRC_EnableCipheringULandDL(eutra_Cell1, v_Auth_Params.AS_Ciphering);

    //@siclog "Step 5" siclog@

    //The SS transmits a SecurityModeCommand message to activate AS security. (Note 3)

    //@siclog "Step 6" siclog@

    //Check: Does the UE transmit a SecurityModeComplete message within 15 (FDD)/18 (TDD) subframes after successful completion of step 5? (Note 2)

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    v_N := 10;

    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,

                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                                       cs_508_SecurityModeCommand(tsc_RRC_TI_Def,

                                                                                  v_Auth_Params.AS_Integrity.Algorithm,

                                                                                  v_Auth_Params.AS_Ciphering.Algorithm)),

                                   -,

                                   car_SRB1_RrcPdu_IND(eutra_Cell1,

                                                       cr_508_SecurityModeComplete(tsc_RRC_TI_Def)),

                                   -,

                                   v_IsFDD,

                                   v_N,

                                   "Test Case 8.2.1.5 Step 6");

    //@siclog "Step 7" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to establish a data radio bearer. (Note 3)

    //@siclog "Step 8" siclog@

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message within 20 (FDD)/23 (TDD) subframes after successful completion of step 7? (Note 2)

    v_Timing := f_EUTRA_GetNextSendOccasion ( eutra_Cell1 );

    v_N := 15;

    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,

                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                                       cs_RRCConnectionReconfiguration_RadioResConfig(tsc_RRC_TI_Def,

                                                                                                      cs_508_RadioResourceConfigDedicated_SRB2({ cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1) },

                                                                                                                                               cs_MAC_MainConfig_Explicit_TA_Infinity,  /* @sic R5s110530 change1; R5-113706 sic@ */

                                                                                                                                               cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                                                          v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                                                          v_AntennaInfo)))),

                                   -,

                                   car_SRB1_RrcPdu_IND(eutra_Cell1,

                                                       cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)),

                                   -,

                                   v_IsFDD,

                                   v_N,

                                   "Test Case 8.2.1.5 Step 8");

    //@siclog "Step 9" siclog@

    //The SS transmits a UECapabilityEnquiry message to request UE radio access capability information for E UTRA only. (Note 3)

    //@siclog "Step 10" siclog@

    //Check: Does the UE transmit a UECapabilityInformation message within 15 (FDD)/18 (TDD) subframes after successful completion of step 9? (Note 2)

    v_Timing := f_EUTRA_GetNextSendOccasion ( eutra_Cell1 );

    v_N := 10;

    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,

                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                                       cs_508_UeCapabilityEnquiry(tsc_RRC_TI_Def)),

                                   -,

                                   car_SRB1_RrcPdu_IND(eutra_Cell1,

                                                       cr_508_UeCapabilityInformation(tsc_RRC_TI_Def, cr_EutraCapInfo)),

                                   -,

                                   v_IsFDD,

                                   v_N,

                                   "Test Case 8.2.1.5 Step 10");

    f_EUTRA_TestBody_Set(false);

    /* Postamble */

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);           // UL grants per scheduling request

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);

  }


Change 8.2.2 - f_TC_8_2_1_6_EUTRA
	  function f_TC_8_2_1_6_EUTRA() runs on EUTRA_PTC

  { /* RRC connection reconfiguration / Radio bearer establishment for transition from RRC_Idle to

       RRC CONNECTED / Success / Latency check / SecurityModeCommand and RRCConnectionReconfiguration transmitted in the same TTI */

    var EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    var PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var O4_Type v_NasCountUL;

    var NAS_KsiValue v_KsiValue;

    var NAS_MSG_Indication_Type v_NasInd;

    var SubFrameTiming_Type v_Timing;

    var integer v_N;

    var boolean v_IsFDD;

    f_EUTRA_Init(c1);

    //Get FDD or TDD mode from Cell-configuration

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info( eutra_Cell1 );

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD( eutra_Cell1 );

    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated( eutra_Cell1 );

    v_IsFDD := (v_EUTRA_FDD_TDD_Mode == FDD);

    //Create and configure cell

    f_EUTRA_CellConfig_Def( eutra_Cell1 );

    /* Preamble to enter UE in E-UTRA RRC_IDLE (state 2) */

    f_EUTRA_Preamble( eutra_Cell1, STATE2_IDLEUPDATE );

    //Get security parameters

    v_Auth_Params := f_EUTRA_Security_Get();

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    //Enable reporting of HARQ errors

    f_EUTRA_SS_CommonL1MacIndCtrlConfig( eutra_Cell1, cas_HarqErrorIndication_REQ( eutra_Cell1 ) );

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //The SS sends a Paging message to the UE on the appropriate paging block, and including the UE identity in one entry of the IE pagingRecordLists.

    f_EUTRA_UE_Page_Def( eutra_Cell1 );

    //@siclog "Step 2" siclog@

    //The UE transmits an RRCConnectionRequest message.

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1, cr_EstablishmentCause_mt_Access);    // @sic R5-120721 change 3 sic@

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_ULGrantAllocation_Def_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PUCCH_Synch_None));  // @sic R5s110469 sic@

    //@siclog "Step 3" siclog@

    //The SS transmit an RRCConnectionSetup message.

    //Note: to avoid unexpected TA commands (resulting in additional HARQ indications) the time alignment timer needs to be set to infinity

    SRB.send( cas_SRB0_RrcPdu_REQ( eutra_Cell1,

                                   cs_TimingInfo_Now,

                                   f_EUTRA_508_RRCConnectionSetup(eutra_Cell1, -, infinity_)));   /* @sic R5-120721 change 3 sic@ */

    //@siclog "Step 4" siclog@

    //The UE transmit an RRCConnectionSetupComplete message.

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,                                       // @sic R5s120121 MCC160 implementation sic@

                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_SERVICE_REQUEST(v_KsiValue)));

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    fl_EUTRA_RRC_InitialTA(eutra_Cell1);     /* @sic R5s120201 sic@ */

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_OneDRB_ConfigAM);    // @sic R5s150338: DRB1 is not configured in f_EUTRA_CellConfig_Def anymore sic@

    v_Auth_Params := f_EUTRA_Authentication_InitAS(v_Auth_Params, v_NasCountUL);        /* Calculate Kenb and KRRCenc, KRRCint, KUPenc */

    // Read PDCP SQN for all RBs and calcuate Ciphering Activation times

    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetInitial(eutra_Cell1);

    f_EUTRA_SS_RRC_EnableIntegrityProtection( eutra_Cell1, v_Auth_Params.AS_Integrity );  /* SecurityModeCommand is integrity protected with the algorithm

                                                                                             being contained in the message; all sub-sequent RRC messages

                                                                                             are integrity protected in UL and DL (36.331 cl. 5.3.4.3)    */

    f_EUTRA_SS_RRC_EnableCipheringULandDL( eutra_Cell1, v_Auth_Params.AS_Ciphering );

    //@siclog "Step 5" siclog@

    //The SS transmits a SecurityModeCommand message to activate AS security. (Note 3)

    //@siclog "Step 6" siclog@

    //The SS transmits a RRCConnectionReconfiguration message to establish a data radio bearer. (Note 3)

    //@siclog "Step 7" siclog@

    //Check: Does the UE transmit an SecurityModeComplete message within 25 (FDD)/28 (TDD) subframes after successful completion of step 5? (Note 2)

    //@siclog "Step 8" siclog@

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message within 25 (FDD)/28 (TDD) subframes after successful completion of step 6? (Note 2)

    v_Timing := f_EUTRA_GetNextSendOccasion( eutra_Cell1 );

    v_N := 20;

    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,

                                   cas_SRB1_RrcPdu_REQ( eutra_Cell1,

                                                        cs_TimingInfo( v_Timing.SFN.Number, v_Timing.Subframe.Number ),

                                                        cs_508_SecurityModeCommand(tsc_RRC_TI_Def,

                                                                                   v_Auth_Params.AS_Integrity.Algorithm,

                                                                                   v_Auth_Params.AS_Ciphering.Algorithm ) ),

                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                                       cs_RRCConnectionReconfiguration_RadioResConfig(tsc_RRC_TI_Def,

                                                                                                      cs_508_RadioResourceConfigDedicated_SRB2({ cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1) },

                                                                                                                                               cs_MAC_MainConfig_Explicit_TA_Infinity,  /* @sic R5-113706 sic@ */

                                                                                                                                               cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                                                          v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                                                          v_AntennaInfo)))),

                                   car_SRB1_RrcPdu_IND( eutra_Cell1,

                                                        cr_508_SecurityModeComplete( tsc_RRC_TI_Def ) ),

                                   car_SRB1_RrcPdu_IND( eutra_Cell1,

                                                        cr_508_RRCConnectionReconfigurationComplete( tsc_RRC_TI_Def ) ),

                                   v_IsFDD,

                                   v_N,

                                   "Test Case 8.2.1.6 Steps 7 and 8");

    f_EUTRA_TestBody_Set(false);

    /* Postamble */

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);           // UL grants per scheduling request

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);

  }


Change 8.2.3 - fl_TC_9_1_3_3_Body
	  function fl_TC_9_1_3_3_Body() runs on EUTRA_PTC

  {

    var HalfOctet_Type v_EpsBearerId := tsc_EpsDefaultBearerId;

    var ProcedureTransactionIdentifier v_EPS_TI := '01'O;

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var EUTRA_SecurityParams_Type v_SecurityParamsPrev; // @sic R5s130201 sic@

    var UE_SecurityCapability v_UeSecurityCapabilities;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var template (omit) ProtocolConfigOptions v_Pco := omit;

    var boolean v_EIT_Flag := false;

    var boolean v_IPallocationViaNasFlag :=  false;

    var NAS_MSG_Indication_Type v_NasInd;  // temporary variable which is used when receiving NAS ASPs

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();   //used to follow TTCN3 v411 standard

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(NORMAL);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellA );

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN,v_LAC, NORMAL);

    var template (omit) MobileIdentity v_MSId := f_GetMSId (NORMAL);

    var template (value) NAS_MSG_Request_Type v_NAS_MSG_Request;

    var template (value) AccessPointName v_APN;

    timer t_Wait := 1.0;

    timer t_T3411min := f_EUTRA_SetTimerToleranceMin (eutra_CellA, nasTimer, 10.0); // @sic R5s130201 sic@

    //+   The UE is switched on.

    //@siclog "Step 1" siclog@

    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //+   The UE transmits an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message.

    //@siclog "Step 2" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_Any,

                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                 cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                 cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                                 f_GetPdnType())));

    v_PdnConnectivityRequest := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    // Get the PCO to be used later (TS 36.508 v.8.2.2 Table 4.7.3-6)

    if (ispresent( v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    v_EIT_Flag := f_CheckEsmInfoTransferFlag(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST);

    v_UeSecurityCapabilities := f_EUTRA_FillNAS_SecurityCapability(v_NasInd.Pdu.Msg.aTTACH_REQUEST);

    //+

    //+   The SS transmits an AUTHENTICATION REQUEST message with SQN out of range and an invalid MAC code, and, "separation bit" in the AMF field of AUTN = 1.

    //+   Note: The present TC is simulating a man-in-the-middle security threat scenario. The man-in-the-middle is not expected to have these parameters right.

    //@siclog "Step 3 Void" siclog@

    v_SecurityParams := f_EUTRA_Authentication_InitNAS(v_SecurityParams,

                                                       v_PLMN,

                                                       f_AuthenticationInit(v_SecurityParams.AuthParams, macErrorPlusSepBit));

    //@sic R5-144345 sic@

    /*SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                cs_AUTHENTICATION_REQUEST(v_SecurityParams.KSIasme,

                                                                          v_SecurityParams.AuthParams.RandValue,

                                                                          v_SecurityParams.AuthParams.AUTN))));

    //+   UE transmits an AUTHENTICATION FAILURE message with EMM cause #20 "MAC failure" or EMM cause #21 "synch failure". */

    //@siclog "Step 4 void" siclog@

    //@sic R5-144345 sic@

    /*SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected,   // @sic R5s130201 sic@

                                                      (cr_AUTHENTICATION_FAIL_synch, cr_AUTHENTICATION_FAIL_mac)))); */

    // At the UE side T3410 is stopped and T3411 is started  @sic R5s130201 sic@

    t_T3411min.start;

    //+   The SS transmits a NAS SECURITY MODE COMMAND message; EIA0 (NULL integrity), EEA0 (NULL ciphering), matched replayed security capabilities.

    //+   Note: 'matched replayed security capabilities' shall be sent to ensure that the SECURITY MODE REJECT is not sent due to problem with this information.

    //@siclog "Step 5" siclog@

    v_SecurityParamsPrev := f_EUTRA_Security_Get();     // save previous Security Parameters @sic R5s130201 sic@

    v_SecurityParams.NAS_Ciphering.Algorithm := '000'B; // EEA0

    v_SecurityParams.NAS_Integrity.Algorithm := '000'B; // EIA0

    v_SecurityParams.KSIasme := '000'B;

    f_EUTRA_Security_Set (v_SecurityParams); // store the new setting, do not configure security at the SS side as UE will not accept it @sic R5s130201 sic@

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_NewSecurityContext, // @sic R5s130201 sic@,  see 24.301 cl. 5.4.3.2

                                                cs_508_SECURITY_MODE_COMMAND(v_SecurityParams.NAS_Ciphering.Algorithm,

                                                                             v_SecurityParams.NAS_Integrity.Algorithm,

                                                                             v_SecurityParams.KSIasme,

                                                                             v_UeSecurityCapabilities)))); // replayedSecurityCapability

    //+   Check: Does the UE transmit a NAS SECURITY MODE REJECT message?

    //@siclog "Step 6" siclog@

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected,  // @sic R5s130201 sic@

                                                      cr_SECURITY_MODE_REJECT((tsc_UeSecurityCapsMismatch, tsc_SecurityModeRejUnspecified)))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //+   The SS transmits an IDENTITY REQUEST message for IMSI (Security not applied)

    //@siclog "Step 7" siclog@

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,  // @sic R5s130201 sic@

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                cs_IDENTITY_REQUEST(tsc_IdType_IMSI))));

    //+   The UE transmits a non security protected IDENTITY RESPONSE message.

    //@siclog "Step 8" siclog@

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,   // @sic R5s130201 sic@

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected,  // @sic R5s130201 sic@

                                                      cr_IDENTITY_RESPONSE(f_Imsi2MobileIdentity(px_IMSI_Def)))));

    //+   EXCEPTION: Steps 9a1 to 9a2 describe behaviour that depends on UE configuration; the "lower case letter" identifies a step sequence that take place

    //+   if the UE has ESM information which needs to be transferred.  -

    if (v_EIT_Flag) {

      //+   The SS transmits an ESM INFORMATION REQUEST message - no integrity protection applied - to initiate exchange of protocol configuration options and/or APN.

      //@siclog "Step 9a" siclog@

      SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                  cs_508_Esm_Information_Request(v_EPS_TI))));

      t_Wait.start;

      //+   Check: Does the UE transmit an ESM INFORMATION RESPONSE message?

      //+   Note: The UE is expected to discard the ESM INFORMATION REQUEST message without security protection.

      //@siclog "Step 9a2" siclog@  // @sic R5s130201 sic@

      alt {

        [] SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,

                                           cr_NAS_Indication(?,

                                                             cr_508_Esm_Information_Response(?)))) {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9a2"); // @sic R5s130201 sic@

        }

        [] t_Wait.timeout {}; // continue

      }

    }
    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_OneDRB_ConfigAM);
    // The SS transmits an ATTACH ACCEPT message- no integrity protection applied. The ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message is piggybacked in ATTACH ACCEPT message

    //@siclog "Step 10" siclog@

    v_IPallocationViaNasFlag :=  f_CheckPCOforIPallocationViaNas (v_Pco); // See TS 36.508 v.8.2.2 Table 4.7.3-6

    if (pc_IMS) {

      v_APN := cs_AccessPointName(f_EUTRA_GetAPNForIMS(eutra_CellA));

    } else {

      v_APN := cs_AccessPointName(px_AccessPointName);

    }

    v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(tsc_SHT_NoSecurityProtection, // @sic R5s130201 sic@

                                                        cs_508_ATTACH_ACCEPT(f_GetEAttachType(NORMAL),

                                                                             cs_TrackingAreaIdListDef_lv,

                                                                             f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams),

                                                                             v_LAI,

                                                                             v_MSId,

                                                                             omit,    // EMM_Cause

                                                                             f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.aTTACH_REQUEST.addUpdateType, NORMAL)),

                                                        cs_508_ActivateDefEpsBearerContextRequest(v_EpsBearerId,

                                                                                                  v_EPS_TI,

                                                                                                  v_APN,

                                                                                                  f_EUTRA_GetPDNAddress(v_IPallocationViaNasFlag),

                                                                                                  omit,

                                                                                                  f_GetDefaultProtocolConfigOptions(v_Pco)));

    f_EUTRA_RRCConnectionReconfiguration_SRB2_1AM_DRB_Send(eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_NAS_MSG_Request); // @sic R5s130201 sic@

    //+   EXCEPTION:    Steps 11a1 to 11b13 describe behaviour that depends on the UE action; the "lower case letter" identifies a step sequence

    //+   that take place if a particular sequential line of behaviour is manifested.

    alt {

      //+   The UE transmits an ATTACH COMPLETE message --> ATTACH COMPLETE

      //@siclog "Step 11a1" siclog@

      //+   The ATTACH COMPLETE message would need to be preceeded by a RRCConnectionReconfigurationComplete // @sic R5s130201 sic@

      [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)))

        {

          repeat;

        }

      [] SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                         cr_NAS_IndicationWithPiggybacking(?,

                                                                           cr_508_ATTACH_COMPLETE,

                                                                           ?)))

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 11a1");

        }

      [] t_T3411min.timeout

        {

          // continue  // @sic R5s130201 sic@

        }

    }

    // Reset SRBs and DRBs

    f_EUTRA_SS_SRBs_DRB1_Reset(eutra_CellA);  // @sic R5s130201 sic@

    //+   Check: Does UE transmit a NOT integrity protected ATTACH REQUEST message including

    //             IMSI and a PDN CONNECTIVITY REQUEST message?

    //@siclog "Step 11b1" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_Any,

                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected, // @sic R5s130201 sic@

                                                                                 cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                 cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                                 f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    v_EPS_TI := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    if (ispresent( v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    f_EUTRA_Security_Set(v_SecurityParamsPrev); // store the previous setting    // @sic R5s130201 sic@

    //+   The attach procedure is completed by executing steps 5 to 16 of the UE registration procedure in TS 36.508 [18] sub clause 4.5.2.3.

    //@siclog "Step 11b2-11b13" siclog@

    f_EUTRA_IdleUpdated_Step5_17 (eutra_CellA, STATE2_IDLEUPDATE, NORMAL, v_NasInd, false);

  }


Change 8.3.4 - f_TC_8_2_1_1_EUTRA
	  function f_TC_8_2_1_1_EUTRA() runs on EUTRA_PTC

  { /* RRC Connection Reconfiguration / Radio Bearer Establishment for transition from RRC_IDLE to RRC_CONNECTED: Success (Default bearer, early bearer establishment) */

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var B3_Type v_KsiValue;

    var O4_Type v_NasCountUL;

    var NAS_MSG_Indication_Type v_NasInd;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;

    f_EUTRA_Init(c1);

    //Get FDD or TDD mode from Cell-configuration

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);

    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell1);

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    /* Preamble to enter UE in E-UTRA RRC_IDLE (state 2) */

    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);

    //Get security parameters

    v_Auth_Params := f_EUTRA_Security_Get();

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //SS sends a Paging message to the UE on the appropriate paging block, and including the UE identity in one entry of the IE pagingRecordLists.

    f_EUTRA_UE_Page_Def(eutra_Cell1);

    //@siclog "Step 2" siclog@

    //UE transmits an RRCConnectionRequest message.

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1, cr_EstablishmentCause_mt_Access);    // @sic R5-120721 change 3 sic@

    //@siclog "Step 3" siclog@

    //SS transmit an RRCConnectionSetup message.

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_Cell1, -, v_RRC_TI);

    //@siclog "Step 4" siclog@

    //The UE transmits an RRCConnectionSetupComplete to confirm the successful completion of the connection establishment. (Note 1)

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,                                       // @sic R5s120121 MCC160 implementation sic@

                                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_SERVICE_REQUEST(v_KsiValue)));

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_OneDRB_ConfigAM);    // @sic R5s150338: DRB1 is not configured in f_EUTRA_CellConfig_Def anymore sic@
    //@siclog "Step 5" siclog@

    //SS transmits a SecurityModeCommand message to activate AS security.

    v_Auth_Params := f_EUTRA_Authentication_InitAS(v_Auth_Params,

                                                   v_NasCountUL);                  /* Calculate Kenb and KRRCenc, KRRCint, KUPenc */

    // Read PDCP SQN for all RBs and calcuate Ciphering Activation times

    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetInitial(eutra_Cell1);

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL(eutra_Cell1,

                                             v_Auth_Params.AS_Integrity,

                                             v_Auth_Params.AS_Ciphering);             /* SecurityModeCommand is integrity protected with the algorithm

                                                                                         being contained in the message; all sub-sequent RRC messages

                                                                                         are integrity protected in UL and DL (36.331 cl. 5.3.4.3);

                                                                                         @sic R5s100576 sic@ */

    // DL ciphering is activated for SRB1=(DL)SQN+1 i.e. for RRCConnectionReconfiguration and for other RBs, not yet configured, immediately.

    // UL ciphering is activated for SRB1=(UL)SQN+1 i.e. for RRCConnectionReconfigurationComplete and for other RBs, not yet configured, immediately.

    SRB.send(cads_SRB1_RrcPdu_REQ_FollowOnTRUE(eutra_Cell1,

                                               cs_TimingInfo_Now,

                                               cs_508_SecurityModeCommand(v_RRC_TI,

                                                                          v_Auth_Params.AS_Integrity.Algorithm,

                                                                          v_Auth_Params.AS_Ciphering.Algorithm)));

    /* Send SecurityModeCommand on SRB1. Follow on flag is set to TRUE

       to ensure that UE receives RRCConnectionReconfiguration message

       before it sends SecurityModeComplete message.  */

    //@siclog "Step 6" siclog@

    //Using the same TTI as the SecurityModeCommand message in step 5, the SS transmits an RRCConnectionReconfiguration message to establish a data radio bearer.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est(v_RRC_TI,

                                                                                  v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                  v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                  v_AntennaInfo)));

    //@siclog "Step 7" siclog@

    //Check: Does the UE transmit a SecurityModeComplete message?

    SRB.receive( car_SRB1_RrcPdu_IND(eutra_Cell1,

                                     cr_508_SecurityModeComplete(v_RRC_TI)));          /* SecurityModeComplete is not ciphered

                                                                                          acc. to 36.331 cl. 5.3.4.3 and 5.3.1.1 */

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.2.1.1 Step 7");

    //@siclog "Step 8" siclog@

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message to confirm the establishment of data radio bearer?

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.2.1.1 Step 8");

    //@siclog "Step 9" siclog@

    //Check: Does the test result of CALL generic procedure in TS 36.508 subclause 6.4.2.3 indicate that UE is in E-UTRA RRC_CONNECTED state?

    f_EUTRA_508Check_ConnectedState(eutra_Cell1);

    f_EUTRA_TestBody_Set(false);

    /* Postamble */

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED); //@sic R5-103875 sic@

  }


Change 8.3.5 - f_TC_8_2_1_8_EUTRA
	  function f_TC_8_2_1_8_EUTRA() runs on EUTRA_PTC

  { /* RRC connection reconfiguration / Radio bearer establishment / Success / Dedicated bearer / ROHC configured */

    var ProcedureTransactionIdentifier v_EpsProcedureTransactionId := '00'O;

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated;

    var template (value) DRX_Config v_DRX_Config; /* @sic R5s120506 sic@ */

    var template (value) MAC_MainConfig_Type v_MAC_MainConfig; /* @sic R5s120506 sic@ */

    var SubFrameTiming_Type v_SubFrameTiming; /* @sic R5s120506 sic@ */

    var template (value) TimingInfo_Type v_TimingInfo := cs_TimingInfo_Now; /* @sic R5s120506 sic@ */

    f_EUTRA_Init(c1);

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);    // @sic R5s150338: f_EUTRA_CellConfig_DRB_Common replaced by f_EUTRA_CellConfig_Def sic@
    /* Preamble to enter UE in state Generic RB Established (state 3) */

    f_EUTRA_Preamble( eutra_Cell1, STATE2_IDLEUPDATE );

    f_EUTRA_RbEst_Def( eutra_Cell1 );

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_OneDRB_ConfigUM(tsc_DRB2));    // @sic R5s150338: DRB2 is not pre-configured anymore sic@
    f_EUTRA_TestBody_Set( true );

    v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS; /* @sic R5s120506 sic@ */

    v_RadioResourceConfigDedicated := cs_RadioResourceConfig_DRB_RECONFIG(cs_DRB_ToAddModList_RECONFIG_8_2_1_8(tsc_DRB2),

                                                                          v_MAC_MainConfig ); /* @sic R5s120506 sic@ */

    /* @sic R5s120506 sic@ */

    if (ispresent(v_MAC_MainConfig.explicitValue.drx_Config) ) {

      v_DRX_Config := valueof(v_MAC_MainConfig.explicitValue.drx_Config);         /* valueof is ok as v_MAC_MainConfig.explicitValue.drx_Config is checked with 'ispresent' beforehand */

      f_EUTRA_SS_DrxCtrlConfig( eutra_Cell1, cs_DrxCtrl_Config( v_DRX_Config ) ); /* configure DRX at the SS */

      v_SubFrameTiming := f_EUTRA_GetNextSendOccasion( eutra_Cell1, 100 );

      v_TimingInfo := cs_TimingInfo( v_SubFrameTiming.SFN.Number, v_SubFrameTiming.Subframe.Number );

    }

    else {

      /* do nothing; for the time being we don't need any DRX configuration at the SS for DRX_L */

    }

    //@siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to establish a data radio bearer.

    SRB.send( cas_SRB1_RrcNasPdu_REQ( eutra_Cell1,

                                      v_TimingInfo, /* @sic R5s120506 sic@ */

                                      cs_RRCConnectionReconfiguration_Common( tsc_RRC_TI_Def,

                                                                              omit,

                                                                              omit,

                                                                              v_RadioResourceConfigDedicated,

                                                                              omit ),

                                      cs_NAS_Request( tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_ActDedicatedEPSBearerCxtReq( tsc_EpsDedicatedBearerId,

                                                                                          v_EpsProcedureTransactionId,

                                                                                          cs_LinkedId(hex2bit( tsc_EpsDefaultBearerId )),

                                                                                          cs_508_EPS_QoS_Dedicated_1, // According to Context#1 in TS 36.508

                                                                                          f_EUTRA_Get_508_TrafficFlowTemplate(1) ) ) ) ); // According to Context#1 in TS 36.508

    //@siclog "Step 2" siclog@

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1,

                                      cr_508_RRCConnectionReconfigurationComplete( tsc_RRC_TI_Def ) ) );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.2.1.8 Step 2");

    //@siclog "Step 3" siclog@

    //The UE transmits an ULInformationTransfer message. This message includes an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message.

    SRB.receive( car_SRB2_NasPdu_IND( eutra_Cell1,

                                      cr_NAS_Indication( tsc_SHT_IntegrityProtected_Ciphered,

                                                         cr_508_ActDedicatedEPSBearerCxtAccept( tsc_EpsDedicatedBearerId ) ) ) );

    //@siclog "Step 4" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.3 indicate that the UE is in E-UTRA RRC_CONNECTED state on Cell 1?

    f_EUTRA_508Check_ConnectedState( eutra_Cell1 );

    f_EUTRA_TestBody_Set( false );

    /* Postamble */

    f_EUTRA_Postamble( eutra_Cell1, E2_CONNECTED );

  }


Change 8.4 – Preconfiguration of DRB1 for cell 2 removed from PDCP test cases
	Function name
	f_TC_7_3_5_2_EUTRA, f_TC_7_3_5_4_EUTRA, f_TC_7_3_5_5_EUTRA

	Reason for change
	In case of handover DRB1 is configured by f_EUTRA_508RRC_IntraLTE_HO_InterCell in PDCP test cases

	Summary of changes
	explicit configuration of DRB1 removed

	TTCN module
	PDCP_Testcases


Change 8.4.1 - f_TC_7_3_5_2_EUTRA
	  function f_TC_7_3_5_2_EUTRA() runs on EUTRA_PTC

  { // @sic R5s100305 sic@ @sic R5-104794 sic@

    var PDCP_SS_State_Type v_PDCP_Rec;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_Info;

    var template (value) PhysicalConfigDedicated v_PhysicalConfigDedicated;

    // Initialization, generic setup, DRB loopback

    f_EUTRA_Init(c1);

    // Use 12 bits SN length

    fl_InitPDCP_Record(v_PDCP_Rec, AM_Mode, PDCP_SNLength12);

    // get FDD_TDD_Info

    v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -79);

    // Cell configuration, with one AM DRB at the SS side

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // Setup cell 2

    f_EUTRA_CellInfo_SetC_RNTI(eutra_Cell2, tsc_C_RNTI_Def3);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    //set null AS ciphering algorithm

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

    //entering UE test state 3, default values with BSR and PHR disabled, dsrTRansMax=n64   // @sic R5s110121 R5-110724 sic@

    v_PhysicalConfigDedicated := f_PhysicalConfigDedicated_Dsr_TransMax(eutra_Cell1);

    f_EUTRA_GenericRbEst_DisableBSR_PHR(eutra_Cell1, 0, 0, -, -, -, v_PhysicalConfigDedicated); // @sic R5s150338 sic@ 

    // reconfigure DRB locally to set the required PDCP test mode

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_DRB1_ConfigAM_PDCP_Mode);

    // closing the test loop (mode A), entering state 4

    f_EUTRA_CloseUE_TestLoopModeA(eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);



    f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

    f_EUTRA_7_3_5_2_TestBody(v_PDCP_Rec);

    f_EUTRA_TestBody_Set(false);

    // leaving UE test state 4

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    //postamble

    f_EUTRA_Postamble(eutra_Cell2, E3_TESTMODE);

  }


Change 8.4.2 - f_TC_7_3_5_4_EUTRA
	  function f_TC_7_3_5_4_EUTRA() runs on EUTRA_PTC

  { // @sic R5s100307 sic@  @sic R5-104794 sic@

    var PDCP_SS_State_Type v_PDCP_Rec;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var template (value) PhysicalConfigDedicated v_PhysicalConfigDedicated;

    // Initialization, generic setup, DRB loopback

    f_EUTRA_Init(c1);

    // Use 12 bits SN length

    fl_InitPDCP_Record(v_PDCP_Rec, AM_Mode, PDCP_SNLength12);

    // get FDD_TDD_Info

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -79);

    // Cell configuration, with one AM DRB at the SS side

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

    // Setup cell 2

    f_EUTRA_CellInfo_SetC_RNTI(eutra_Cell2, tsc_C_RNTI_Def3);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    //set null AS ciphering algorithm

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

    //entering UE test state 3, default values with BSR and PHR disabled, DsrTransMax=n64 // @sic R5-110724 sic@

    v_PhysicalConfigDedicated := f_PhysicalConfigDedicated_Dsr_TransMax(eutra_Cell1);

    f_EUTRA_GenericRbEst_DisableBSR_PHR(eutra_Cell1, 0, 0, -, -, -, v_PhysicalConfigDedicated); // @sic R5s150338 sic@ 

    // reconfigure DRB locally to set the required PDCP test mode

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_DRB1_ConfigAM_PDCP_Mode);

    // closing the test loop (mode A), entering state 4

    f_EUTRA_CloseUE_TestLoopModeA(eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);



    f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

    f_EUTRA_7_3_5_4_TestBody(v_PDCP_Rec);

    f_EUTRA_TestBody_Set(false);

    // leaving UE test state 4

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    //postamble

    f_EUTRA_Postamble(eutra_Cell2, E3_TESTMODE);

  }


Change 8.4.3 - f_TC_7_3_5_5_EUTRA
	  function f_TC_7_3_5_5_EUTRA() runs on EUTRA_PTC

  { // @sic R5s100309 sic@  @sic R5-104794 sic@

    var PDCP_SS_State_Type v_PDCP_Rec;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    // Initialization, generic setup, DRB loopback

    f_EUTRA_Init(c1);

    // Use 12 bits SN length

    fl_InitPDCP_Record(v_PDCP_Rec, AM_Mode, PDCP_SNLength12);

    // do we need a record per cell ?

    // get FDD_TDD_Info

    v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell1);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -79);

    // Cell configuration, with one AM DRB at the SS side

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // Setup cell 2

    f_EUTRA_CellInfo_SetC_RNTI(eutra_Cell2, tsc_C_RNTI_Def3);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    //set null AS ciphering algorithm

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

    //entering UE test state 3, default values with BSR and PHR disabled

    f_EUTRA_RbEst_Def_DisableBSR_PHR(eutra_Cell1);

    // reconfigure DRB locally to set the required PDCP test mode

    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_Cell1, cs_DRB1_ConfigAM_PDCP_Mode);

    // closing the test loop (mode A), entering state 4

    f_EUTRA_CloseUE_TestLoopModeA(eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);



    f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@

    f_EUTRA_7_3_5_5_TestBody(v_PDCP_Rec);

    f_EUTRA_TestBody_Set(false);

    // leaving UE test state 4

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    //postamble

    f_EUTRA_Postamble(eutra_Cell2, E3_TESTMODE);

  }


Change 9 –EUTRA IPCAN
	Function name
	f_EUTRA_IPCAN_Init , f_EUTRA_IPCAN_ConnectionRelease, f_EUTRA_IPCAN_MainLoop

	Reason for change
	1. No need for pre-configuration of DRBs in f_EUTRA_IPCAN_Init anymore
2. All DRBs are automatically released by f_EUTRA_RRC_ConnectionRelease (see change 6) ( no DRB configuration need to be considered in f_EUTRA_IPCAN_ConnectionRelease
3. p_TestConfiguration not needed for f_EUTRA_IPCAN_Init anymore
4. No need for pre-configuration of DRBs for emergency call in f_EUTRA_IPCAN_StartProcedure

	Summary of changes
	1. pre-configuration of DRBs removed from f_EUTRA_IPCAN_Init
2. DRB configuration removed from f_EUTRA_IPCAN_ConnectionRelease
3. no TestConfiguration handed over to f_EUTRA_IPCAN_Init in f_EUTRA_IPCAN_MainLoop
4. pre-configuration of DRBs removed from f_EUTRA_IPCAN_StartProcedure

	TTCN module
	IPCAN_EUTRA


Change 9.1 - f_EUTRA_IPCAN_Init
	  function f_EUTRA_IPCAN_Init(IMS_IPCAN_CO_ORD_PORT p_Port,

                              EUTRA_CellId_Type p_CellId
) runs on EUTRA_PTC

  { /* Open issues:

     * @sic R5-131807, R5-131132 and R5-131135 - new configuration for XCAP; additional change: parameter of IMS_TestConfiguration_Type instead of IMS_TestProcedure_Type sic@ */




    f_EUTRA_Init(c1); 
    f_EUTRA_CellConfig_Def(p_CellId, DISCARD_UM_DATA, USE_BIG_GRANTS);





















    f_IMS_IPCAN_SendCoOrdMsg(p_Port, cms_IPCAN_IMS_Response(IPCAN_INIT));

  }


Further change: p_TestConfiguration removed from f_UTRAN_IPCAN_Init (has not been used anyway)

Change 9.2 - f_EUTRA_IPCAN_ConnectionRelease
	  function f_EUTRA_IPCAN_ConnectionRelease(IMS_IPCAN_CO_ORD_PORT p_Port,

                                           EUTRA_CellId_Type p_CellId,

                                           IMS_TestProcedure_Type p_ProcedureType,

                                           boolean p_DedicatedBearersEstablished,

                                           boolean p_IsXCAP) runs on EUTRA_PTC

  { /* @sic R5-145787: new parameter p_IsXCAP sic@ */

    /* NOTE: p_IsXCAP is indepedent from p_ProcedureType */


    var EPS_BearerIdentity v_EpsBearerId;

    var boolean v_Keep2ndPDN := p_IsXCAP and pc_XCAP_over_Internet_APN;


    select (p_ProcedureType) {

      case (IPCAN_InitialRegistration, IPCAN_MO_IMS_Signalling, IPCAN_MT_IMS_Signalling) {

        // just release RRC connection

      }

      case (IPCAN_EmergencyCall_NormalService) {


        f_EUTRA_IPCAN_DeactivateEPSBearer(p_CellId, tsc_EpsDedicatedBearerId);  /* For emergency call the default bearer shall be released

                                                                                   (2nd default bearer in case of normal service, 1st in case of limited service)

                                                                                   => for normal service - as long as there is only signalling on PDN1 - we also release EPS bearer '6'H and DRB2 */

      }

      case (IPCAN_EmergencyCall_LimitedService) {

        // to be implemented

      }

      case (IPCAN_MO_SpeechCall, IPCAN_MT_SpeechCall) {

        if (p_DedicatedBearersEstablished) {       // NOTE: in test cases for unsuccessful behaviour the EPS bearer may not be implemented

          f_EUTRA_IPCAN_DeactivateEPSBearer(p_CellId);

        }

      }

      case (IPCAN_MO_VideoCall, IPCAN_MT_VideoCall) {

        // release dedicated bearers


        f_EUTRA_IPCAN_DeactivateEPSBearer(p_CellId, tsc_EpsDedicatedBearerId);

        f_EUTRA_IPCAN_DeactivateEPSBearer(p_CellId, tsc_EpsDedicatedBearerId2);

      }

      case (IPCAN_XCAP_Signalling) {                          /* @sic R5s141051 change 7: release the 2ndPDN sic@ */

        v_EpsBearerId := tsc_EpsDefaultBearerId2ndPDN;                                                  // @sic R5-145787 sic@

        f_EUTRA_IPCAN_DeactivateEPSBearer(p_CellId, v_EpsBearerId);  /* For XCAP TCs, 2nd PDN needs to be released */

      }

      case (IPCAN_MO_Speech_EmergencyCall) {                  /* @sic R5s141286 change 7 sic@ */



        f_EUTRA_IPCAN_DeactivateEPSBearer(p_CellId, tsc_EpsDedicatedBearerId3);   /* release dedicated EPS bearer associated with tsc_DRB4 */

        f_EUTRA_IPCAN_DeactivateEPSBearer(p_CellId, tsc_EpsDedicatedBearerId);    /* release dedicated EPS bearer associated with tsc_DRB2 */

      }

    }


    f_EUTRA_RRC_ConnectionRelease(p_CellId);
    if (p_ProcedureType==IPCAN_InitialRegistration and tsc_EUTRA_Registration_MultiplePDN) {  // @sic R5-144797 sic@

      if (not v_Keep2ndPDN) {                                                                 // @sic R5-145787 sic@

        f_EUTRA_IdleUpdated_ReleaseSecondPDN(p_CellId);

      }

    }

    f_IMS_IPCAN_SendCoOrdMsg(p_Port);

  }


Change 9.3 - f_EUTRA_IPCAN_MainLoop
	  function f_EUTRA_IPCAN_MainLoop() runs on EUTRA_PTC

  {

    var EUTRA_CellId_Type v_CellId := tsc_IMS_EUTRA_CellId;

    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;

    var IMS_IPCAN_Command_Type v_Command;

    var integer v_PdnIndex;

    var boolean v_DedicatedBearersEstablished := false;  // will be set to true e.g. after successful call establishment (note: there are test cases with unsuccessful call establishment)

    var boolean v_ConnectionEstablished := false;      /* @sic R5s140490 change 7 MCC160 implementation sic@ */

    var boolean v_Done := false;

    var integer v_NextAvailableEpsBearerId := hex2int(tsc_EpsDedicatedBearerId);  // the next EPS bearer available is to be used as default for IMS Normal Emg Call

    var boolean v_IsXCAP := false;

    while (not v_Done) {

      alt {

        [] IMS[tsc_Index_PDN1].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_PDN1;

        }

        [] IMS[tsc_Index_PDN2].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_PDN2;

        }

      }

      v_Command := v_IMS_IPCAN_Coordination_MSG.IMS_IPCAN_Command;

      select (v_Command.Name) {

        case (IPCAN_INIT) {

          f_EUTRA_IPCAN_Init(IMS[v_PdnIndex], v_CellId);

          v_IsXCAP := (v_Command.TestConfiguration == IPCAN_XCAP);    /* @sic R5-145787: we need to know when it is an XCAP case sic@ */

        }

        case (IPCAN_CONFIG) {

          f_EUTRA_IPCAN_ModifyConfig(IMS[v_PdnIndex], v_CellId, v_Command.CellConfiguration);

        }

        case (IPCAN_STARTPROCEDURE) {

          if (f_EUTRA_IPCAN_StartProcedure(IMS[v_PdnIndex],

                                           v_CellId,

                                           v_Command.TestProcedure,

                                           v_NextAvailableEpsBearerId,

                                           v_ConnectionEstablished) == NORMAL) {

            v_DedicatedBearersEstablished := true;

            select (v_Command.TestProcedure) {                       /* @sic R5s140490 change 9 MCC160 implementation: v_NextAvailableEpsBearerId to be incremented only when procedure has not been aborted sic@ */

              case (IPCAN_MO_SpeechCall, IPCAN_MT_SpeechCall, IPCAN_EmergencyCall_LimitedService) {

                v_NextAvailableEpsBearerId := v_NextAvailableEpsBearerId + 1;

              }

              case (IPCAN_MO_VideoCall, IPCAN_MO_VideoCall, IPCAN_EmergencyCall_NormalService) {

                v_NextAvailableEpsBearerId := v_NextAvailableEpsBearerId + 2;

              }

              case else {

                // no new bearer added

              }

            }

          }

          v_ConnectionEstablished := true;      /* @sic R5s140490 change 7 MCC160 implementation sic@ */

        }

        case (IPCAN_ENDPROCEDURE) {

          f_EUTRA_IPCAN_ConnectionRelease(IMS[v_PdnIndex], v_CellId, v_Command.TestProcedure, v_DedicatedBearersEstablished, v_IsXCAP);

          /* => RRC connection is released */

          v_DedicatedBearersEstablished := false;

          v_ConnectionEstablished := false;     /* @sic R5s140490 change 7 MCC160 implementation sic@ */

        }

        case (IPCAN_RELEASE) {

          f_EUTRA_IPCAN_Release(IMS[v_PdnIndex], v_CellId);

          v_Done := true;

        }

        case (IPCAN_QUERY) {  /* @sic R5s130133 additional changes: handling of queries from the IMS PTC sic@ */

          IMS[v_PdnIndex].send(cms_IPCAN_IMS_QueryResponse(f_EUTRA_IMS_IpcanInfo()));

        }

        case else {

          FatalError(__FILE__, __LINE__, "invalid command");

        }

      }

    }

  }


Change 9.4 - f_EUTRA_IPCAN_StartProcedure
	  function f_EUTRA_IPCAN_StartProcedure(IMS_IPCAN_CO_ORD_PORT p_Port,

                                        EUTRA_CellId_Type p_CellId,

                                        IMS_TestProcedure_Type p_ProcedureType,

                                        integer p_NextAvailableEpsBearerId,

                                        boolean p_ConnectionEstablished) runs on EUTRA_PTC return TriggerResult_Type

  { /* @sic R5s130681: DISCARD_UM_DATA sic@ */

    /* @sic R5s140490 change 7 MCC160 implementation: p_ConnectionEstablished sic@ */

    var TriggerResult_Type v_Result := NORMAL;

    var EPS_BearerIdentity v_EpsBearerId := int2hex(p_NextAvailableEpsBearerId, 1);

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get(); // @sic R5s150091 sic@

    select (p_ProcedureType) {

      case (IPCAN_InitialRegistration) {                                        // initial registration

        f_EUTRA_IPCAN_InitialRegistration(p_Port, p_CellId);

      }

      case (IPCAN_EmergencyCall_NormalService) {                                // emergency call, normal service, uses 112 as default emg number

        f_EUTRA_IPCAN_EmergencyCall(p_CellId, v_EpsBearerId, p_ConnectionEstablished);

      }

      case (IPCAN_MO_Speech_EmergencyCall) {                                    /* @sic R5s141286 change 6 sic@

                                                                                   to add an emergency call even when there is a speech call already (been put on hold);

                                                                                   NOTE: In case there is a speech call already (put on hold, e.g. 19.1.5)

                                                                                   the default EPS bearer (v_EpsBearerId) is '7'H whereas it is '6'H in normal cases */








        f_EUTRA_IPCAN_EmergencyCall(p_CellId, v_EpsBearerId, p_ConnectionEstablished);

      }

      case (IPCAN_EmergencyCall_LimitedService) {                               // emergency call, limited service

        // to be implemented

      }

      case (IPCAN_MO_SpeechCall) {                                              // MO speech call

        v_Result := f_EUTRA_IPCAN_MO_SpeechCall(p_Port, p_CellId);

      }

      case (IPCAN_MT_SpeechCall) {                                              // MT speech call

        v_Result := f_EUTRA_IPCAN_MT_SpeechCall(p_Port, p_CellId, v_EpsBearerId);

      }

      case (IPCAN_MO_VideoCall) {                                               // MO video call

        v_Result := f_EUTRA_IPCAN_MO_VideoCall(p_Port, p_CellId);

      }

      case (IPCAN_MT_VideoCall) {                                               // MT video call

        v_Result := f_EUTRA_IPCAN_MT_VideoCall(p_Port, p_CellId);

      }

      case (IPCAN_MO_AddVideo, IPCAN_MT_AddVideo) {                             // add video to existing speech call

        f_EUTRA_IPCAN_AddVideo(p_Port, p_CellId, tsc_EpsDedicatedBearerId2);

      }

      case (IPCAN_ReleaseVideo) {                                               // Release video but keep speech call

        f_EUTRA_IPCAN_ReleaseVideo(p_Port, p_CellId, tsc_EpsDedicatedBearerId2);

      }

      case (IPCAN_MO_IMS_Signalling) {                                          // MO IMS signalling

        f_EUTRA_IPCAN_MO_IMS_Signalling(p_CellId);

      }

      case (IPCAN_MT_IMS_Signalling) {                                          // MT IMS signalling

        f_EUTRA_IPCAN_MT_IMS_Signalling(p_Port, p_CellId);                      /* @sic R5s130183 change 2 sic@ */

      }

      case (IPCAN_XCAP_Signalling) {                                            // XCAP signalling

        v_EpsBearerId := tsc_EpsDefaultBearerId2ndPDN;

        f_EUTRA_IPCAN_XCAP_Signalling(p_Port, p_CellId, v_EpsBearerId);         /* @sic R5s141051 change 6: v_EpsBearerId sic@ */

      }

    }

    return v_Result;

  }


Change 10 – Explicit DRB release 

	Function name
	f_EUTRA_DeactivateEPS_BearerContext, f_EUTRA_RRCConnectionReconfiguration_DRB_Release, f_TC_13_1_1_EUTRA, f_TC_8_2_3_1_EUTRA, fl_DeactivateEPSBearer, fl_TC_10_4_1_Body, fl_TC_10_4_2_Body, fl_TC_10_8_4_Body

	Reason for change
	When a default or dedicated EPS bearer is relased the associated DRB at the UE is released by an RRCConnectionReconfiguration with drb_ToReleaseList containing the DRB. For dynamic DRB configuration this DRB needs to be released at the SS too

	Summary of changes
	1. Release of DRB wherever cs_RRCConnectionReconfiguration_DRB_Release is used
2. f_EUTRA_SS_Release_SingleDRB removed from f_EUTRA_IdleUpdated_ReleaseSecondPDN

	TTCN module
	EUTRA_InitialRegistration, EUTRA_RRCSteps, MultiLayer_Procedures, RRC_ConnReconfig, NAS_TrackingArea_UG, ESM_Testcases, 


Change 10.1 - f_EUTRA_DeactivateEPS_BearerContext
	  function f_EUTRA_DeactivateEPS_BearerContext(EUTRA_CellId_Type p_CellId,

                                               EPS_BearerIdentity p_EpsDefBearerId := tsc_EpsDefaultBearerId,

                                               ProcedureTransactionIdentifier p_PTId_UE := tsc_PTI_Unassigned,

                                               B8_Type p_EsmCauseValue := tsc_ESM_Cause36_RegularDeactivation,

                                               template (omit) GPRS_Timer3 p_T3396 := omit)

    runs on EUTRA_PTC

  { // @sic R5-140747 sic@

    /* @sic R5-144797: new parameters p_EsmCauseValue, p_T3396 sic@ */

    var EPS_BearerIdentity v_EpsBearerId := p_EpsDefBearerId;

    var DRB_Identity v_DrbId := f_EUTRA_EpsBearerAssociatedDRB(v_EpsBearerId);

    var template (value) ESM_Cause v_EsmCause := cs_ESM_Cause_v(p_EsmCauseValue);

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST.

    SRB.send(cas_SRB1_RrcNasPdu_REQ(p_CellId,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release(tsc_RRC_TI_Def, v_DrbId),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(v_EpsBearerId,

                                                                                    p_PTId_UE,

                                                                                    v_EsmCause,

                                                                                    p_T3396))));

    interleave { /* @sic R5-145774 sic@ */

      // receive RRCConnectionReconfigurationComplete

      [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId,

                                         cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def))) {}

      // The UE transmits a DEACTIVATE EPS BEARER CONTEXT ACCEPT message.

      [] SRB.receive(car_SRB2_NasPdu_IND(p_CellId,   // @sic R5s140579 sic@

                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                           cr_508_DeactivateEPSBearerCxtAccept(v_EpsBearerId)))) {}

    } 
    f_EUTRA_SS_Release_SingleDRB(p_CellId, v_DrbId);
  }


Change 10.2 - f_EUTRA_RRCConnectionReconfiguration_DRB_Release
	  function f_EUTRA_RRCConnectionReconfiguration_DRB_Release(EUTRA_CellId_Type p_CellId,

                                                            RRC_TransactionIdentifier p_RRC_TI,

                                                            DRB_Identity p_DRB_Id)

    runs on EUTRA_PTC

  {

    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo_Now, cs_RRCConnectionReconfiguration_DRB_Release(tsc_RRC_TI_Def, p_DRB_Id)));

    SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_508_RRCConnectionReconfigurationComplete(p_RRC_TI)));
    f_EUTRA_SS_Release_SingleDRB(p_CellId, p_DRB_Id);
  }


Change 10.3 - f_TC_13_1_1_EUTRA
	  function f_TC_13_1_1_EUTRA() runs on EUTRA_PTC

  { /* Activation and deactivation of additional data radio bearer in E-UTRA */

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template(value) NAS_MSG_Request_Type v_NAS_MSG_Request;

    var template (value) ESM_Cause v_EsmCause36 := cs_ESM_Cause_v ('00100100'B); // "Regular deactivation"  // @sic R5-101118: ESM_Cause sic@

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var EPS_BearerIdentity v_DedicatedEpsBeareId := tsc_EpsDedicatedBearerId;

    var EPS_BearerIdentity v_LinkedEpsBearerId := tsc_EpsDefaultBearerId;

    var UInt16_Type v_SourcePort := 31160; // @sic R5s100068 sic@ // @sic R5s100135 sic@

    var UInt16_Type v_DestPort := 61000; // @sic R5s100068 sic@ // @sic R5s100135 sic@

    var PDCP_SDUList_Type v_PDCP_SDUList_DRB1 := { f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE()) };      /* @sic R5s120470 sic@ */

    var octetstring v_AnyUdpPacket;

    var charstring v_IP_Address1_NW;

    var B3_Type v_PdnType;

    var boolean v_UseIPv4;

    f_EUTRA_Init(c1);

    v_NAS_MSG_Request := cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                        cs_508_DeactivateEPSBearerCxtReq(v_DedicatedEpsBeareId,

                                                                         v_EPS_TI,

                                                                         v_EsmCause36)); // @sic R5-101118: ESM_Cause sic@

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);  /* @sic R5s150338: f_EUTRA_CellConfig_DRB_Common replaced by f_EUTRA_CellConfig_Def sic@ */

    /* Preamble The UE is in state Generic RB Established, UE test mode activated (state 3A) according to [18] using the UE TEST LOOP MODE B. */

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeB);

    f_EUTRA_RbEst_Def(eutra_Cell1);

    f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-101050 sic@

    //@siclog "Step 1 - 2" siclog@

    /* Step 1: SS transmits an RRCConnectionReconfiguration message to establish a data radio bearer.

     *         The SS transmits a ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message activating a new EPS bearer context.

     *         The NAS message is included in a RRCConnectionReconfiguration message.

     * Step 2: The UE transmit a RRCConnectionReconfigurationComplete message to confirm the establishment of data radio bearer. */

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_Cell1,

                                        tsc_DRB2,

                                        v_DedicatedEpsBeareId,

                                        v_EPS_TI,

                                        cs_LinkedId(hex2bit(v_LinkedEpsBearerId)),

                                        cs_508_EPS_QoS_Dedicated_1, // According to Context#1 in TS 36.508

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1)); // According to Context#1 in TS 36.508

    //@siclog "Step 3" siclog@

    /* The UE transmits an ACTIVATE DEDICATE EPS BEARER CONTEXT ACCEPT message. */

    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_DedicatedEpsBeareId))));

    //@siclog "Step 4" siclog@

    /* The SS closes the UE test loop mode. */

    f_EUTRA_CloseUE_TestLoopModeB(eutra_Cell1, '00'O);

    //@siclog "Step 5" siclog@

    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context. */

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1, tsc_RbId_DRB1, cs_TimingInfo_Now, v_PDCP_SDUList_DRB1));     /* @sic R5s120470 sic@ */

    //@siclog "Step 6" siclog@

    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    DRB.receive(car_DRB_COMMON_IND_PDCP_SDUList(eutra_Cell1, tsc_RbId_DRB1, ?, v_PDCP_SDUList_DRB1));              /* @sic R5s120470 sic@ */

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 6");

    //Check PDN type and select IP version accordingly

    v_PdnType := f_EUTRA_MobileInfo_GetAssignedPdnType();

    v_UseIPv4 := (v_PdnType == tsc_PdnType_IPv4);  /* @sic R5-112600: clarification to conditions for IP address configuration sic@ */

    v_IP_Address1_NW := f_LoopbackModeB_IP_Address_NW(v_UseIPv4);      /* @sic R5s120470 sic@ */

    v_AnyUdpPacket := f_IPv4IPv6_AnyUdpPacket(v_IP_Address1_NW, v_IP_Address1_NW, v_SourcePort, v_DestPort);         /* @sic R5s120470 sic@ */

    //@siclog "Step 7" siclog@

    /* The SS transmits one IP packet to the UE on the DRB associated with the dedicated EPS bearer context. */

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,

                                         tsc_RbId_DRB2,

                                         cs_TimingInfo_Now,

                                         { v_AnyUdpPacket })); // @sic R5s100135 sic@

    //@siclog "Step 8" siclog@

    /* Check: Does the UE loop back the IP packet on the DRB associated with the dedicated EPS bearer context? */

    DRB.receive(car_DRB_COMMON_IND_PDCP_SDUList(eutra_Cell1,

                                                tsc_RbId_DRB2,

                                                ?,

                                                { v_AnyUdpPacket })); // @sic R5s100135 sic@

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 8");

    //@siclog "Step 9" siclog@

    /* SS transmits an RRCConnectionReconfiguration message to release a data radio bearer.

     * The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the dedicated EPS bearer.

     * The NAS message is included in a RRCConnectionReconfiguration message. */

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_Cell1,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release(v_RRC_TI, tsc_DRB2),

                                    v_NAS_MSG_Request));

    //@siclog "Step 10" siclog@

    /* The UE transmits an RRCConnectionReconfigurationComplete message. */

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //@siclog "Step 11" siclog@

    /* The UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT. */

    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_DeactivateEPSBearerCxtAccept(v_DedicatedEpsBeareId))));

    f_EUTRA_SS_Release_SingleDRB(eutra_Cell1, tsc_DRB2);
    //@siclog "Step 12" siclog@

    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context. */

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1, tsc_RbId_DRB1, cs_TimingInfo_Now, v_PDCP_SDUList_DRB1));     /* @sic R5s120470 sic@ */

    //@siclog "Step 13" siclog@

    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    DRB.receive(car_DRB_COMMON_IND_PDCP_SDUList(eutra_Cell1, tsc_RbId_DRB1, ?, v_PDCP_SDUList_DRB1));              /* @sic R5s120470 sic@ */

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 13");

    f_EUTRA_TestBody_Set(false);

    /* Postamble */

    f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE); //@sic R5-103875 sic@

  }


Change 10.4 - f_TC_8_2_3_1_EUTRA
	  function f_TC_8_2_3_1_EUTRA() runs on EUTRA_PTC

  { /* RRC Connection Reconfiguration / Radio Bearer Release: Success */

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var EPS_BearerIdentity v_DedicatedEpsBearerId := tsc_EpsDedicatedBearerId;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template(value) NAS_MSG_Request_Type v_NAS_MSG_Request;

    var template (value) ESM_Cause v_EsmCause36 := cs_ESM_Cause_v ('00100100'B); // "Regular deactivation"  // @sic R5-100492: ESM_Cause sic@

    f_EUTRA_Init(c1);

    v_NAS_MSG_Request := cs_NAS_Request( tsc_SHT_IntegrityProtected_Ciphered,

                                         cs_508_DeactivateEPSBearerCxtReq( v_DedicatedEpsBearerId,

                                                                           v_EPS_TI,

                                                                           v_EsmCause36)); //@sic R5-100492: ESM_Cause sic@

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);    // @sic R5s150338: f_EUTRA_CellConfig_DRB_Common replaced by f_EUTRA_CellConfig_Def sic@

    /* Preamble to enter UE in state Generic RB Established (state 3)

       Note: Also 1 AM DRB is established here compared to default state 3 */

    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_RbEst_508_AM(eutra_Cell1); // @sic R5-100492: UM bearer changed to AM sic@

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionReconfiguration message with a drb-ToReleaseList

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_Cell1,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release(v_RRC_TI, tsc_DRB2),

                                    v_NAS_MSG_Request));

    //@siclog "Step 2" siclog@

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.2.3.1 Step 2");

    //@siclog "Step 2a" siclog@

    //The UE transmits an ULInformationTransfer message.

    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_DeactivateEPSBearerCxtAccept(v_DedicatedEpsBearerId)))); 
    f_EUTRA_SS_Release_SingleDRB(eutra_Cell1, tsc_DRB2);
    //@siclog "Step 3" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.3 indicate that the UE is in E-UTRA RRC_CONNECTED state on Cell 1?

    f_EUTRA_508Check_ConnectedState(eutra_Cell1);

    f_EUTRA_TestBody_Set(false);

    /* Postamble */

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED); //@sic R5-103875 sic@

  }


Change 10.5 - fl_DeactivateEPSBearer
	  function fl_DeactivateEPSBearer(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  {

    const ProcedureTransactionIdentifier tsc_Eps_Ti := '00'O;

    var template (value) ESM_Cause v_EsmCause43 := cs_ESM_Cause_v('00100100'B); // #36 regular deactivation

    // The SS sends DEACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST

    // [email discussion wit Lena@Qualcomm] Message does not exist use DEACTIVATE EPS BEARER CONTEXT REQUEST

    SRB.send(cas_SRB1_RrcNasPdu_REQ(p_CellId,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release(tsc_RRC_TI_Def, tsc_DRB2),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(tsc_EpsDedicatedBearerId,

                                                                                    tsc_Eps_Ti,

                                                                                    v_EsmCause43))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_DeactivateEPSBearerCxtAccept(tsc_EpsDedicatedBearerId)))); 
    f_EUTRA_SS_Release_SingleDRB(p_CellId, tsc_DRB2);
  }


Change 10.6 - fl_TC_10_4_1_Body
	  function fl_TC_10_4_1_Body() runs on EUTRA_PTC

  {

    var EPS_BearerIdentity v_EpsBearerId2 := '6'H; // default bearer on additional PDN

    var EPS_BearerIdentity v_EpsDedicatedBearerId2 := '7'H; // dedicated bearer id on additional PDN

    var ProcedureTransactionIdentifier v_EpsTi2 := tsc_PTI_Unassigned;

    var ProcedureTransactionIdentifier v_PTId_UE;

    var template (omit) ProtocolConfigOptions v_Pco;  // @sic R5s100761 sic@

    var template (value) AccessPointName v_AccessPointName;

    var PDN_Address v_PDN_Address2;

    var NasCount_Type v_NasCountUL;

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var NAS_MSG_Indication_Type v_NasInd;

    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;

    var template (value) ESM_Cause v_EsmCause43 := cs_ESM_Cause_v ('00101011'B); // Invalid EPS bearer identity

    var template (value) ESM_Cause v_EsmCause36 := cs_ESM_Cause_v ('00100100'B); // Regular deactivation

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var GutiParameters_Type v_GutiParams;

    var NAS_PlmnId v_PLMN;

    var TrackingAreaCode v_Tac;

    var SRB_COMMON_IND v_ReceivedAsp;

    var template (value) TrackingAreaIdList v_TaiList;

    var template (value) CellPowerList_Type v_CellPowerList;

    var M_TMSI_Type v_M_Tmsi_Def := tsc_M_TMSI1;

    var GutiParameters_Type v_GutiParameters := f_EUTRA_CellInfo_GetGuti ( eutra_CellA );

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info ( eutra_CellA);

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA);

    var octetstring v_APN_Value := tsc_APN1; // APN-1(New PDN name), different from default

    var template (omit) LocAreaId v_LAI := omit;

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(NORMAL); // @sic R5s110176 sic@

    var template (value) RadioBearerList_Type v_DrbConfigList := {cs_OneDRB_ConfigAM(tsc_DRB1), cs_OneDRB_ConfigAM(tsc_DRB2), cs_OneDRB_ConfigAM(tsc_DRB3)};  // @sic R5s130077 sic@

    timer t_Wait := 1.0;

    // Set the cell type of cell A to the ''Serving cell''.

    // Set the cell type of cell B to the ''Suitable cell''.

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_ServingCellRS_EPRE  ),

      cs_CellPower ( eutra_CellB, tsc_SuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList ); // @sic R5s110498 sic@

    // Cause the UE to request connectivity to an additional PDN.

    //@siclog "Step 1" siclog@

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_string); // @sic R5s100466 sic@

    // The UE transmits a SERVICE REQUEST

    //@siclog "Step 1A" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    // The SS establishes SRB2 and DRB associated with default EPS bearer context obtained during the attach procedure.

    //@siclog "Step 1B" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The UE transmits a PDN CONNECTIVITY REQUEST message as specified to request an additional PDN.

    //@siclog "Step 2" siclog@

    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_CellA,

                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

    v_AccessPointName := cs_AccessPointName(v_APN_Value);

    v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    // Change from DRB1 to DRB2 to handle U Plane data

    f_EUTRA_StartIPfor2ndPDN(eutra_CellA);  // @sic R5s110449, R5-113734 change 8 sic@

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message with IE EPS Bearer Identity set to new EPS bearer context.

    // Note: the SS allocates a PDN address of a PDN type which is compliant with the PDN type requested by the UE.

    //@siclog "Step 3" siclog@

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), PDN_2); // @sic R5s110011 R5-120234 sic@

    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefaultProtocolConfigOptions(v_Pco));

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 4" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling

    f_EUTRA_DelayForUserPlaneSignalling();

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST relative to the additional PDN.

    // Note: the same PDN address is applicable because the linked EPS bearer ID refers to the default EBC allocated in step 3.

    //@siclog "Step 5" siclog@

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,

                                        v_EpsDedicatedBearerId2,

                                        v_EpsTi2,

                                        cs_LinkedId(hex2bit( v_EpsBearerId2)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2));

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 6" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId2))));

    //  The SS releases the RRC connection.

    //@siclog "Step 7" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);  // @sic R5s130077 sic@  @sic R5s150338: no v_DrbReleaseList anymore sic@

    //  The SS transmits a Paging message to the UE using S-TMSI with CN domain indicator set to "PS".

    //@siclog "Step 8" siclog@

    f_EUTRA_UE_Page(eutra_CellA, cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParameters.MME_Code, v_M_Tmsi_Def)));

    //  The UE transmits the SERVICE REQUEST message.

    //@siclog "Step 9" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));

    // The SS establishes SRB2 and DRBs associated with two default EPS bearer context (a first PDN obtained during the attach procedure and an additional PDN).

    //@siclog "Step 9A" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL, v_DrbConfigList);   /* @sic R5s150338 change 3.3.2: v_DrbConfigList sic@ */

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2({cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1),

                                                                                                                  cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB2), // @sic R5s110365 sic@

                                                                                                                  cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB3)},

                                                                                                                 cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                 cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                            v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                            v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the dedicated EPS bearer to the additional PDN.

    //@siclog "Step 10" siclog@

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release (tsc_RRC_TI_Def, tsc_DRB3), // @sic R5s110275 sic@

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(v_EpsDedicatedBearerId2,

                                                                                    v_EpsTi2,

                                                                                    v_EsmCause36))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    //@siclog "Step 11" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_DeactivateEPSBearerCxtAccept(v_EpsDedicatedBearerId2,

                                                                                      v_EpsTi2)))); 
    f_EUTRA_SS_Release_SingleDRB(eutra_CellA, tsc_DRB3);   /* @sic R5s150338 sic@ */    
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST relative to additional PDN.

    // Note: the same PDN address is applicable because the linked EPS bearer ID refers to the default EBC allocated in step 3.

    //@siclog "Step 12" siclog@

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,

                                        v_EpsDedicatedBearerId2,

                                        v_EpsTi2,

                                        cs_LinkedId(hex2bit( v_EpsBearerId2)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2));

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 13" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId2))));

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the dedicated EPS bearer to the additional PDN.

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    //@siclog "Step 14-15" siclog@

    f_EUTRA_DeactivateEPS_BearerContext(eutra_CellA, v_EpsBearerId2);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 15");

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the dedicated EPS bearer.

    // This message is included in a DLInformationTransfer message.

    //@siclog "Step 16" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsDedicatedBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             f_EUTRA_Get_508_TrafficFlowTemplate_2_Modify(v_EpsDedicatedBearerId2))))); // @sic R5-103625 sic@

    // Check: Does UE transmits a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43?

    //@siclog "Step 17" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsDedicatedBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST that does not point an existing EPS bearer context.

    //@siclog "Step 18" siclog@

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release (tsc_RRC_TI_Def, tsc_DRB2),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(v_EpsBearerId2,

                                                                                    v_EpsTi2,

                                                                                    v_EsmCause36))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    //@siclog "Step 19" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_DeactivateEPSBearerCxtAccept(v_EpsBearerId2,

                                                                                      v_EpsTi2)))); 
    f_EUTRA_SS_Release_SingleDRB(eutra_CellA, tsc_DRB2);   /* @sic R5s150338 sic@ */    
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 19");

    //  The SS releases the RRC connection.

    //@siclog "Step 20" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);  // @sic R5s130077 sic@   @sic R5s150338: no v_DrbReleaseList anymore sic@

    //  Cause the UE to request connectivity to an additional PDN.

    //@siclog "Step 21" siclog@

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_string); // APN-1(New PDN name), different from default @sic R5s100466 sic@

    // The UE transmits a SERVICE REQUEST.

    //@siclog "Step 21A" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    // The SS establishes SRB2 and DRBs associated with two default EPS bearer context (a first PDN obtained during the attach procedure and an additional PDN).

    //@siclog "Step 21B" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The UE transmits a PDN CONNECTIVITY REQUEST message as specified to request an additional PDN.

    //@siclog "Step 22" siclog@

    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_CellA,

                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

    v_AccessPointName := cs_AccessPointName(v_APN_Value);

    v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message with IE EPS Bearer Identity set to new EPS bearer context.

    // Note: the SS allocates a PDN address of a PDN type which is compliant with the PDN type requested by the UE.

    //@siclog "Step 23" siclog@

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), PDN_2); // @sic R5s110011 R5-120234 sic@

    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefaultProtocolConfigOptions(v_Pco));

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 24" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling

    f_EUTRA_DelayForUserPlaneSignalling();

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST relative to additional PDN.

    // Note: the same PDN address is applicable because the linked EPS bearer ID refers to the default EBC allocated in step 3.

    //@siclog "Step 25" siclog@

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,

                                        v_EpsDedicatedBearerId2,

                                        v_EpsTi2,

                                        cs_LinkedId(hex2bit( v_EpsBearerId2)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2));

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 26" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId2))));

    // The SS releases the RRC connection.

    //@siclog "Step 27" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);  // @sic R5s130077 sic@   @sic R5s150338: no v_DrbReleaseList anymore sic@

    // The SS transmits a Paging message to the UE using S-TMSI with CN domain indicator set to "PS".

    //@siclog "Step 28" siclog@

    f_EUTRA_UE_Page ( eutra_CellA,

                      cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_GutiParameters.MME_Code, v_M_Tmsi_Def )) );

    // The UE transmits SERVICE REQUEST message.

    //@siclog "Step 29" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_508_SERVICE_REQUEST(v_SecurityParams.KSIasme)));

    // activate security in AS

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    // The SS performs a radio bearer establishment procedure.

    // The RRCConnectionReconfiguration message doesn't include the EPS bearer ID of the additional PDN.

    //@siclog "Step 30" siclog@

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    // Check: Does UE transmit a RRCConnectionReconfigurationComplete message?

    //@siclog "Step 31" siclog@

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_CellA, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 31");

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the default EPS bearer to the additional PDN.

    // This message is included in a DLInformationTransfer message.

    //@siclog "Step 32" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             omit))));

    // Check: Does UE transmit a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43?

    //@siclog "Step 32A" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 32A");

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the dedicated EPS bearer.

    //@siclog "Step 32B" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsDedicatedBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             f_EUTRA_Get_508_TrafficFlowTemplate_2_Modify(v_EpsDedicatedBearerId2))))); // @sic R5-103625 sic@

    // Check: Does UE transmit a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43? (see Note 4)

    //@siclog "Step 32C" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsDedicatedBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 32C");

    // The SS releases the RRC connection.

    //@siclog "Step 33" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);  // @sic R5s130077 sic@   @sic R5s150338: no v_DrbReleaseList anymore sic@

    //  Cause the UE to request connectivity to an additional PDN.

    //@siclog "Step 34" siclog@

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_string); // APN-1(New PDN name), different from default @sic R5s100466 sic@

    // The UE transmits a SERVICE REQUEST

    //@siclog "Step 34A" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cr_508_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get())));

    // The SS establishes SRB2 and DRB associated with default EPS bearer context obtained during the attach procedure

    //@siclog "Step 34B" siclog@

    // activate security in AS with parameters from AKA performed in preamble

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // The UE transmits a PDN CONNECTIVITY REQUEST message as specified to request an additional PDN.

    //@siclog "Step 35" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_PDNConnReqAdditionalPDN(f_GetPdnType(), cr_AccessPointName(v_APN_Value)))))-> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

    v_AccessPointName := cs_AccessPointName(v_APN_Value);

    v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message

    // with IE EPS Bearer Identity set to new EPS bearer context.

    //@siclog "Step 36" siclog@

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), PDN_2); // @sic R5s110011 R5-120234 sic@

    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefaultProtocolConfigOptions(v_Pco));

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 37" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling

    f_EUTRA_DelayForUserPlaneSignalling();

    //  The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST relative to additional PDN.

    //@siclog "Step 38" siclog@

    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_CellA,

                                        tsc_DRB3,

                                        v_EpsDedicatedBearerId2,

                                        v_EpsTi2,

                                        cs_LinkedId(hex2bit( v_EpsBearerId2)),

                                        cs_508_EPS_QoS_Dedicated_1,

                                        f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2));

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 39" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId2))));

    // The SS releases the RRC connection.

    //@siclog "Step 40" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);  // @sic R5s130077 sic@   @sic R5s150338: no v_DrbReleaseList anymore sic@

    //  Cell A is a suitable cell, Cell B is the serving cell

    //@siclog "Step 41" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_SuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE  )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    // The UE transmit a TRACKING AREA UPDATE REQUEST message as specified on Cell B.

    //@siclog "Step 42" siclog@

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,

                                               tsc_RRC_TI_Def,

                                               ?,

                                               cr_NAS_Indication (tsc_SHT_IntegrityProtected,

                                                                  cdr_TAU_Request_EPSBearerCtxtStatus( v_UpdateType,

                                                                                                       f_EUTRA_SecurityKSIasme_Get (),

                                                                                                       cr_DRXparameter_Any ('5C'O) ifpresent, // @sic R5s120210, R5-123117 sic@

                                                                                                       v_AdditionalUpdateType, // @sic R5-103681 sic@

                                                                                                       cs_EPSBearerCtxtStatus_3Ctxts)));

    // The SS establishes SRB2 and DRB associated with default EPS bearer context obtained during the attach procedure.

    //@siclog "Step 42A" siclog@

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;  // @sic R5s141327 sic@

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellB, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellB,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                                                                                             v_AntennaInfo)),

                                                                        omit)));

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_CellB, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );

    // The SS transmits a TRACKING AREA UPDATE ACCEPT indicating only one EPS bearer (default EBId-1) active in the EPS bearer context status IE.

    // The EPS bearer ID linked to the additional PDN is deactivated by SS.

    //@siclog "Step 43" siclog@

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellB);

    v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellB );

    v_TaiList := cds_TAIListNonConsecutive_tlv( v_PLMN, { bit2oct(v_Tac) } );

    if (v_UpdateType == tsc_EpsUpdate_Combined_TaLaUpdate) // @sic R5s110086 sic@

    {

      v_LAI := f_EUTRA_GetCellLAI(eutra_CellB, NORMAL); // @sic R5s110176 sic@

    }

    SRB.send ( cas_SRB2_NasPdu_REQ ( eutra_CellB, // @sic R5s141327 sic@

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_TAU_Accept ( v_UpdateType,

                                                                          f_GutiParameters2MobileIdentity (tsc_IEI_Guti, v_GutiParams ),

                                                                          v_TaiList,

                                                                          cs_EPSBearerCtxtStatus_DefaultBearerOnly,

                                                                          v_LAI,

                                                                          omit, // v_MSId

                                                                          f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    // The UE transmits TRACKING AREA UPDATE COMPLETE message.

    //@siclog "Step 43AA" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB, // @sic R5s141327 sic@

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete))) -> value v_ReceivedAsp;

    // VOID   @sic R5-113188 sic@

    //@siclog "Step 43AB" siclog@

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the default EPS bearer to the additional PDN.

    // This message is included in a DLInformationTransfer message.

    //@siclog "Step 43A" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellB, // @sic R5s110593 sic@ @sic R5s141327 sic@

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             omit))));

    // Check: Does UE transmits a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43?

    //@siclog "Step 43B" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,  // @sic R5s110593 sic@ @sic R5s141327 sic@

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 43B");

    // Step 43C. The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message including the EPS bearer identity of the dedicated EPS bearer.

    //@siclog "Step 43C" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellB,  // @sic R5s110593 sic@ @sic R5s141327 sic@

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsDedicatedBearerId2,

                                                                             v_EpsTi2,

                                                                             cds_508_EPS_QoS_Dedicated_1_tlv, // @sic R5-103625 sic@

                                                                             f_EUTRA_Get_508_TrafficFlowTemplate_2_Modify(v_EpsDedicatedBearerId2))))); // @sic R5-103625 sic@

    // Check: Does UE transmits a MODIFY EPS BEARER CONTEXT REJECT message with ESM cause #43?

    //@siclog "Step 43D" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellB,  // @sic R5s110593 sic@ @sic R5s141327 sic@

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsDedicatedBearerId2,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 43D");

    //  The SS releases the RRC connection.

    //@siclog "Step 44" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);

    // EXCEPTION: Step 45 & 46 describe behaviour that depends on the UE capability.

    if ( pc_ESM_MO_Bearer_Allocation == true ) {  // @sic R5-121749 sic@

      //  Cause the UE to request bearer resource allocation of dedicated EPS bearer associated with additional PDN connectivity.

      //@siclog "Step 45" siclog@

      f_UT_RequestActivateDedicatedBearer (UT,

                                           hex2int(v_EpsDedicatedBearerId2),

                                           hex2int(v_EpsBearerId2),

                                           cs_508_EPS_QoS_Dedicated_1,

                                           f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId2),

                                           LOCAL_CNF_REQUIRED); // @sic R5s110055 sic@

      // Check: Does the UE transmit SERVICE REQUEST?

      //@siclog "Step 46" siclog@

      t_Wait.start(3.0);

      alt {

        [] SRB.receive ( car_SRB1_NasPdu_IND ( eutra_CellB,

                                               cr_NAS_Indication ( ?,

                                                                   ?)))

          {

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 46");

            t_Wait.stop;

          }

        [] t_Wait.timeout

          {

          }

      }

    }

  }


Change 10.7 - fl_TC_10_4_2_Body
	  function fl_TC_10_4_2_Body() runs on EUTRA_PTC

  {

    var NAS_MSG_Indication_Type v_NasInd;

    var ProcedureTransactionIdentifier v_EpsTi2 := tsc_PTI_Unassigned;

    var template (value) ESM_Cause v_EsmCause36 := cs_ESM_Cause_v('00100100'B); // Regular deactivation

    var SRB_COMMON_IND v_ReceivedAsp;

    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;

    var ProcedureTransactionIdentifier v_PTId_UE;

    var template (omit) ProtocolConfigOptions v_Pco;

    var template (value) AccessPointName v_APN;

    var PDN_Address v_PDN_Address;

    var PDN_Index_Type v_PdnIndex := PDN_1; /* for IMS */

    var B3_Type v_PDNType := f_GetPdnType();

    var BEARER_CONTEXT_TYPE v_BearerContextNumber := DEF_2; /* @sic R5s141375 change 4 - additional change: DEF_2 instead of DEF_1 sic@ */

    var DRB_Identity v_DRB_Id_IMS := f_EUTRA_EpsBearerAssociatedDRB(tsc_EpsDefaultBearerId);

    var template (value) DRB_ToAddModList v_DRB_ToAddModList := { cs_508_DRB_ToAddMod_DEFAULT_AM(v_DRB_Id_IMS) }; /* for IMS */

    var EPS_BearerIdentity v_EpsBearerId_IMS := tsc_EpsDefaultBearerId;

    // The UE is switched on.

    //@siclog "Step 0" siclog@

    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //@siclog "Steps 0A-0Q" siclog@

    // Steps 1 to 16c1 of the generic test procedure for UE registration (TS 36.508 4.5.2.3-1).

    // Now continue with the procedure

    v_NasInd := f_EUTRA_IdleUpdated_Step1_4 (eutra_CellA, PREAMBLE, NORMAL);

    f_EUTRA_IdleUpdated_Step5_17 (eutra_CellA, STATE2_IDLEUPDATE, NORMAL, v_NasInd, false); // up to step 16 only

    // The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the default EPS bearer.

    // Same value as in ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST for the IMS APN form preamble

    //@siclog "Step 1" siclog@

    // @sic R5s141375 change 4.2: no bearer id needs to be redefined sic@

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release (tsc_RRC_TI_Def, v_DRB_Id_IMS),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(v_EpsBearerId_IMS,

                                                                                    v_EpsTi2,

                                                                                    v_EsmCause36))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Check: Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT?

    //@siclog "Step 2" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_DeactivateEPSBearerCxtAccept(v_EpsBearerId_IMS,

                                                                                      v_EpsTi2)))); 
    f_EUTRA_SS_Release_SingleDRB(eutra_CellA, v_DRB_Id_IMS);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    //@siclog "Step 3" siclog@

    // Check: Does the UE transmit a PDN CONNECTIVITY REQUEST ?

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDNConnReqAdditionalPDN(v_PDNType,

                                                                                                                                   -,

                                                                                                                                   -,

                                                                                                                                   cr_ESMInfoTransferFlag_Any ifpresent))))

      -> value v_ReceivedAsp;

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

    v_APN := f_ESM_EvaluateAPN(eutra_CellA, secondPDN, v_PDNConnReq.accessPointName);    /* get APN without any ESM information transfer */

    v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    v_PDN_Address := f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas(v_Pco), v_PdnIndex);

    v_Pco := f_GetDefaultProtocolConfigOptions(v_Pco, v_PdnIndex);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    //@siclog "Steps 4-12 Void" siclog@

    // Reset IMS DRB


    f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB(eutra_CellA, cs_OneDRB_ConfigAM(v_DRB_Id_IMS));        // @sic R5s141375 change 4.1 sic@

    f_EUTRA_SS_RRC_RestartSecurity(eutra_CellA);                                                        // @sic R5s141375 change 4.1 sic@

    //@siclog "Step 13" siclog@

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST.

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Reconfig(tsc_RRC_TI_Def, v_DRB_ToAddModList),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_ActivateDefEpsBearerContextRequest(v_EpsBearerId_IMS,

                                                                                             v_PTId_UE,

                                                                                             v_APN,

                                                                                             v_PDN_Address,

                                                                                             -,

                                                                                             v_Pco,

                                                                                             v_BearerContextNumber))));

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // EXCEPTION: In parallel to the event described in step 14 below, if initiated by the UE the generic procedure for IP address allocation

    // in the U-plane specified in TS 36.508 subclause 4.5A.1 takes place performing IP address allocation in the U-plane.

    // EXCEPTION: In parallel to the event described in step 14 below the generic procedure for IMS signalling in the U-plane specified in table 10.4.2.3.2-2 takes place.

    //@siclog "Step 14" siclog@

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message.

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId_IMS))));

    //Wait Trigger from IMS

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA, tsc_EpsDefaultBearerId2ndPDN); // @sic R5s150172 change 1.2 sic@      

  }


Change 10.8 - fl_TC_10_8_4_Body
	  function fl_TC_10_8_4_Body() runs on EUTRA_PTC

  {

    var ProcedureTransactionIdentifier v_EpsPti := '01'O;

    var ProcedureTransactionIdentifier v_EpsPtiUnassigned := tsc_PTI_Unassigned;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var EPS_BearerIdentity v_EpsBearerId := tsc_EpsDedicatedBearerId; // bearer for additional PDN

    var template (value) ESM_Cause v_EsmCause36 := cs_ESM_Cause_v ('00100100'B);   // Regular deactivation, #36

    var template (present) ESM_Cause v_EsmCause43 := cr_ESM_Cause_v ('00101011'B); // Invalid EPS bearer identity, #43

    var template TrafficFlowTemplate v_Tft;

    var SRB_COMMON_IND v_ReceivedAsp;

    // Cause the UE to request bearer resource release of dedicated EPS bearer associated with first PDN connectivity.

    //@siclog "Step 1" siclog@

    f_UT_DeactivateBearer (UT, hex2int(v_EpsBearerId));

    //  Check: Does the UE transmit a BEARER RESOURCE MODIFICATION REQUEST message? (1,P)

    //@siclog "Step 2" siclog@

    v_Tft := cr_Tft(-, '101'B);  // Delete packet filters from existing TFT

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cdr_BearerResourceModReq(?,

                                                                               v_EpsBearerId,

                                                                               cr_ESM_Cause_tv ('00100100'B),

                                                                               v_Tft)))) -> value v_ReceivedAsp; // BEARER RESOURCE MODIFICATION REQUEST message with cause "regular deactivation", #36

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // Store PTI for SS responses

    v_PdnConnectivityRequest := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;

    v_EpsPti := v_PdnConnectivityRequest.bEARER_RESOURCE_MODIFICATION_REQUEST.procedureTransactionIdentifier;

    //  The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST

    //@siclog "Step 3" siclog@

    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_CellA,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Release (tsc_RRC_TI_Def, tsc_DRB2),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_DeactivateEPSBearerCxtReq(v_EpsBearerId,

                                                                                    v_EpsPti,

                                                                                    v_EsmCause36))));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA,

                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Check: Does the UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT message? (2,P)

    //@siclog "Step 4" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_DeactivateEPSBearerCxtAccept(v_EpsBearerId,

                                                                                      v_EpsPtiUnassigned)))); 
    f_EUTRA_SS_Release_SingleDRB(eutra_CellA, tsc_DRB2);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    // The SS transmits a MODIFY EPS BEARER CONTEXT REQUEST message

    //@siclog "Step 5" siclog@

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_ModifyEPSBearerCxtReq(v_EpsBearerId,

                                                                             v_EpsPtiUnassigned,

                                                                             cds_EPS_QoS_Dedi_nonGBR_QCI_tlv,  // @sic R5-115537 sic@

                                                                             f_EUTRA_Get_508_TrafficFlowTemplate_2_Modify(v_EpsBearerId) //ask to replace packet filters in existing TFT

                                                                             ))));

    // Check: Does the UE transmit a MODIFY EPS BEARER CONTEXT REJECT message? (2,P)

    //@siclog "Step 6" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtReject (v_EpsBearerId,

                                                                                       v_EsmCause43))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");

  }


Change 10.9 - f_EUTRA_IdleUpdated_ReleaseSecondPDN
	  function f_EUTRA_IdleUpdated_ReleaseSecondPDN(EUTRA_CellId_Type p_CellId,

                                                EPS_BearerIdentity p_EpsBearerId := tsc_EpsDefaultBearerId2ndPDN) runs on EUTRA_PTC

  {

    var DRB_Identity v_DrbForInternet := f_EUTRA_EpsBearerAssociatedDRB(p_EpsBearerId);

    var LinkedEpsBearerIdentity v_LinkedEpsBearerId := {idValue := hex2bit(p_EpsBearerId)};

    var SRB_COMMON_IND v_ReceivedAsp;

    var ProcedureTransactionIdentifier v_EpsTi;

    var B8_Type v_EsmCauseValue;

    var AccessStratumRelease v_Release;

    var template (omit) GPRS_Timer3 v_T3396 := omit;

    if (pc_UE_supports_user_initiated_PDN_disconnect) {

      // Step 1

      f_UT_DeactivateBearer(UT, hex2int(p_EpsBearerId));

      // Step 2-8

      f_EUTRA_ServiceRequestAndActivate_SRB2_DRB(p_CellId, -, -, -, -, -, -, -, -, -, -, -, -, v_DrbForInternet);  // @sic R5s150338: additional parameter p_DrbConfigAtSS sic@

      // Step 9: The UE transmits PDN DISCONNECT REQUEST

      SRB.receive(car_SRB2_NasPdu_IND(p_CellId, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDNDisconnectReq(v_LinkedEpsBearerId)))) -> value v_ReceivedAsp;

      v_EpsTi := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_DISCONNECT_REQUEST.procedureTransactionIdentifier;

      v_EsmCauseValue := tsc_ESM_Cause36_RegularDeactivation;

    }

    else {

      // Step 1-9

      f_EUTRA_GenericRbEst_Common(p_CellId, 0, 0, -, -, -, -, -, -, -, -, -, v_DrbForInternet);  // @sic R5s150338: additional parameter p_DrbConfigAtSS sic@

      v_EpsTi := tsc_PTI_Unassigned;

      v_EsmCauseValue := tsc_ESM_Cause26_InsufficientResources;

      v_Release := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease();

      select (v_Release) {

        case (rel8, rel9) {}

        case else { v_T3396 := cs_GprsTimer3(tsc_GprsTimerUnit_deact, '11111'B); }

      }

    }

    f_EUTRA_DeactivateEPS_BearerContext(p_CellId, p_EpsBearerId, v_EpsTi, v_EsmCauseValue, v_T3396);


    f_IP_Handling_Stop(IP, cs_IP_StopPDN(PDN_Internet));

    f_EUTRA_RRC_ConnectionRelease(p_CellId);

  }


