
3. Corrections to TC 8.5.4.1
Change 1 – Correction to template ‘cr_UE_EUTRA_Capability_v9a0_IEs_r11’
	Template name
	cr_UE_EUTRA_Capability_v9a0_IEs_r11 ()

	Reason for change
	As per the ASN.1 definition of UE-EUTRA-Capabillity information element in TS 36.331 clause 6.3.6, the interRAT-ParametersUTRA-v9c0, phyLayerParameters-v9d0 and rf-Parameters-v9e0 are optional. However, the UECapabilityInformation message definition in TS 36.523-1 Table 8.5.4.1.3.3-2 and its TTCN implementation suggest that presence of the UE-EUTRA-Capability-v9c0 sub-structure is expected even though the single IE elements may be empty.

(Note: change 1 requires a Prose CR).

	Summary of change
	Changed the template to allow any value or even absence of the complete sub structure with IEs interRAT-ParametersUTRA-v9c0, phyLayerParameters-v9d0 and rf-Parameters-v9e0.

	TTCN module
	8_5\RRC_OtherFunctions.ttcn

	MCC160 Comment
	


Before change

...
    template (present) UE_EUTRA_Capability_v9a0_IEs cr_UE_EUTRA_Capability_v9a0_IEs_r11 := // @sic R5s130195 sic@

  { /* @status    APPROVED (LTE) */

    featureGroupIndRel9Add_r9 := *,

    fdd_Add_UE_EUTRA_Capabilities_r9 := *,

    tdd_Add_UE_EUTRA_Capabilities_r9 := *,

    nonCriticalExtension := {

      interRAT_ParametersUTRA_v9c0 := *,

      nonCriticalExtension := {

        phyLayerParameters_v9d0 := *,

        nonCriticalExtension := {

          rf_Parameters_v9e0 := *,

          nonCriticalExtension := *

        }

      }

    }
  };
...
After change

...
  template (present) UE_EUTRA_Capability_v9a0_IEs cr_UE_EUTRA_Capability_v9a0_IEs_r11 := // @sic R5s130195 sic@

  { /* @status    APPROVED (LTE) */

    featureGroupIndRel9Add_r9 := *,

    fdd_Add_UE_EUTRA_Capabilities_r9 := *,

    tdd_Add_UE_EUTRA_Capabilities_r9 := *,

    nonCriticalExtension := *
  };
... 
Change 2 – Correction to function ‘f_CheckUE_SupportedEUTRA_Bands’
	Function name
	f_CheckUE_SupportedEUTRA_Bands ()

	Reason for change
	f_CheckUE_SupportedEUTRA_Bands_Chunk is modified and needs to be invoked as per the new version.

	Summary of change
	Modified the calls of  f_CheckUE_SupportedEUTRA_Bands_Chunk as needed for the new version.

	TTCN module
	EUTRA_CapabilityFunctions.ttcn

	MCC160 Comment
	


Before change

...
  function f_CheckUE_SupportedEUTRA_Bands(UE_EUTRA_Capability p_ReceivedCapabilityMsg)

  {

    //@sic R5s140566 sic@

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;

    var integer i;

    var integer v_BandEUTRA;

    var integer v_BitStringIndex;

    var bitstring v_SuppEutraBandsAccToPICS := f_DeriveSuppEutraBandsFromPics();

    var integer v_NumEutraBandsReported;

    var bitstring v_SuppEutraBandsAccToPICS_Chunk;

    var integer v_BitStringIndexReported := 0; // index in reported bands list
    v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;

    v_NumEutraBandsReported := lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA);

    // It is no longer checked that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    // There may be more bands reported than there are PICS defined for bands

    // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS

    // Check only the bands for which there are PICS; do not check other bands which may or may not be reported

    v_SuppEutraBandsAccToPICS_Chunk := substr(v_SuppEutraBandsAccToPICS, 0, 14); // bands 1 - 14

    v_BitStringIndexReported := f_CheckUE_SupportedEUTRA_Bands_Chunk (v_ReceivedCapabilityMsg, v_SuppEutraBandsAccToPICS_Chunk, 0, v_BitStringIndexReported, 14);

    if (v_BitStringIndexReported != 0) {    

      v_SuppEutraBandsAccToPICS_Chunk := substr(v_SuppEutraBandsAccToPICS, 16, 15); // bands 17 - 31

      v_BitStringIndexReported := f_CheckUE_SupportedEUTRA_Bands_Chunk (v_ReceivedCapabilityMsg, v_SuppEutraBandsAccToPICS_Chunk, 16, v_BitStringIndexReported,15);

    } 

    if (v_BitStringIndexReported != 0) {    

      v_SuppEutraBandsAccToPICS_Chunk := substr(v_SuppEutraBandsAccToPICS, 32, 11); // bands 33 - 43

      v_BitStringIndexReported := f_CheckUE_SupportedEUTRA_Bands_Chunk (v_ReceivedCapabilityMsg, v_SuppEutraBandsAccToPICS_Chunk, 32, v_BitStringIndexReported, 11);

    }
    // interFreqEUTRA-BandList SEQUENCE (SIZE (1..maxBands)) OF SEQUENCE {

...

  } // f_CheckUE_SupportedEUTRA_Bands

...

After change

...
  function f_CheckUE_SupportedEUTRA_Bands(UE_EUTRA_Capability p_ReceivedCapabilityMsg)

  {

    //@sic R5s140566 sic@

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;

    var integer i;

    var integer v_BandEUTRA;

    var integer v_BitStringIndex;

    var bitstring v_SuppEutraBandsAccToPICS := f_DeriveSuppEutraBandsFromPics();

    var integer v_NumEutraBandsReported;

    var bitstring v_SuppEutraBandsAccToPICS_Chunk;

    v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;

    v_NumEutraBandsReported := lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA);

    // It is no longer checked that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    // There may be more bands reported than there are PICS defined for bands

    // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS

    // Check only the bands for which there are PICS; do not check other bands which may or may not be reported

    v_SuppEutraBandsAccToPICS_Chunk := substr(v_SuppEutraBandsAccToPICS, 0, 14); // bands 1 - 14

    f_CheckUE_SupportedEUTRA_Bands_Chunk (v_ReceivedCapabilityMsg, v_SuppEutraBandsAccToPICS_Chunk, 0, 14);

    v_SuppEutraBandsAccToPICS_Chunk := substr(v_SuppEutraBandsAccToPICS, 16, 15); // bands 17 - 31

    f_CheckUE_SupportedEUTRA_Bands_Chunk (v_ReceivedCapabilityMsg, v_SuppEutraBandsAccToPICS_Chunk, 16, 15);

    v_SuppEutraBandsAccToPICS_Chunk := substr(v_SuppEutraBandsAccToPICS, 32, 11); // bands 33 - 43

    f_CheckUE_SupportedEUTRA_Bands_Chunk (v_ReceivedCapabilityMsg, v_SuppEutraBandsAccToPICS_Chunk, 32, 11);
    // interFreqEUTRA-BandList SEQUENCE (SIZE (1..maxBands)) OF SEQUENCE {
... 

  } // f_CheckUE_SupportedEUTRA_Bands

... 

... 
Change 3 – Correction to function ‘f_CheckUE_SupportedEUTRA_Bands_Chunk’
	Function name
	f_CheckUE_SupportedEUTRA_Bands_Chunk ()

	Reason for change
	It is assumed that the supported EUTRA bands are provided in ascending order. Therefore, the “chunks” of defined PICS for EUTRA bands are compared against “chunks” of reported bands.

	Summary of change
	Compared the “chunks” of defined PICS for EUTRA bands against the complete list of reported bands as the supported EUTRA bands may be provided in any order.

	TTCN module
	EUTRA_CapabilityFunctions.ttcn

	MCC160 Comment
	


Before change

...
  function  f_CheckUE_SupportedEUTRA_Bands_Chunk(UE_EUTRA_Capability p_ReceivedCapabilityMsg,

                                                 bitstring p_DerivedChunk,

                                                 integer p_OffsetPICS,

                                                 integer p_OffsetReport,

                                                 integer p_LengthPICS) return integer
  {

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;

    var bitstring v_DerivedChunk := p_DerivedChunk;

    var integer v_OffsetPICS := p_OffsetPICS;

    var integer v_OffsetReport := p_OffsetReport;

    var integer v_LengthPICS := p_LengthPICS;

    var integer i;

    var integer v_BandEUTRA;

    var integer v_BitStringIndex;

    var bitstring v_SuppEutraBandsReported := int2bit(0, 64); // 1... maxbands

    var integer v_MaxBandsReported := lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA);

    // It is no longer checked that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    // There may be more bands reported than there are PICS defined for bands

    // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS

    for (i := 0; i < v_LengthPICS; i := i + 1) {

      //It is assumed that UE reports the bands in order in the list

      if ((i + v_OffsetReport) >= v_MaxBandsReported) {return  0}; // there are no more bands reported which need to be checked

      v_BandEUTRA := v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[i + v_OffsetReport].bandEUTRA;

      if (v_BandEUTRA > (v_OffsetPICS + v_LengthPICS)) { return  (i + v_OffsetReport)}; // if the bands reported become greater than the range of bands checked we are done @sic R5s150194 sic@      

      v_BitStringIndex := v_BandEUTRA - 1;

      if (not f_Bitstring_BitIsSet(v_DerivedChunk, v_BitStringIndex - v_OffsetPICS)) { // @sic R5s150194 sic@

        f_SetVerdict(fail, __FILE__, __LINE__, "PICS does not match reported capability for band nr. " & int2str(v_BandEUTRA));  // @sic R5-133612 sic@

      } else {

        if (f_Bitstring_BitIsSet(v_SuppEutraBandsReported, v_BitStringIndex)) {

          f_SetVerdict(fail, __FILE__, __LINE__, "support of band nr. " & int2str(v_BandEUTRA) &  " is reported more than once");  // @sic R5-133612 sic@

        }

        v_SuppEutraBandsReported[v_BitStringIndex] := '1'B;

      }

      // NOTE:  number of bands reported by the UE and number of bands acc. to PICS setting may not be equal. None of the bands is reported twice by the UE.

    }

    return  (i + v_OffsetReport);

  }
...

After change

...
  function f_CheckUE_SupportedEUTRA_Bands_Chunk(UE_EUTRA_Capability p_ReceivedCapabilityMsg,

                                                bitstring p_DerivedChunk,

                                                integer p_OffsetPICS,

                                                integer p_LengthPICS)

  {

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;

    var bitstring v_DerivedChunk := p_DerivedChunk;

    var integer v_OffsetPICS := p_OffsetPICS;

    var integer v_LengthPICS := p_LengthPICS;

    var integer i,j;

    var integer v_BandEUTRA;

    var integer v_BitStringIndex;

    var bitstring v_SuppEutraBandsReported := int2bit(0, 64); // 1... maxbands

    var integer v_MaxBandsReported := lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA);

    // It is no longer checked that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    // There may be more bands reported than there are PICS defined for bands

    // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS

    for (i := 0; i < v_LengthPICS; i := i + 1) {

      //It is assumed that UE reports the bands in any order => compare against the complete list

      if (v_DerivedChunk [i] == '1'B) { // PICS for band i in chunk + 1 is set

        for (j := 0; j < v_MaxBandsReported; j := j + 1) { // check against all reported bands

          v_BandEUTRA := v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[j].bandEUTRA;

          v_BitStringIndex := v_BandEUTRA - 1;

          if ( (i + v_OffsetPICS) == v_BitStringIndex) {

            if (v_SuppEutraBandsReported[v_BitStringIndex] == '1'B) {

              f_SetVerdict(fail, __FILE__, __LINE__, "support of band nr. " & int2str(v_BandEUTRA) &  " is reported more than once"); // error: duplicate found              

            } else {

              v_SuppEutraBandsReported[v_BitStringIndex] := '1'B;

            };

          } 

          else {// do nothing

          }

        }

        if (v_SuppEutraBandsReported[i + v_OffsetPICS] != '1'B) { // PICS set,  but expected band was not in the list

          f_SetVerdict(fail, __FILE__, __LINE__, "expected (per PICS) band nr. " & int2str(i + v_OffsetPICS + 1) &  " is not reported"); // error: PICS set, but expected band not reported              

        }        

      }        

      else { // PICS is not set => non support

        for (j := 0; j < v_MaxBandsReported; j := j + 1) { // check against all reported bands

          v_BandEUTRA := v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[j].bandEUTRA;

          v_BitStringIndex := v_BandEUTRA - 1;

          if ( (i + v_OffsetPICS) == v_BitStringIndex) {

            f_SetVerdict(fail, __FILE__, __LINE__, "PICS does not match reported capability for band nr. " & int2str(v_BandEUTRA)); // PICS mismatch

          } 

          else { // do nothing 

          }

        }

      }             

      // NOTE:  number of bands reported by the UE and number of bands acc. to PICS setting may not be equal. None of the bands is reported twice by the UE.

    }

  } // f_CheckUE_SupportedEUTRA_Bands_Chunk

... 
