
Change 1
	Function name
	f_CheckUE_SupportedEUTRA_Bands_Chunk

	Reason for change
	· The range of bands to be checked is incremented every iteration, this should be set according the input parameters of the function. Current implementation causes problems as the function does not leave when it is supposed.
· Current implementation will fail when the function checks for bands higher than band 14. The v_OffsetPICS needs so be subtracted from v_BitStringIndex so we will compare correct bits, otherwise it will cause issue (for example with band 18).

	Summary of change
	In the if statements  replace  i parameter with v_OffsetPICS to handle that range of bands will be checked according to the input parameter.
Replace v_BitStringIndex with v_BitStringIndex - v_OffsetPICS to compare a correct bits during the checking bands

	TTCN module
	Common\IRAT\EUTRA_CapabilityFunctions.ttcn


Before change

      function  f_CheckUE_SupportedEUTRA_Bands_Chunk(UE_EUTRA_Capability p_ReceivedCapabilityMsg,

                                                 bitstring p_DerivedChunk,

                                                 integer p_OffsetPICS,

                                                 integer p_OffsetReport,

                                                 integer p_LengthPICS) return integer

  {

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;

    var bitstring v_DerivedChunk := p_DerivedChunk;

    var integer v_OffsetPICS := p_OffsetPICS;

    var integer v_OffsetReport := p_OffsetReport;

    var integer v_LengthPICS := p_LengthPICS;

    var integer i;

    var integer v_BandEUTRA;

    var integer v_BitStringIndex;

    var bitstring v_SuppEutraBandsReported := int2bit(0, 64); // 1... maxbands

    var integer v_MaxBandsReported := lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA);

    // It is no longer checked that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    // There may be more bands reported than there are PICS defined for bands

    // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS

    for (i := 0; i < v_LengthPICS; i := i + 1) {

      //It is assumed that UE reports the bands in order in the list

      if ((i + v_OffsetReport) >= v_MaxBandsReported) {return  0}; // there are no more bands reported which need to be checked

      v_BandEUTRA := v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[i + v_OffsetReport].bandEUTRA;

      if (v_BandEUTRA > (i + v_LengthPICS)) { return  (i + v_OffsetReport)}; // if the bands reported become greater than the range of bands checked we are done

      v_BitStringIndex := v_BandEUTRA - 1;

      if (not f_Bitstring_BitIsSet(v_DerivedChunk, v_BitStringIndex)) {

        f_SetVerdict(fail, __FILE__, __LINE__, "PICS does not match reported capability for band nr. " & int2str(v_BandEUTRA));  // @sic R5-133612 sic@

      } else {

        if (f_Bitstring_BitIsSet(v_SuppEutraBandsReported, v_BitStringIndex)) {

          f_SetVerdict(fail, __FILE__, __LINE__, "support of band nr. " & int2str(v_BandEUTRA) &  " is reported more than once");  // @sic R5-133612 sic@

        }

        v_SuppEutraBandsReported[v_BitStringIndex] := '1'B;

      }

      // NOTE:  number of bands reported by the UE and number of bands acc. to PICS setting may not be equal. None of the bands is reported twice by the UE.

    }

    return  (i + v_OffsetReport);

  }
After change

function  f_CheckUE_SupportedEUTRA_Bands_Chunk(UE_EUTRA_Capability p_ReceivedCapabilityMsg,

                                                 bitstring p_DerivedChunk,

                                                 integer p_OffsetPICS,

                                                 integer p_OffsetReport,

                                                 integer p_LengthPICS) return integer

  {

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;

    var bitstring v_DerivedChunk := p_DerivedChunk;

    var integer v_OffsetPICS := p_OffsetPICS;

    var integer v_OffsetReport := p_OffsetReport;

    var integer v_LengthPICS := p_LengthPICS;

    var integer i;

    var integer v_BandEUTRA;

    var integer v_BitStringIndex;

    var bitstring v_SuppEutraBandsReported := int2bit(0, 64); // 1... maxbands

    var integer v_MaxBandsReported := lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA);

    // It is no longer checked that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    // There may be more bands reported than there are PICS defined for bands

    // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS

    for (i := 0; i < v_LengthPICS; i := i + 1) {

      //It is assumed that UE reports the bands in order in the list

      if ((i + v_OffsetReport) >= v_MaxBandsReported) {return  0}; // there are no more bands reported which need to be checked

      v_BandEUTRA := v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[i + v_OffsetReport].bandEUTRA;

      if (v_BandEUTRA > (v_OffsetPICS + v_LengthPICS)) { return  (i + v_OffsetReport)}; // if the bands reported become greater than the range of bands checked we are done  //Change 1
      v_BitStringIndex := v_BandEUTRA - 1;

      if (not f_Bitstring_BitIsSet(v_DerivedChunk, v_BitStringIndex - v_OffsetPICS)) { //Change2
        f_SetVerdict(fail, __FILE__, __LINE__, "PICS does not match reported capability for band nr. " & int2str(v_BandEUTRA));  // @sic R5-133612 sic@

      } else {

        if (f_Bitstring_BitIsSet(v_SuppEutraBandsReported, v_BitStringIndex)) {

          f_SetVerdict(fail, __FILE__, __LINE__, "support of band nr. " & int2str(v_BandEUTRA) &  " is reported more than once");  // @sic R5-133612 sic@

        }

        v_SuppEutraBandsReported[v_BitStringIndex] := '1'B;

      }

      // NOTE:  number of bands reported by the UE and number of bands acc. to PICS setting may not be equal. None of the bands is reported twice by the UE.

    }

    return  (i + v_OffsetReport);

  }  

